

EAB 770 - Chapter 2

Miguel Fudolig

1 Chapter 2: 2×2 tables

1.1 Outline

- Contingency Tables
- Chi-Square Statistics
- Exact Tests
- Difference in Proportions
- Odds Ratio and Relative Risk
- Sensitivity and Specificity
- McNemar's Test
- Incidence Densities
- Power Analysis

2 Contingency Tables

2.1 Contingency Tables

- Summarizes categorical data
 - Response variable has two levels
 - Exploratory variable has two levels
- Example:
 - Treatment/Placebo in RCT
 - Case/Control in a retrospective study

Risk\Case	Case	Control	Total
Present			
Absent			
Total			

2.2 Contingency Tables

- Main Objective: Determine whether there is an association between the row and column variables.
 - Is the risk variable associated with the prevalence rate?
- Recall: Intro to Statistics
 - Chi-square tests of homogeneity H_0 : The two populations (rows) have the same distribution of the column variable
 - Chi-square tests of independence H_0 : The row and column variables are independent of each other.

2.3 Contingency Tables: Convention

The indices of contingency tables denote the row and column of each number.

- n_{xy} denotes the number at row x and column y
- Seeing a (+) sign indicates that it is a marginal sum.
 - n_{x+} denotes the sum of all columns in row x .
 - n_{+y} denotes the sum of all rows in column y .
- The placement of the + sign denotes the margin in which it was summed.
- n is the total sample size.

Row\Col	C ₁	C ₂	Total
R ₁	n_{11}	n_{12}	n_{1+}
R ₂	n_{21}	n_{22}	n_{2+}
Total	n_{+1}	n_{+2}	n

3 Chi-square Statistics

3.1 Chi-square Statistics

- The null hypothesis of no association between row and column variables imply that the totals are fixed.
- The probabilities of realization of each cell is determined by the [hypergeometric distribution](#).

3.2 Chi-square Statistics

- Recall: Pearson Chi-square statistics

$$Q_P = \sum_{i=1}^2 \sum_{j=1}^2 \frac{(n_{ij} - m_{ij})^2}{m_{ij}}$$

i Note

m_{ij} is the expected value of the number of observations in row i and column j , n_{ij} .

3.3 Mantel-Haenszel Chi-square statistic

- Another form of the Chi-square statistic is the Mantel-Haenszel chi-square statistic, Q . For 2×2 contingency tables,

$$Q = \frac{(n_{11} - m_{11})^2}{v_{11}}$$

$$m_{11} = n_{1+}n_{+1}/n; v_{11} = \frac{n_{1+}n_{2+}n_{+1}n_{+2}}{n^2(n-1)}$$

i Note

This statistic is based on the condition that all marginal totals are fixed and the probability distribution of random allocation of units to groups is given by the hypergeometric distribution. m_{11} and v_{11} are the mean and variance of the hypergeometric distribution based on the 2×2 table, respectively.

i Note

Q is also known as the randomization chi-square.


3.4 Pearson Chi-square vs. Mantel-Haenszel Chi-square

- MH Chi-square is more powerful than Pearson Chi-square
- MH Chi-square tests need less samples than Pearson Chi-square, in general.
- MH Chi-square tests can control for confounding variables.

$$Q_P = \frac{n}{n-1}Q$$

3.5 Chi-square distribution

- The chi-square distribution, χ_k^2 , is defined by the number of degrees of freedom k , where $k \geq 1$.
- Properties include:
 - The value of χ^2 can only assume non-negative values.
 - The χ^2 distribution is asymmetric.
 - The sum of two or more independent chi-square variables also follows a chi-square distribution.
 - Mean: k , Variance: $2k$



3.6 Chi-square distribution

For large n , both the MH/randomization chi-square and the Pearson chi-square approach a chi-square distribution with 1 degree of freedom (χ_1^2).

⚠ Warning

Rule of thumb: The expected value for each cell should be at least 5, i.e. $m_{ij} \geq 5$ for all i, j


3.7 Hypothesis testing

The null hypothesis of no association states that there is no association between row and column variables. - Assuming this is true and assumptions are satisfied, Q and Q_P follow a χ_1^2 distribution for 2×2 tables.

i Note

The p-values are given by $P(\chi^2 > Q_P)$ and $P(\chi^2 > Q)$ for Pearson and MH Chi-squares, respectively.

3.8 Hypothesis Testing: SAS



3.9 FREQ procedure

- The **weight** statement tells SAS if there is an aggregated count variable present in the data set.
- The **tables** statement tells SAS to create a contingency table using the stated variables.
 - Under tables, you can include the **cmh** or the **chisq** option to calculate the randomization and Pearson chi-square values.

Below is a sample code.

```
proc freq;  
weight count;  
tables a*b/ cmh chisq;  
run;
```

3.10 Example

The table shows the information from a randomized clinical trial that compared an experimental treatment to a placebo treatment for a respiratory disorder.

- What is the Pearson chi-square value?
- What is the Mantel-Haenszel chi-square value?
- Is there strong evidence of an association between treatment and outcome?

Treatment	Outcome	
	Favorable	Unfavorable
Placebo	16	48
Test	40	20

3.11 Example

The table shows the information from a randomized clinical trial that compared an experimental treatment to a placebo treatment for a respiratory disorder.

- What is the Pearson chi-square value? **21.7**
- What is the Mantel-Haenszel chi-square value? **21.53**
- Is there strong evidence of an association between treatment and outcome?
 - $p < 0.0001$; There is strong evidence of an association between the test treatment and the outcome.

3.11.1 SAS Code

```
1 SAScode <- "  
2 data respire;  
3 input treat $ outcome $ count;  
4 datalines;  
5 test f 40  
6 test u 20  
7 placebo f 16  
8 placebo u 48  
9 ;  
10  
11 proc freq data=respire order=data;  
12 weight count;
```

```

13 tables treat*outcome/chisq;
14 run;
15 "

```

3.11.2 R Code

```

dat <- expand.grid(Treatment=c("Placebo","Test"),
                  Outcome = c("Favorable","Unfavorable")) |>
mutate(count = c(16,40,48,20))

xtabs(count ~ .,data=dat) -> dat_cc

```

Pearson Chi-square (without continuity adjustment)

```

#install.packages("tidyverse")
library(tidyverse)
dat <- expand.grid(Treatment=c("Placebo","Test"),
                  Outcome = c("Favorable","Unfavorable")) |>
mutate(count = c(16,40,48,20))

xtabs(count ~ .,data=dat)

```

	Outcome	
Treatment	Favorable	Unfavorable
Placebo	16	48
Test	40	20

```
chisq.test(dat_cc, correct = F)
```

Pearson's Chi-squared test

```

data: dat_cc
X-squared = 21.709, df = 1, p-value = 3.174e-06

```

Randomization Chi-square

	Chisq	Df	Prob
cor	21.53361	1	3.476808e-06
rmeans	21.53361	1	3.476808e-06
cmeans	21.53361	1	3.476808e-06
general	21.53361	1	3.476808e-06

```
# install.packages("vcdExtra")
CMH_CC <- vcdExtra::CMHtest(dat_cc)
CMH_CC$table |> as.data.frame() |> gt::gt(rownames_to_stub = T)
```

3.12 Exercise

Suppose we wish to determine if the frequency of falls is related to wheelchair use for Polio survivors. The data is shown in the table.

- Write the SAS code to enter the data and analyze using the FREQ procedure.
- What is the Pearson chi-square value?
- What is the Mantel-Haenszel chi-square value?
- Using a significance level of 0.05, do we have sufficient evidence to warrant the conclusion that wheelchair use and falling are associated?

```
# install.packages("vcdExtra")
dat <- expand.grid(Fall=c("Fallers","Nonfallers"),
                  Wheelchair = c("Yes","No")) |>
mutate(count = c(62,18,121,32))

xtabs(count ~ .,data=dat) -> dat_cc
dat_cc
```

	Wheelchair	
Fall	Yes	No
Fallers	62	121
Nonfallers	18	32

3.13 Exercise

Suppose we wish to determine if the frequency of falls is related to wheelchair use for Polio survivors. The data is shown in the table.

- Write the SAS code to enter the data and analyze using the FREQ procedure.
- What is the Pearson chi-square value? **0.0783**
- What is the Mantel-Haenszel chi-square value? **0.0780**
- Using a significance level of 0.05, do we have sufficient evidence to warrant the conclusion that wheelchair use and falling are associated?

– $p = 0.78$, no evidence that wheelchair use and falling are associated.

3.13.1 SAS

```

1 data wheelchair;
2 input fall $ wheelchair $ count;
3 datalines;
4 f yes 62
5 n yes 18
6 f no 121
7 n no 32
8 ;
9
10 proc freq data=wheelchair order=data;
11 weight count;
12 tables fall*wheelchair/chisq;
13 run;

```

3.13.2 R

Pearson Chi-square (without continuity adjustment)

```

#install.packages("tidyverse")

chisq.test(dat_cc, correct = F)

```

Pearson's Chi-squared test

```

data: dat_cc
X-squared = 0.078299, df = 1, p-value = 0.7796

```

Randomization Chi-square

	Chisq	Df	Prob
cor	0.07796307	1	0.780077
rmeans	0.07796307	1	0.780077
cmeans	0.07796307	1	0.780077
general	0.07796307	1	0.780077

```
# install.packages("vcdExtra")
CMH_CC <-vcdExtra::CMHtest(dat_cc)
CMH_CC$table |> as.data.frame()|> gt::gt(rownames_to_stub = T)
```

4 Exact Methods

4.1 Fisher's Exact Test

The chi-square distribution assumption for Q and Q_P includes a large sample assumption.

Warning

What if $m_{ij} < 5$?

The **Fisher's Exact Test** can be used to analyze these contingency tables.

4.2 Fisher's Exact Test

Fisher's Exact Test is based on the hypergeometric distribution.

- Calculates an *exact* p-value, unlike the chi-square test which only uses an approximation.
- Easy to calculate for 2×2 and total sample sizes below 50, but computationally expensive as the number increases.

4.3 Hypergeometric distribution

The hypergeometric distribution describes the probability of the cell count for the (i,j) cell in a 2×2 table assuming the null hypothesis of no association assuming the null hypothesis of no association is true.

$$P(n_{11}) = \frac{n_{1+}!n_{2+}!n_{+1}!n_{+2}!}{n!n_{11}!n_{12}!n_{21}!n_{22}!}$$

- One of margins can be regarded as fixed depending on the treatment assignment/sampling method.
- To get the **exact** p-value, we sum the probabilities of tables that are as *likely or less likely* given the fixed margins.

4.4 Fisher's Exact Test: Hypotheses

Null hypothesis H_0 : there is no association between the row and column variables

- One sided H_a : There is a positive/negative association between the row and column variables. OR One combination of the row and column variable levels is more favored compared to the others.
- Two-sided H_a : there is an association between the row and column variables.

Note

Only add probabilities of outcomes that have lower probabilities than the observed value based on the hypergeometric distribution.

4.5 Fisher's Exact Test: SAS

Including the **exact** or the **chisq** option under the **tables** statement will yield the results of the exact test.

```
proc freq;
weight count;
tables a*b/ exact chisq;
run;
```

4.6 Example

The table shown shows data from a study on treatments for healing severe infections. Calculate the exact p-value of this contingency table using the Fisher's Exact Test.

- Under the alternative hypothesis: The test treatment is positively associated with the outcome.
- Under the alternative hypothesis: The test treatment is associated with the outcome.
- At a significance level of 0.10, do you have enough evidence to conclude that the treatment is associated with the outcome?

	Outcome	
Treatment	Favorable	Unfavorable
Test	10	2
Control	2	4

4.7 Example

4.7.1 SAS Code

```
data severe;
input treat $ outcome $ count;
datalines;
Test f 10
Test u 2
Control f 2
Control u 4
;
proc freq order=data data=severe;
weight count;
tables treat*outcome / chisq nocol;
run;
```

4.7.2 Manual Calculation through R

- One-sided hypothesis (+)

```
c(`Prob_n11=10`=factorial(12)^2*factorial(6)^2/(factorial(18)*factorial(10)*factorial(2)^2*f
`Prob_n11=11`= factorial(12)^2*factorial(6)^2/(factorial(18)*factorial(11)*factorial(1)^2*f
`Prob_n11=12`= factorial(12)^2*factorial(6)^2/(factorial(18)*factorial(12)*factorial(0)^2*f
```

```
Prob_n11=10 Prob_n11=11 Prob_n11=12
0.0533290239 0.0038784745 0.0000538677
```

```
factorial(12)^2*factorial(6)^2/(factorial(18)*factorial(10)*factorial(2)^2*factorial(4)) + #P(
factorial(12)^2*factorial(6)^2/(factorial(18)*factorial(11)*factorial(1)^2*factorial(5)) + #P(
factorial(12)^2*factorial(6)^2/(factorial(18)*factorial(12)*factorial(0)^2*factorial(6)) #P(
```

```
[1] 0.05726137
```

- Two-sided hypothesis (needs to include values < 10.)

```
c(`Prob_n11=10`=factorial(12)^2*factorial(6)^2/(factorial(18)*factorial(10)*factorial(2)^2*f
`Prob_n11=11` = factorial(12)^2*factorial(6)^2/(factorial(18)*factorial(11)*factorial(1)^2*f
`Prob_n11=12` = factorial(12)^2*factorial(6)^2/(factorial(18)*factorial(12)*factorial(0)^2*f
`Prob_n11=9`  = factorial(12)^2*factorial(6)^2/(factorial(18)*factorial(9)*factorial(3)^2*f
`Prob_n11=8`  = factorial(12)^2*factorial(6)^2/(factorial(18)*factorial(8)*factorial(4)^2*f
`Prob_n11=7`  = factorial(12)^2*factorial(6)^2/(factorial(18)*factorial(7)*factorial(5)^2*f
`Prob_n11=6`  = factorial(12)^2*factorial(6)^2/(factorial(18)*factorial(6)*factorial(6)^2*f
```

```
      Prob_n11=10  Prob_n11=11  Prob_n11=12  Prob_n11=9  Prob_n11=8  Prob_n11=7
0.0533290239 0.0038784745 0.0000538677 0.2370178841 0.3999676794 0.2559793148
      Prob_n11=6
0.0497737557
```

```
factorial(12)^2*factorial(6)^2/(factorial(18)*factorial(10)*factorial(2)^2*factorial(4)) + #
factorial(12)^2*factorial(6)^2/(factorial(18)*factorial(11)*factorial(1)^2*factorial(5)) + #
factorial(12)^2*factorial(6)^2/(factorial(18)*factorial(12)*factorial(0)^2*factorial(6)) + #
factorial(12)^2*factorial(6)^2/(factorial(18)*factorial(6)*factorial(6)^2*factorial(0)) #P($
```

```
[1] 0.1070351
```

4.7.3 Answers

- Under the alternative hypothesis: The test treatment is positively associated with the outcome. **p=0.0573**
- Under the alternative hypothesis: The test treatment is associated with the outcome. **p=0.1070**
- At a significance level of 0.10, do you have enough evidence to conclude that the treatment is associated with the outcome? **At $\alpha = 0.10$, we do not have sufficient evidence to conclude that treatment is associated with the outcomes in healing severe infections.**

Note

Be careful when selecting which one-sided alternative hypothesis to interpret from the results!

4.8 Exercise

Fisher's experiment: A number of individuals gathered together for afternoon tea. A lady in the group said that she could differentiate between whether the milk or the tea was poured

first into a cup. Fisher suggested to pour tea first into four cups and to pour milk first into four cups. The lady was blinded to which cups had milk or tea poured first, but she knew there were four of each. The contingency table is shown below. At $\alpha = 0.05$, do we have sufficient evidence to conclude that the lady can differentiate tea based on how they were poured?

```
dat <- expand.grid(Actual=c("Milk","Tea"),
                  Guess = c("Milk","Tea")) |>
mutate(count = c(3,1,1,3))

xtabs(count ~ .,data=dat) -> dat_cc
dat_cc
```

	Guess	
Actual	Milk	Tea
Milk	3	1
Tea	1	3

4.9 Exercise

At $\alpha = 0.05$, do we have sufficient evidence to conclude that the lady can differentiate tea based on how they were poured?

- The data can be analyzed through SAS or mathematically through the hypergeometric distribution
- Use one-sided alternative hypothesis because direction is implied in the problem. Thus we consider $P(n_{11}=3) + P(n_{11} = 4)$.
- $P(n_{11}=3) + P(n_{11} = 4) = 0.229+0.014 = 0.243$

4.9.1 SAS Code

```
data tea;
input actual$ guess $ count;
datalines;
milk milk 3
milk tea 1
tea milk 1
tea tea 3
;

proc freq data=tea order=data;
```

```
weight count;
tables actual*guess/chisq;
run;
```

4.9.2 R (Math)

```
c(`Prob_n11=3`=factorial(4)^4/(factorial(8)*factorial(3)^2),
`Prob_n11=4`=factorial(4)^4/(factorial(8)*factorial(4)^2) )
```

```
Prob_n11=3 Prob_n11=4
0.22857143 0.01428571
```

```
factorial(4)^4/(factorial(8)*factorial(3)^2) + #P($n_11$=3)
factorial(4)^4/(factorial(8)*factorial(4)^2) # P($n_11$ = 4)
```

```
[1] 0.2428571
```

5 Difference in Proportions

5.1 Difference in Proportions

- We are often interested in comparing two independent populations
- Categorical response: proportions
- Recall from Intro to Stats course: Difference of two proportions
- Confidence interval using the standard normal distribution
- Z-test to test for a significant difference in proportions of two populations

5.2 Difference in Proportions: Estimation

- Independence of the two populations (Groups 1 and 2) mean that we have two independent binomial distributions.
- Estimating the difference of the two proportions involves calculating the confidence intervals.
- There are different methods to calculate the confidence interval for the difference in proportion.

Table 2.5 2×2 Contingency Table

	Yes	No	Total	Proportion Yes
Group 1	n_{11}	n_{12}	n_{1+}	$p_1 = n_{11}/n_{1+}$
Group 2	n_{21}	n_{22}	n_{2+}	$p_2 = n_{21}/n_{2+}$
Total	n_{+1}	n_{+2}	n	

5.3 Confidence Intervals: Wald

The commonly used confidence interval in estimating difference in proportions is the **Wald** confidence interval.

$$p_1 - p_2 \pm z_{1-\alpha/2} \sqrt{\frac{p_1(1-p_1)}{n_{1+}} + \frac{p_2(1-p_2)}{n_{2+}}}$$

Warning

This CI fails when p_1 or p_2 are close to the extreme values (0 and 1). this CI does not account for the discreteness of the data.

5.4 Confidence Intervals: Corrected Wald

We introduce a correction term to the Wald confidence interval to account for the continuity of the data.

$$p_1 - p_2 \pm z_{1-\alpha/2} \sqrt{\frac{p_1(1-p_1)}{n_{1+}} + \frac{p_2(1-p_2)}{n_{2+}}} + 1/2(1/n_{1+} + 1/n_{2+})$$

Warning

This CI still fails when p_1 or p_2 are close to the extreme values (0 and 1).

5.5 Confidence Intervals: Agresti-Caffo

We introduce a correction term to the Wald confidence interval to account for the extreme values.

$$p_1 - p_2 \pm z_{1-\alpha/2} \sqrt{\frac{\tilde{p}_1(1-\tilde{p}_1)}{n_{1+} + 2} + \frac{\tilde{p}_2(1-\tilde{p}_2)}{n_{2+} + 2}}$$

where

$$\tilde{p}_1 = (n_{11} + 1)/(n_{1+} + 2); \tilde{p}_2 = (n_{21} + 1)/(n_{2+} + 2)$$

i Note

This correction keeps the coverage of the confidence interval at 95% for all possible values of p_1 and p_2 .

5.6 Confidence Intervals: Other

Confidence Intervals	SAS code	Adjustment Type
Wald	<code>wald</code>	Default
Corrected Wald	<code>correct</code>	Continuity correction
Agresti-Caffo	<code>ac</code>	Improved Coverage
Miettinen-Nurminen	<code>mn</code>	Uses goodness-of-fit to improve coverage
Newcombe	<code>newcombe</code>	hybrid interval that does well for small counts
Hauck-Anderson	<code>ha</code>	Accounts for row totals
Exact	<code>exact</code>	Maximizes p-value overall possible values

5.7 Confidence Intervals: SAS

- The FREQ procedure has all mentioned confidence intervals
 - The options included in the `tables` statement are included in the previous slide.
 - The `riskdiff` option of the `tables` statement allows us to specify which confidence intervals we need for the difference as well as the estimates for the group proportions.
 - The `alpha` option can be used to change the confidence level.

Sample SAS code:

```
1 proc freq;
2 weight count;
3 tables treat*outcome/riskdiff(correct cl=(mn ac)) alpha=0.05;
4 run;
```

5.8 Example

The contingency table shows the information from a randomized clinical trial that compared an experimental treatment to a placebo treatment for a respiratory disorder. Calculate the 95% confidence interval for the difference in proportion of favorable outcomes for both treatment groups.

- Wald
- Corrected Wald
- Agresti-Caffo
- Miettinen-Nurminen

Treatment	Outcome	
	Favorable	Unfavorable
Placebo	16	48
Test	40	20

5.9 Example

The contingency table shows the information from a randomized clinical trial that compared an experimental treatment to a placebo treatment for a respiratory disorder. Calculate the 95% confidence interval for the difference in proportion of favorable outcomes for both treatment groups.

- Wald: **0.2570-0.6763**
- Corrected Wald: **0.2409-0.5924**
- Agresti-Caffo: **0.2456-0.5416**
- Miettinen-Nurminen **0.2460-0.5427**

Treatment	Outcome	
	Favorable	Unfavorable
Placebo	16	48
Test	40	20

SAS Code

```
1 data respire;
2 input treat $ outcome $ count;
3 datalines;
4 test f 40
5 test u 20
6 placebo f 16
7 placebo u 48
8 ;
9
10 proc freq data=respire order=data;
11 weight count;
12 tables treat*outcome/riskdiff(cl=(mn ac)) alpha=0.05;
13 tables treat*outcome/riskdiff(correct) alpha=0.05;
14 run;
```

5.10 Exercise

One of the most famous and largest clinical trials ever performed was in 1954. The data obtained from this RCT is shown in the contingency table. Calculate the 90% confidence interval for the difference in proportion of polio-free children between the vaccine and placebo group.

- Corrected Wald
- Miettinen-Nurminen
- Newcombe

Salk vaccine clinical trial results; data source is Francis et al. (1955, p. 25).

	Polio	Polio free	Total
Vaccine	57	200,688	200,745
Placebo	142	201,087	201,229
Total	199	401,775	401,974

5.11 Exercise

One of the most famous and largest clinical trials ever performed was in 1954. The data obtained from this RCT is shown in the contingency table. Calculate the 90% confidence interval for the difference in proportion of polio-free children between the vaccine and placebo group.

- Corrected Wald (**0.0003,0.0005**)
- Miettinen-Nurminen (**0.0003,0.0005**)
- Newcombe (**0.0003,0.0005**)

Salk vaccine clinical trial results; data source is Francis et al. (1955, p. 25).

	Polio	Polio free	Total
Vaccine	57	200,688	200,745
Placebo	142	201,087	201,229
Total	199	401,775	401,974

SAS Code

```

1 data vaxx;
2 length treat $7.;
3 input treat $ outcome $ count;
4 datalines;
5 vax n 200688
6 vax p 57
7 placebo n 201087
8 placebo p 142
9 ;
10
11 proc freq data=vaxx order=data;
12 weight count;
13 tables treat*outcome/riskdiff(correct cl=(mn newcombe))
14 alpha=0.10;
15 run;

```

6 Odds Ratio and Relative Risk

6.1 Odds Ratio

- Odds of success: the ratio of probability of success and failure of an outcome.

$$Odds = p/(1 - p)$$

- Odds ratio between two groups: the ratio of the odds of success for Group 1 and Group 2.

$$OR = \frac{p_1/(1 - p_1)}{p_2/(1 - p_2)} = (n_{11}n_{22}/n_{21}n_{12})$$

- Commonly used in retrospective studies

6.2 Relative Risk

- The ratio of the risk developing a particular condition for individuals with risk factors present and absent

$$RR = p_1/p_2 = \frac{(n_{11}/n_{1+})}{(n_{21}/n_{2+})}$$

- Commonly used in prospective studies

6.3 Odds Ratio & Relative Risk

- OR and RR are approximately equal when the condition is rare.
 - <10% of the time
- For cross-sectional data, the relative risk is known as the prevalence ratio because the disease and risk factor are assessed at the same time.
- When OR, $RR > 1$, Group 1 is **more likely** than Group 2 to record success.
- When OR, $RR < 1$, Group 1 is **less likely** to record success.
- When OR, $RR = 1$, there is **no association** between group and success because the two groups are equally likely to record success.

6.4 Odds Ratio & Relative Risk: SAS

- OR and RR are calculated using the measures option in the TABLES statement in PROC FREQ.
- For small sample sizes, the exact confidence limits can be calculated for the odds ratios.

```
1 proc freq;  
2 weight count;  
3 tables a*b/measures;  
4 exact or rr;  
5 run;
```

6.5 Example

The contingency table shows data from a study on treatments for severe infections. Calculate the asymptotic and exact confidence limits for the odds ratio of favorable outcomes for test and control groups.

Treatment	Outcome	
	Favorable	Unfavorable
Test	10	2
Control	2	4

6.6 Example

The contingency table shows data from a study on treatments for severe infections. Calculate the asymptotic and exact confidence limits for the odds ratio of favorable outcomes for test and control groups.

Treatment	Outcome	
	Favorable	Unfavorable
Test	10	2
Control	2	4

SAS Code

```
data severe;
input treat $ outcome $ count;
datalines;
Test f 10
Test u 2
Control f 2
Control u 4
;
proc freq order=data data=severe;
weight count;
tables treat*outcome / chisq nocol;
exact or;
run;
```

Method	Value and CI
Asymptotic	10 (1.0256, 97.5005)
Exact	10 (0.6896,166.3562)

The two methods lead to different conclusions.

6.7 Exercise

Two groups having the same symptoms of respiratory illness were sampled. One group was treated with an experimental treatment and the other group was treated with a placebo. The participants were asked whether there was an overall treatment in overall respiration.

- Calculate the odds ratio of respiratory improvement between test and placebo groups.
- What is the relative risk of having respiratory improvement if we consider the treatment as a risk factor?
- Calculate the asymptotic 95% CI for each answer.

	Outcome	
Treatment	Yes	No
Test	29	16
Placebo	14	31

6.8 Exercise

Two groups having the same symptoms of respiratory illness were sampled. One group was treated with an experimental treatment and the other group was treated with a placebo. The participants were asked whether there was an overall treatment in overall respiration.

- Calculate the odds ratio of respiratory improvement between test and placebo groups. **4.01 (1.67,9.66)**
- What is the relative risk of having respiratory improvement if we consider the treatment as a risk factor? **2.07 (1.27,3.37)**
- Calculate the asymptotic 95% CI for each answer.

	Outcome	
Treatment	Yes	No
Test	29	16
Placebo	14	31

6.8.1 SAS Code

```
data resp;
input Treatment $ Outcome $ count;
datalines;
Test Yes 29
Placebo Yes 14
Test No 16
Placebo No 31
;
proc freq data=resp order=data;
weight count;
tables treatment*outcome/measures;
run;
```

6.8.2 R Code

Odds ratio

```
library(epitools)
```

Attaching package: 'epitools'

The following object is masked from 'package:vcdExtra':

expand.table

The following object is masked from 'package:vcd':

oddsratio

```
dat_cc |> oddsratio(rev=c("both"), method="wald", conf.level=0.95) -> or
or$measure
```

```
      odds ratio with 95% C.I.
Treatment estimate    lower    upper
Placebo 1.000000      NA      NA
Test    4.013393 1.668039 9.656441
```

Relative Risk

```
dat_cc |> riskratio(rev=c("both"),method="wald")->rr
rr$measure
```

```
      risk ratio with 95% C.I.
Treatment estimate  lower    upper
Placebo  1.000000      NA      NA
Test     2.071429  1.274194  3.367474
```

7 Sensitivity and specificity

7.1 Sensitivity

True proportion of positive results that a test elicits when performed on subjects known to have the disease.

$$Sens = TP / (TP + FN) = n_{11} / n_{1+}$$

Sensitivity is the conditional probability of a positive test given that the subject has the disease ($P(PT|HD)$).

7.2 Specificity

True proportion of negative results that a test elicits when performed on subjects known to be disease free.

$$Sens = TN / (TN + FP) = n_{22} / n_{2+}$$

Specificity is the conditional probability of a negative test when the subject does not have the disease ($P(NT|ND)$).

7.3 Sensitivity and Specificity: SAS

The FREQ procedure can also calculate sensitivity and specificity.

- `riskdiff` approximates these metrics using the risk estimates.
 - Sensitivity: Row 1 risk estimate for Column 1
 - Specificity: Row 2 risk estimate for Column 2

7.4 Example

The following table shows the results of a study investigating a new screening device for a skin disease. Calculate the sensitivity and specificity of the screening device.

Status	Test	
	Positive	Negative
Disease Present	52	8
Disease Absent	20	100

7.5 Example

The following table shows the results of a study investigating a new screening device for a skin disease. Calculate the sensitivity and specificity of the screening device.

- Sensitivity: 0.87 (0.75,0.94)
- Specificity: 0.83 (0.75,0.90)

Status	Test	
	Positive	Negative
Disease Present	52	8
Disease Absent	20	100

SAS Code

```
data screening;
input disease $ outcome $ count @@;
datalines;
present + 52 present - 8
absent + 20 absent - 100
;
proc freq data=screening order=data;
weight count;
tables disease*outcome / riskdiff;
run;
```

7.6 Exercise

A pharmaceutical company developed a screening test that can be used to detect infection of the Lassa (virus). The newly developed test was said to be cheaper than the tests currently available to the public, but its performance was yet to be evaluated. The screening test was applied to 115 diagnosed cases (w/ Lassa virus) and 120 control participants (w/o Lassa virus) of the Lassa fever. The resulting contingency table is shown in the table.

- What is the sensitivity of the test?
- What is the specificity of the test?

Status	Test	
	Positive	Negative
Disease Present	97	18
Disease Absent	45	75

7.7 Exercise

A pharmaceutical company developed a screening test that can be used to detect infection of the Lassa (virus). The newly developed test was said to be cheaper than the tests currently available to the public, but its performance was yet to be evaluated. The screening test was applied to 115 diagnosed cases (w/ Lassa virus) and 120 control participants (w/o Lassa virus) of the Lassa fever. The resulting contingency table is shown in the table.

- What is the sensitivity of the test? **0.8435 (0.76, 0.90)**
- What is the specificity of the test? **0.6250 (0.53, 0.71)**

Status	Test	
	Positive	Negative
Disease Present	97	18
Disease Absent	45	75

7.7.1 SAS Code

```
data screening;
input disease $ outcome $ count @@;
datalines;
present + 97 present - 18
absent + 45 absent - 75
;
```

```
proc freq data=screening order=data;
weight count;
tables disease*outcome / riskdiff;
run;
```

8 McNemar's Test

8.1 McNemar's Test

- When the 2x2 table provides information from matched pairs, the two responses are related to each other. Hence, independence cannot be assumed.
- Null hypothesis: The marginal probabilities of each outcome are the same.
- The test statistic for the McNemar's test is Q_M given by

$$Q_M = (n_{12} - n_{21})^2 / (n_{12} + n_{21})$$

8.2 McNemar's Test: SAS

- The `agree` option in the `tables` statement can be used to perform the McNemar's test.
- Exact p-values can be calculated using the `exact` statement.

8.2.1 Sample SAS Code

```
proc freq;
weight count;
tables a*b/agree;
exact mcnem;
run;
```

8.3 Example

The table displays data collected by political science researchers who polled husbands and wives on whether they approved of one of their U.S. senators. The cell counts represent the number of pairs of husbands and wives who fit the configurations indicated by the row and column levels. Use the McNemar's test to determine whether there is no difference in husband and wife approvals on the state senator.

	Wife	
Husband	Yes	No
Yes	20	5
No	10	10

8.4 Example

The table displays data collected by political science researchers who polled husbands and wives on whether they approved of one of their U.S. senators. The cell counts represent the number of pairs of husbands and wives who fit the configurations indicated by the row and column levels. Use the McNemar's test to determine whether there is no difference in husband and wife approvals on the state senator.

- Exact p-value = 0.3018, which implies that there is no evidence of a difference between the senator approval ratings of the husbands and wives.

	Wife	
Husband	Yes	No
Yes	20	5
No	10	10

8.4.1 SAS Code

```

1  data approval;
2  input hus_resp $ wif_resp $ count;
3  datalines;
4  yes yes 20
5  yes no 5
6  no yes 10
7  no no 10
8  ;
9
10 proc freq order=data;
11 weight count;
12 tables hus_resp*wif_resp / agree;
13 exact mcnem;
14 run;

```

8.5 Exercise

The tables show the data from a study that compared the diagnostic accuracy of magnetic resonance imaging (MRI) and ultrasound in patients who had been established as having localized prostate cancer by a gold standard test.

- Use the McNemar's test to determine whether the two methods have a different probabilities in diagnosing localized prostate cancer. (*Hint: Small cell counts*)

MRI	Ultrasound	
	Localized	Advanced
Localized	4	6
Advanced	3	3

8.6 Exercise

The tables show the data from a study that compared the diagnostic accuracy of magnetic resonance imaging (MRI) and ultrasound in patients who had been established as having localized prostate cancer by a gold standard test.

MRI	Ultrasound	
	Localized	Advanced
Localized	4	6
Advanced	3	3

8.6.1 Answer

- Use the McNemar's test to determine whether the two methods have a different probabilities in diagnosing localized prostate cancer. (*Hint: Small cell counts*)
 - Use exact p-value
 - $p = 0.5078$. No evidence of any difference in probabilities in correctly diagnosing localized prostate cancer.

8.6.2 SAS Code

```
data imaging;
input mri $ ult $ count;
datalines;
loc loc 4
loc adv 6
```

```

adv loc 3
adv adv 3
;

proc freq order=data;
weight count;
tables mri*ult / agree;
exact mcnem;
run;

```

9 Incidence Densities

9.1 Incidence Densities

- Counts of subjects who responded within a duration of exposure to an event.
- Poisson distribution: counts
- Examples:
 - Bacteria colony count
 - Disease incidence
 - Accidents
 - Vaccine failure
- Common measure of duration of exposure/ time-at-risk: person-time

9.2 Poisson Distribution

Recall that the Poisson distribution is described by the following density function:

$$P(x) = e^{-\lambda} \lambda^x / x!$$

where λ is the expected number of events at a given unit of time.

- is also referred to as incidence density.
- For t units of time, the expected number of events scales to λt .
- Main assumption: The time to event is a random variable following an exponential distribution.

9.3 Incidence Density Ratios

- The table shows the number of infections (events) and person years for participants of an RCT.
 - n_v and n_p are Poisson random variables with respective incidence densities equal to λ_v and λ_p
- If λ_v is the incidence density for the vaccine group, then the expected number of events is $\lambda_v N_v$.
- Similarly, the expected number of events in the placebo group is $\lambda_p N_p$
- We are interested in testing the null hypothesis $\lambda_v/\lambda_p = 1$.
 - λ_v/λ_p is referred to as the **Incidence Density Ratio (IDR)**

Table 2.14 Vaccine Study

Treatment	Events	Person Years
Vaccine	n_v	N_v
Placebo	n_p	N_p

9.4 Incidence Density Ratios

The IDR can be estimated using conditional binomial distributions assuming n_v and n_p are independent Poisson processes.

$$P(n_v | n_v + n_p) \sim \text{Bin} \left(N = n_v + n_p, p = \frac{\lambda_v N_v}{\lambda_v N_v + \lambda_p N_p} \right)$$

9.5 Incidence Density Ratios: Confidence Interval

The binomial proportion can be estimated using the proportion of events in the treatment group:

$$p = \frac{n_v}{n_v + n_p} = \frac{\text{IDR}(N_v/N_p)}{1 + (\text{IDR})(N_v/N_p)}$$

Using algebra (Problem Set 1), *it is easy to show that*:

$$IDR = \frac{p}{1-p} (N_p/N_v)$$

9.6 Incidence Density Ratios: Confidence Interval

You can calculate the confidence interval estimates for p using the binomial distribution assumption (p_L, p_U) and apply the scaling factor to each confidence limit.

$$95\%CI : \left(\frac{p_L}{1-p_L} (N_p/N_v), \frac{p_U}{1-p_U} (N_p/N_v) \right)$$

9.7 Incidence Density Ratios: SAS

- The `binomial` option in the `tables` statement produces the estimate and exact confidence limits for the binomial proportion p , i.e. p_L and p_U .
- We use the estimate and exact confidence limits of p in calculating the IDR.
- The `IML` procedure, which is the procedure used to perform mathematical equations, is used to calculate the IDR confidence limits.
 - Later on, we will use the `GENMOD` procedure to calculate exact confidence intervals for the IDR.

9.8 Example

The table shows the results of an international study of the effect of a rotavirus vaccine in nearly 70,000 infants on health care events. The events counted were hospitalizations or emergency room visits.

- Estimate the IDR of hospitalization/emergency room visits between the two groups.
- Calculate the 95% CI.

Table 2.14 Vaccine Study

Treatment	Events	Person Years
Vaccine	n_v	N_v
Placebo	n_p	N_p

9.9 Example

The table shows the results of an international study of the effect of a rotavirus vaccine in nearly 70,000 infants on health care events. The events counted were hospitalizations or emergency room visits.

- Estimate the IDR of hospitalization/emergency room visits between the two groups.
IDR = 0.05
- Calculate the 95% CI. **(0.01, 0.15)**

Table 2.13 Vaccine Study Results for U.S.

Treatment	Events	Person Years
Vaccine	3	7500
Placebo	58	7250

9.9.1 SAS Code

```
1 data vaccine2;
2 input Outcome $ Count @@;
3 datalines;
4 vaccine 3 placebo 58
5 ;
6
7 ods select BinomialCLs; *this statement filters the output.;
8
9
10 proc freq order=data data=vaccine2;                                ①
11 weight count;
12 tables Outcome / binomial (exact); *calculates the exact confidence limits for the binomial p
13 ods output BinomialCLs=BinomialCLs; *outputs a dataset that includes the binomial CLs.;
14 run;
15
16 proc iml ;
17 Use BinomialCLs var{LowerCL UpperCL};
18 read all into CL;
19 print CL;
20 q = { 3, 7500, 58 , 7250 }; *vector of numbers from the table;
21 C= q[2]/q[4]; *q[i] represents the ith number in the q vector defined above. C is the person
22 P= q[1]/ (q[1] + q[3]); *estimate of the binomial distribution;
```

IDR	IDR_lower	IDR_higher
0.05	0.01002014	0.1535468

```

23 R= P/ ((1-P) *C) ; * IDR estimate;
24 CI= CL[1]/((1-CL[1])*C) || CL[2]/((1-CL[2])*C) ; *IDR CL;
25 print r CI;
26 quit;

```

① order=data is important so as to keep the proportions consistent.

9.9.2 R Code

```

events <- c(3,58)
person_years <- c(7500, 7250)

mult <- person_years[2]/person_years[1] #multiplier
DescTools::BinomCI(events[1],sum(events),method = "clopper-pearson") -> p_est

data.frame(IDR = p_est[1]/(1-p_est[1])*mult,
           IDR_lower = p_est[2]/(1-p_est[2])*mult,
           IDR_higher = p_est[3]/(1-p_est[3])*mult) |> gt::gt()

```

9.10 Exercise

The table lists the number of accidents for day and night shift workers at a certain factory in Southeast Asia. The company wants you to investigate if there is any evidence of night shift workers having a higher accident density, i.e., higher incidence density of accidents, compared to the day shift.

Shift	Metric	
	Accidents	Person-Hours
Night	25	1150
Day	20	1240

9.11 Exercise

The table lists the number of accidents for day and night shift workers at a certain factory in Southeast Asia. The company wants you to investigate if there is any evidence of night shift workers having a higher accident density, i.e., higher incidence density of accidents, compared to the day shift.

- IDR: 1.37
- 95% CI: (0.72,2.56)

	Metric	
Shift	Accidents	Person-Hours
Night	25	1150
Day	20	1240

9.11.1 SAS Code

```
data accident;
/*data only has the events. Person-Hours are used later.*/
input Outcome $ Count @@;
datalines;
night 25 day 20
;

ods select BinomialCLs; *this statement filters the output.;

proc freq order=data data=accident;
weight count;
tables Outcome / binomial (exact) alpha=0.05; *calculates the exact confidence limits for the
ods output BinomialCLs=BinomialCLs; *outputs a dataset that includes the binomial CLs.;
run;

proc iml ;
Use BinomialCLs var{LowerCL UpperCL};
read all into CL;
print CL;
q = { 25 1150 20 1240 }; *vector of numbers from the table;
C= q[2]/q[4]; *q[i] represents the ith number in the q vector defined above. C is the person
P= q[1]/ (q[1] + q[3]); *estimate of the binomial distribution;
R= P/ ((1-P) *C) ; * IDR estimate;
CI= CL[1]/((1-CL[1])*C) || CL[2]/((1-CL[2])*C) ; *IDR CL;
```

IDR	IDR_lower	IDR_higher
1.347826	0.7187612	2.559057

```
print r CI;
quit;
```

① “Success” is an event happening during the night shift.

9.11.2 R Code

```
events <- c(25,20)
person_years <- c(1150,1240)

mult <- person_years[2]/person_years[1] #multiplier
DescTools::BinomCI(events[1],sum(events),method = "clopper-pearson") -> p_est

data.frame(IDR = p_est[1]/(1-p_est[1])*mult,
            IDR_lower = p_est[2]/(1-p_est[2])*mult,
            IDR_higher = p_est[3]/(1-p_est[3])*mult) |> gt::gt()
```

10 Power Analysis

10.1 Power Analysis

- Power is defined as the probability of rejecting the null hypothesis when it is false.
 - Acceptable range: 80% and above.
- Also known as (a priori) Sample Size Calculations
 - How many samples do I need to detect a set difference between the treatment and placebo group at a significance level of 0.05?

10.2 Power Analysis: SAS

- The POWER procedure enables us to calculate how many samples are needed to detect a set effect size in examining the difference between two proportions from two different populations.
- Specifically, the TWOSAMPLEFREQ option
 - RCT: treatment vs. control
- We need to set an effect size (theory-based), significance level, and power level.
 - Other options can be found [here](#)

10.3 Example

Suppose prior literature states that the incidence rate of a chronic disease is 0.40. Suppose we want to perform a prospective study to test the association between the disease incidence and family history, i.e. whether other family members were diagnosed with the chronic disease. We want to detect a proportion difference of 0.10 between groups with power of 80% (0.80) at a significance level of 0.05.

10.4 Example

Suppose prior literature states that the incidence rate of a chronic disease is 0.40. Suppose we want to perform a prospective study to test the association between the disease incidence and family history, i.e. whether other family members were diagnosed with the chronic disease. We want to detect a proportion difference of 0.10 between groups with power of 80% (0.80) at a significance level of 0.05.

- We need 388 units per group (family +, family -). This brings us to 776 total units.

10.4.1 SAS Code

```
proc power ;  
twosamplefreq  
  refproportion=0.4  
  alpha=0.05  
  power=0.80  
  proportiondiff=0.1  
  npergroup=.;  
  
run;
```

10.4.2 RCode

```
power.prop.test(p1=0.4,p2=0.5,power=0.8)
```

Two-sample comparison of proportions power calculation

```
      n = 387.3385
      p1 = 0.4
      p2 = 0.5
sig.level = 0.05
  power = 0.8
alternative = two.sided
```

NOTE: n is number in *each* group