

EAB 770 - Chapter 5

Miguel Fudolig

1 $s \times r$ tables

1.1 Outline

- Association
 - Test for General Association
 - Mean Score Test
 - Correlation Test
 - Exact Tests
- Measures of Association
 - Ordinal
 - Nominal
- Observer Agreement (inter-rater reliability)
- Test for Ordered Differences

1.2 $s \times r$ tables

We now evolve the contingency table from a $2 \times 2 \rightarrow s \times r$, where s is the number of rows and r is the number of columns.

1.3 Nominal Categorical variables

The response variables can now have $r \geq 2$ levels and be strictly nominal.

Table 5.2 $s \times r$ Contingency Table

Group	Response Variable Categories				Total
	1	2	...	r	
1	n_{11}	n_{12}	...	n_{1r}	n_{1+}
2	n_{21}	n_{22}	...	n_{2r}	n_{2+}
$\vdots$	$\vdots$	$\vdots$		$\vdots$	$\vdots$
s	n_{s1}	n_{s2}	...	n_{sr}	n_{s+}
Total	n_{+1}	n_{+2}	...	n_{+r}	n

1.4 General Association Statistic

- The groups and the response variable categories can be nominally scaled.
 - H_0 : there is no general association between the groups and response variables.
 - H_A : there is a general association between the two variables

1.5 General Association Statistic

The randomization statistic based on Mantel-Haenszel statistics can be calculated from the Pearson chi-square statistic, Q_P .

$$Q = \left(\frac{n-1}{n} \right) Q_P$$


$$Q_P = \sum_{i=1}^s \sum_{j=1}^r \frac{(n_{ij} - m_{ij})^2}{m_{ij}}$$

! Important

When n is large, $Q \approx Q_P$. Like the Pearson chi-square, Q follows a chi-square distribution with $(s-1)(r-1)$ degrees of freedom.

1.6 Mosaic Plot

- A mosaic plot visualizes the proportion in each cell of the contingency table.
 - Horizontal space is divided by sizes corresponding to the relative count of x-axis variable levels
 - Vertical space is divided up according to relative sizes of the y-axis variable levels.



1.7 General Association: SAS

- In SAS, the Pearson chi-square statistic Q_P is the value referred to as ‘Chi-square’
- Note that Q is **NOT** the Mantel-Haenszel statistic in the `chisq` output in SAS.
 - Q is the “General Association” cell in the `cmh` output.
- We can use the `plots` option in the `tables` statement to create a mosaic plot.

```
proc freq;  
weight count;  
tables a*b/cmh plots=mosaicplot;  
run;
```

①

- ① `cmh` calculates the general association Mantel-Haenszel statistic. `plots=mosaicplot` tells SAS to create a mosaic plot with a and b .

1.8 Example

The table contains data from a study concerning the distribution of party affiliation in a city suburb. The interest was whether there was an association between registered political party and neighborhood.

- What is the Pearson chi-square value?
- What is the randomization statistic?
- Is there sufficient evidence of an association between political party and neighborhood?

Table 5.1 Distribution of Parties in Neighborhoods

Party	Neighborhood			
	Bayside	Highland	Longview	Sheffield
Democrat	221	160	360	140
Independent	200	291	160	311
Republican	208	106	316	97

1.9 Example

The table contains data from a study concerning the distribution of party affiliation in a city suburb. The interest was whether there was an association between registered political party and neighborhood.

- What is the Pearson chi-square value?
- What is the randomization statistic?
- Is there sufficient evidence of an association between political party and neighborhood?

1.9.1 SAS Code

```
data neighbor;  
length party $ 11 neighborhood $ 10;  
input party $ neighborhood $ count @@;  
datalines;  
democrat longview 360 democrat bayside 221  
democrat sheffield 140 democrat highland 160  
republican longview 316 republican bayside 208
```

```

republican sheffield 97 republican highland 106
independent longview 160 independent bayside 200
independent sheffield 311 independent highland 291
;

proc freq data=neighbor;
weight count;
tables party*neighborhood /
plots=mosaicplot chisq cmh nocol nopct;
run;

```

①

① Included both `chisq` and `cmh` to show discrepancy.

1.9.2 Answer

$Q_P = 273.92$, $Q = 273.81$. The corresponding p-value $p < 0.001$ suggests there is an association between party affiliation and neighborhood.

- Note that the Mantel-Haenszel statistic reported in the `chisq` output corresponds to the correlation statistic which doesn't make sense in this case (discussed later).

1.10 Exercise

Secondary school teachers were randomly sampled from four school districts in Nevada and asked what marker color they used in grading students' work.

- What is the Pearson chi-square statistic?
- What is the randomization statistic?
- Generate a mosaic plot for this example. Can you see a trend between the colors and/or the counties?
- At a significance level of 0.05, do we have sufficient evidence of a general association between the two variables?

Data set:

```

data color;
input county $ color $ count @@;
datalines;
cl blu 22 cl red 57 cl green 12 cl blk 15
ca blu 12 ca red 30 ca green 6 ca blk 15
do blu 17 do red 34 do green 17 do blk 10
wa blu 5 wa red 56 wa green 7 wa blk 9

```

```
;
run;
```

1.11 Exercise

Secondary school teachers were randomly sampled from four school districts in Nevada and asked what marker color they used in grading students' work.

- What is the Pearson chi-square statistic?
- What is the randomization statistic?
- Generate a mosaic plot for this example. Can you see a trend between the colors and/or the counties?
- At a significance level of 0.05, do we have sufficient evidence of a general association between the two variables?

1.11.1 SAS Code

```
data color;
input county $ color $ count @@;
datalines;
cl blu 22 cl red 57 cl green 12 cl blk 15
ca blu 12 ca red 30 ca green 6 ca blk 15
do blu 17 do red 34 do green 17 do blk 10
wa blu 5 wa red 56 wa green 7 wa blk 9
;
run;

proc freq data=color;
weight count;
tables county*color/
plots=mosaicplot chisq cmh nocol nopct measures cl;
run;
```

1.11.2 Answer

$Q_P = 24.36$, $Q = 24.28$, $p = 0.004$, which means there is strong evidence of an association between school district and marker color used for grading.

Attaching package: 'ggmosaic'


The following objects are masked from 'package:vcd':

mosaic, spine

Warning: The `scale\_name` argument of `continuous\_scale()` is deprecated as of ggplot2 3.5.0.

Warning: The `trans` argument of `continuous\_scale()` is deprecated as of ggplot2 3.5.0.
i Please use the `transform` argument instead.

Warning: `unite\_()` was deprecated in tidyr 1.2.0.
i Please use `unite()` instead.
i The deprecated feature was likely used in the ggmosaic package.
Please report the issue at <<https://github.com/haleyjeppson/ggmosaic>>.



1.12 Mean Score Tests

- Used when the column variable is ordinal.
- Row variable could be nominal
 - H_0 : there is no difference in mean scores across the groups
 - H_a : at least one group has a different mean score compared to the others
- Scores are calculated for each row
- Recall: 2xr tables

1.13 Mean Score Tests

- The mean score statistic Q_S can be used to test the difference between the mean scores.
 - Consistent with the ANOVA approach
- Recall 2xr table: Sample size requirement
- Set a cutoff column, the totals to the left and right of this column should be greater than 5.
 - Row totals should be **greater than 20**
- Included in the `cmh` table as “Row Mean Scores Differ”

1.14 Example

The data shown comes from a study on headache pain relief. A new treatment was compared with the standard treatment and a placebo. Researchers measured the number of hours of substantial relief from headache pain

- Is the chi-square approximation valid for this case?
- What kind of scoring method is most appropriate for this data set?
- Make the appropriate conclusion based on the result of the mean score tests using a significance level $\alpha = 0.05$.

Table 5.3 Pain Study Data

Treatment	Hours of Relief				
	0	1	2	3	4
Placebo	6	9	6	3	1
Standard	1	4	6	6	8
Test	2	5	6	8	6

1.15 Example

The data shown comes from a study on headache pain relief. A new treatment was compared with the standard treatment and a placebo. Researchers measured the number of hours of substantial relief from headache pain

- Is the chi-square approximation valid for this case?
- What kind of scoring method is most appropriate for this data set?
- Make the appropriate conclusion based on the result of the mean score tests using a significance level $\alpha = 0.05$.

1.15.1 SAS Code

```
data pain;
input treatment $ hours count @@;
datalines;
placebo 0 6 placebo 1 9 placebo 2 6 placebo 3 3 placebo 4 1
standard 0 1 standard 1 4 standard 2 6 standard 3 6 standard 4 8
test 0 2 test 1 5 test 2 6 test 3 8 test 4 6
;

proc freq data=pain order=data;
weight count;
tables treatment*hours/ cmh nocol nopct;
run;
```

1.15.2 Answer

- Use column 3 (Hours = 2) as cutoff. The use of the chi-square approximation is justified!

- The integer scoring method is most appropriate.
- $Q_S = 13.7$, $p=0.001$. There is sufficient evidence to conclude that at least one treatment group is different compared to the others.

1.16 Exercise

The table shows data from a study that investigates intent to breastfeed of women who are interested in having children. The women who participated in the study were asked if they intended to breastfeed your child up to 6 months after birth. The researchers want to know if there is an effect of the order of pregnancy on their intent to breastfeed.

```
data preg;
input pregnancy $ outcome $ score count @@ ;
datalines;
npnc sd 1 8 npnc d 2 22 npnc n 3 24 npnc a 4 27 npnc sa 5 15
npc sd 1 15 npc d 2 23 npc n 3 15 npc a 4 19 npc sa 5 9
fp sd 1 18 fp d 2 8 fp n 3 22 fp a 4 23 fp sa 5 11
pwc sd 1 35 pwc d 2 13 pwc n 3 10 pwc a 4 9 pwc sa 5 6
;
```

Variables

- npnc: not pregnant, no children
- npc: not pregnant with children
- fp: first pregnancy
- pwc: pregnant with children
- sd: strongly disagree
- d: disagree
- n: neutral
- a: agree
- sa: strongly agree

1.17 Exercise

The table shows data from a study that investigates intent to breastfeed of women who are interested in having children. The women who participated in the study were asked if they intended to breastfeed your child up to 6 months after birth. The researchers want to know if there is an effect of the order of pregnancy on their intent to breastfeed.

1.17.1 SAS Code

```
proc freq data=preg order=data;
weight count;
tables pregnancy*outcome / cmh nocol nopct;
run;
```

```
proc means data=preg order=data;
freq count;
class pregnancy;
var score;
run;
```

①

- ① The `means` procedure is shown to display the mean scores and how they indeed differ for each pregnancy order.

1.17.2 Answer

$Q_S = 27.2$, $p < 0.0001$. There is sufficient evidence of an effect of pregnancy order on intent to breastfeed.

1.18 Correlation Tests

- When both and row variables are ordinally scaled, we can look at correlations between them.
- Recall $s \times 2$ tables: Correlation statistic Q_{CS}
 - $s \times r$ tables: Assign scores to both rows and columns
 - Q_{CS} can be calculated using matrix algebra.

$$Q_{CS} = (n - 1)r_{ac}^2$$

where r_{ac} is the Pearson correlation coefficient.

1.19 Correlation Tests

Note

Q_{CS} follows a chi-square distribution with 1 degree of freedom

- Sample size requirement is the same as the $s \times 2$ tables.
- Uses 1 degree of freedom regardless of the size of the contingency table

1.20 Correlation Tests: SAS

Q_{CS} is the Mantel-Haenszel chi-square statistic displayed by the `chisq` output.

- `chisq` and `cmh` will both yield the correlation statistic.

1.21 Example

A water treatment company is studying water additives and investigating how they affect clothes washing. The treatments studied were no treatment (plain water), the standard treatment, and a double dose of the standard treatment, called super. Washability was measured as low, medium, and high. The company claims that the dose of water treatment affects the washability of water. Do we have sufficient evidence to support this claim?

Table 5.4 Washability Data

Treatment	Washability			Total
	Low	Medium	High	
Water	27	14	5	46
Standard	10	17	26	53
Super	5	12	50	67

1.22 Example

A water treatment company is studying water additives and investigating how they affect clothes washing. The treatments studied were no treatment (plain water), the standard treatment, and a double dose of the standard treatment, called super. Washability was measured as low, medium, and high. The company claims that the dose of water treatment affects the washability of water. Do we have sufficient evidence to support this claim?

1.22.1 SAS Code

```
data wash;
input treatment $ washability $ count @@;
datalines;
water low 27 water medium 14 water high 5
standard low 10 standard medium 17 standard high 26
super low 5 super medium 12 super high 50
;

proc freq order=data data=wash;
weight count;
tables treatment*washability / chisq cmh;
tables treatment*washability / scores=modridit cmh
noprint nocol nopct;
run;
```

1.22.2 Answer

$Q_{CS} = 50.6$, $p < 0.0001$. There is sufficient evidence to claim that the dose of water treatment is associated with the washability of the water.

If modridit scores are used, $Q_{CS} = 49.5$.

1.23 Exercise

A random sample of 400 college students are classified according to class status and drinking habits. Test the hypothesis that class status and drinking habits are not correlated to each other.

```
data drink;
input habit $ status $ count @@;
datalines;
non fres 55 non soph 34 non jun 27 non sen 17
mod fres 32 mod soph 29 mod jun 36 mod sen 39
hea fres 29 hea soph 41 hea jun 33 hea sen 28
;
```

Variables

non: nondrinkers; mod: moderate; hea: heavy; fres: freshman; soph: sophomore; jun: junior; sen:senior

1.24 Exercise

A random sample of 400 college students are classified according to class status and drinking habits. Test the hypothesis that class status and drinking habits are not correlated to each other.

1.24.1 SAS Code

```
data drink;
input habit $ status $ count @@;
datalines;
non f 55 non soph 34 non j 27 non sen 17
mod f 32 mod soph 29 mod j 36 mod sen 39
hea f 29 hea soph 41 hea j 33 hea sen 28
;

proc freq data=drink order=data;
weight count;
tables habit*status/cmh chisq;
run;
```

1.24.2 Answer

Table Scores: $Q_{CS} = 9.2$, $p=0.0024$. There is sufficient evidence to support the claim that class status and drinking habits are correlated to each other.

1.25 Exact Tests

- Recall: Fisher's Exact Test
 - `sxr`: Generalized to a multivariate hypergeometric distribution
- For general association, we can get exact p-values for the Pearson chi-square, and likelihood ratio chi-square.
- Unlike Fisher's Exact Test for 2x2 tables, there are no one-sided alternative hypothesis.

- Exact p-value for randomization statistic is the same as the Pearson chi-square p-value.

1.26 Exact Tests

- Recall: Exact testing of $s \times 2$ tables
 - The correlation statistic is suitable when sample size requirements are not met, but we can't use the chi-square distribution.
 - Exact p-value is the sum of p-values associated with contingency tables where the test statistic is larger than the observed table.

1.27 Exact Tests: SAS

- The `exact` statement can be used to calculate exact p-values and perform exact tests.
 - `Fisher` – Fisher's exact test
 - `PCHI` – Pearson chi-square statistic
 - `LRCHI` – likelihood ratio chi-square statistic
 - `MHCHI` – Mantel-Haenszel statistic (Correlation tests)

1.28 Exact Tests: SAS

Warning

Exact tests are computationally intensive. More cells lead to more calculations, which makes exact computation not feasible.

- You can set the `MAXTIME` option in the `exact` statement to tell SAS to stop computations beyond a certain number of seconds.

1.29 Example

A marketing research firm organized a focus group to consider issues of new car marketing. Members of the group included people who had purchased a car from a local dealer in the last month. Researchers were interested in whether there was an association between the type of car bought and how group members found out about the car in the media. Cars were classified as sedans, sporty, and utility. The types of media included television, magazines, newspapers, and radio.

Table 5.5 Car Marketing Data

Type of Car	Advertising Source				Total
	TV	Magazine	Newspaper	Radio	
Sedan	4	0	0	2	6
Sporty	0	3	3	4	10
Utility	5	5	2	2	14

1.30 Example

A marketing research firm organized a focus group to consider issues of new car marketing. Members of the group included people who had purchased a car from a local dealer in the last month. Researchers were interested in whether there was an association between the type of car bought and how group members found out about the car in the media. Cars were classified as sedans, sporty, and utility. The types of media included television, magazines, newspapers, and radio.

1.30.1 SAS Code

```
data market;
length AdSource $ 9. ;
input car $ AdSource $ count @@;
datalines;
sporty paper 3 sporty radio 4 sporty tv 0 sporty magazine 3
sedan paper 0 sedan radio 2 sedan tv 4 sedan magazine 0
utility paper 2 utility radio 2 utility tv 5 utility magazine 5
;
proc freq;
weight count;
table car*AdSource / norow nocol nopct;
exact fisher pchi lrchi;
run;
```

1.30.2 Answer

We need to use Fisher's exact test because of the small sample size. The p-value was calculated to be 0.04, which is sufficient evidence to conclude that there is an association between car type and media.

1.31 Exercise

A university professor wanted to test the general association between the most used social media app and the type of high school attended by freshmen. The professor asked 23 students in his class about their social media habits and respective high schools.

	Tiktok	Instagram	Snapchat	Other
Priv., in-state	3	2	3	0
Pub., in-state	3	1	0	0
Priv., out-of-state	4	1	0	1
Pub., out-of-state	1	0	0	1

1.32 Exercise

A university professor wanted to test the general association between the most used social media app and the type of high school attended by freshmen. The professor asked 23 students in his class about their social media habits and respective high schools.

1.32.1 SAS Code

```
data sm;
input school $ social $ count @@;
datalines;
Pr_IS t 3 Pr_IS i 2 Pr_IS s 3 Pr_IS o 0
Pu_IS t 3 Pu_IS i 1 Pu_IS s 3 Pu_IS o 0
Pr_OS t 4 Pr_OS i 1 Pr_OS s 0 Pr_OS o 1
Pu_OS t 1 Pu_OS i 0 Pu_OS s 0 Pu_OS o 1
;

proc freq data=sm order=data;
weight count;
tables school*social;
exact fisher pchi lrchi;
run;
```

1.32.2 Answer

The Fisher's exact test yielded $p=0.45$, which means there is insufficient evidence to claim that social media habits are associated with the type of high school attended by the students.

2 Measures of Association

2.1 Measures of Association

- We only have information on whether the row and column variables are associated, but not by how much.
- Depends on the scale of the measurements
 - Ordinal vs. Nominal
 - Are levels equally spaced?

2.2 Ordinal variables

When both row and column variables are ordinal, we can use correlation coefficients.

Note

When scores are equally spaced, the **Pearson** correlation coefficient is appropriate. For data without an obvious scale, the **Spearman** correlation coefficient is more appropriate.

2.3 Ordinal Variables

- Association can also be measured by classifying all possible pairs of subjects as concordant or discordant pairs.
 - Concordant: Subject ranks higher on the row variable also ranks higher on the column variable
 - Discordant: Subject that ranks higher on the row variable also ranks lower on the column variable
 - Depends on relative ordering of the levels of variables

Note

Measures of association that use these pairs are Gamma γ , Kendall's τ_B , Stuart's τ_C , Somers' Delta.

- All measures range from -1 to 1, inclusive.

2.4 Measures of Association

These measures differ in adjusting for ties and sample size.

- Somers' D depends on the location of the explanatory variable.
- Stuart's τ_C includes an adjustment for table size to a correction for ties. Recommended when two ordinal variables under consideration have different numbers of values.
- Kendall's τ_B is preferred when the contingency table is square ($m \times m$)
- Gamma does not account for ties, so it is only recommended if there are no ties.

2.5 Measures of Association

When the sample size is large enough, 95% confidence intervals can be calculated for these measures using a normal approximation.

$$measure \pm 1.96 * ASE$$

where ASE is the asymptotic standard error reported in the `measures` output from the `freq` procedure. - The confidence interval can be calculated using the `ci` option in the `freq` procedure.

2.6 Exercise

Consider the water treatment example. Calculate the following measures of association with the corresponding 95% CI.

- Pearson correlation
- Spearman Correlation
- Gamma
- Kendall's τ_b
- Stuart's τ_c
- Somers' Delta

Table 5.4 Washability Data

Treatment	Washability			Total
	Low	Medium	High	
Water	27	14	5	46
Standard	10	17	26	53
Super	5	12	50	67

2.7 Exercise

Consider the water treatment example. Calculate the following measures of association with the corresponding 95% CI.

2.7.1 SAS Code

```
data wash;
input treatment $ washability $ count @@;
datalines;
water low 27 water medium 14 water high 5
standard low 10 standard medium 17 standard high 26
super low 5 super medium 12 super high 50
;
proc freq order=data data=wash;
weight count;
tables treatment*washability / measures cl;
run;
```

2.7.2 Answer

- Pearson Correlation: 0.55 (0.44,0.67)
- Spearman Correlation: 0.55 (0.43, 0.66)
- Gamma: 0.6974 (0.57-0.82)
- τ_b : 0.50 (0.39,0.61)
- τ_C : 0.48 (0.37,0.59)
- Somers' D C|R: 0.49 (0.38,0.59)

2.8 Exercise

A random sample of 400 college students are classified according to class status and drinking habits. Calculate the following measures of association and the 95% CI.

- Pearson correlation
- Spearman Correlation
- Gamma
- Kendall's τ_b
- Stuart's τ_c
- Somers' Delta

```
data drink;
input habit $ status $ count @@;
datalines;
non fres 55 non soph 34 non jun 27 non sen 17
mod fres 32 mod soph 29 mod jun 36 mod sen 39
hea fres 29 hea soph 41 hea jun 33 hea sen 28
;
```

①

① non: nondrinkers; mod: moderate; hea: heavy; fres: freshman; soph: sophomore; jun: junior; sen:senior

2.9 Exercise

A random sample of 400 college students are classified according to class status and drinking habits. Calculate the following measures of association and the 95% CI.

2.9.1 SAS Code

```
data drink;
input habit $ status $ count @@;
datalines;
non f 55 non soph 34 non j 27 non sen 17
mod f 32 mod soph 29 mod j 36 mod sen 39
hea f 29 hea soph 41 hea j 33 hea sen 28
;

proc freq data=drink order=data;
weight count;
tables habit*status/measures cl;
run;
```

2.9.2 Answer

- Pearson Correlation: 0.15 (0.06,0.25)
- Spearman Correlation: 0.16 (0.06, 0.25)
- Gamma: 0.18 (0.07,0.30)
- τ_b : 0.13 (0.05,0.21)
- τ_C : 0.14 (0.05,0.22)
- Somers' D C|R: 0.14 (0.05,0.22)

2.10 Nominal Variables

It is slightly tricky to define association between two nominal variables. There are two measures of association that could measure the strength of association between nominal variables.

- Lambda λ
- Uncertainty coefficients

Both of these measures are included in the `measures` output.

Warning

Note that SAS will calculate all measures despite the nature of the variables. It is your job as an analyst to only report the relevant measures.

Do not use ordinal measures for nominal variables. Do not use nominal measures for ordinal variables

2.11 Lambda λ

Lambda measures the probable improvement in predicting the column variable given knowledge of the row variable.

- Asymmetric: Can be C|R or R|C depending on where the explanatory variable is located
- Symmetric: weighted average of the two asymmetric λ values.

2.12 Uncertainty coefficient

The uncertainty coefficient, similar to R^2 , measures the proportion of uncertainty in column variable as explained by the row variable.

- Asymmetric: Can be C|R or R|C depending on where the explanatory variable is located
- Symmetric: weighted average of the two asymmetric uncertainty coefficients.

2.13 Exercise

The table contains data from a study concerning the distribution of party affiliation in a city suburb. The interest was whether there was an association between registered political party and neighborhood.

- What are the appropriate measures of association to report?
- Provide the estimate and the 95% CI of these measures.

Table 5.1 Distribution of Parties in Neighborhoods

Party	Neighborhood			
	Bayside	Highland	Longview	Sheffield
Democrat	221	160	360	140
Independent	200	291	160	311
Republican	208	106	316	97

2.14 Exercise

The table contains data from a study concerning the distribution of party affiliation in a city suburb. The interest was whether there was an association between registered political party and neighborhood.

- What are the appropriate measures of association to report?
- Provide the estimate and the 95% CI of these measures.

2.14.1 SAS Code

```
data neighbor;
length party $ 11 neighborhood $ 10;
input party $ neighborhood $ count @@;
datalines;
democrat longview 360 democrat bayside 221
democrat sheffield 140 democrat highland 160
republican longview 316 republican bayside 208
republican sheffield 97 republican highland 106
independent longview 160 independent bayside 200
independent sheffield 311 independent highland 291
;

proc freq data=neighbor;
weight count;
tables party*neighborhood /
measures cl;
run;
```

2.14.2 Answer

You can choose to report the asymmetric λ , symmetric λ , asymmetric uncertainty coefficient, or the symmetric uncertainty coefficient.

The asymmetric uncertainty coefficient R|C is estimated to be 0.05 (95% CI: 0.04,0.06). The estimated proportion of uncertainty in party affiliation explained by the neighborhood is estimated to be 0.05.

3 Observer Agreement

3.1 Observer Agreement

Errors are inevitable in making measurements!

- Biggest error: Observer error

Note

How do we account for observer error, especially for multiple observer measurements?

3.2 Observer Agreement: κ

The kappa statistic κ measures the extent of agreement between two observers.

- $\kappa=0$: any agreement is made by chance
- $\kappa=1$: perfect agreement
- $\kappa \geq 0.8$: excellent agreement
- $\kappa \geq 0.4$: moderate agreement

Note

This is similar to the intraclass correlation coefficient in ANOVA models.

3.3 Observer Agreement: κ

κ is calculated from the following equation:

$$\kappa = (\Pi_O - \Pi_E) / (1 - \Pi_E)$$

where Π_O is the probability that observers agreed based on the data, Π_E is the expected probability of agreement if the two observer ratings are assumed to be independent.

3.4 Observer Agreement

- When there is order in the categories, we might need to account how different the disagreements are.
- We can use weighted κ measurements such as:
 - Cicchetti-Allison weights puts more weight on cells closest to the diagonal.
 - Fleiss-Cohen kappa weights use a squared distance between categories, which penalizes highly discordant pairs.

3.5 Bowker's Test

- Tests whether the contingency table is symmetric about the diagonals.
 - Works for square tables
 - Same as McNemar's test when the table is 2×2

3.6 Observer Agreement: SAS

- The AGREE option in the tables statement generates the kappa statistics and Bowker's symmetry test.
 - AGREE (WT=FC) specifies to use the *Fleiss-Cohen* criterion for the Weighted Kappa statistic.
- The agreement plot is also shown.
 - Comparison of shaded portions of the rectangles to the entire area of the rectangle provides information about the agreement between the two raters.
- Exact testing can also be done by specifying KAPPA in the EXACT statement.
 - Small cell counts with multiple 0 cell counts.

3.7 Example

- The data comes from a study concerning the diagnostic classification of multiple sclerosis patients. Patients from Winnipeg and New Orleans were classified into one of four diagnostic classes by both a Winnipeg neurologist and a New Orleans neurologist.
 - Calculate the estimate and 95% confidence interval for the simple and weighted Kappa statistic.
 - What can you say about the agreement between the New Orleans and Winnipeg neurologists?

Table 5.8 Ratings of Neurologists

New Orleans Neurologist	Winnipeg Neurologist			
	1	2	3	4
1	38	5	0	1
2	33	11	3	0
3	10	14	5	6
4	3	7	3	10

3.7.1 SAS Code

```
data classify;
input no_rater w_rater count @@;
datalines;
1 1 38 1 2 5 1 3 0 1 4 1
2 1 33 2 2 11 2 3 3 2 4 0
3 1 10 3 2 14 3 3 5 3 4 6
4 1 3 4 2 7 4 3 3 4 4 10
;
```

3.8 Example

- The data comes from a study concerning the diagnostic classification of multiple sclerosis patients. Patients from Winnipeg and New Orleans were classified into one of four diagnostic classes by both a Winnipeg neurologist and a New Orleans neurologist.
 - Calculate the estimate and 95% confidence interval for the simple and weighted Kappa statistic.

- What can you say about the agreement between the New Orleans and Winnipeg neurologists?

3.8.1 SAS Code

```
data classify;
input no_rater w_rater count @@;
datalines;
1 1 38 1 2 5 1 3 0 1 4 1
2 1 33 2 2 11 2 3 3 2 4 0
3 1 10 3 2 14 3 3 5 3 4 6
4 1 3 4 2 7 4 3 3 4 4 10
;

proc freq;
weight count;
tables no_rater*w_rater / agree norow nocol nopct;
run;
```

3.8.2 Answer

Simple κ : 0.21 (0.11,0.31) Weighted κ : 0.38 (0.28,0.48)

Since the weighted κ is higher than the simple κ , we can infer that there were more disagreements closer to the diagonal. There is a slight to moderate agreement between the two neurologists.

3.9 Exercise

Two radiologists rated 85 patients with respect to liver lesions. The ratings were designated on an ordinal scale as:

0 = 'Normal'; 1 = 'Benign'; 2 = 'Suspected'; 3 = 'Cancer'

- Calculate the simple and weighted kappa statistics and the corresponding 95% confidence intervals.
- Compare the asymptotic and exact p-values in testing whether there is an agreement between the two radiologists.

```

data radiology;
input rater1 rater2 count;
cards;
0 0 21
0 1 12
0 2 0
0 3 0
1 0 4
1 1 17
1 2 1
1 3 0
2 0 3
2 1 9
2 2 15
2 3 2
3 0 0
3 1 0
3 2 0
3 3 1
;
run;

```

3.10 Exercise

Two radiologists rated 85 patients with respect to liver lesions. The ratings were designated on an ordinal scale as:

0 = 'Normal'; 1 = 'Benign' ; 2 = 'Suspected'; 3 = 'Cancer'

- Calculate the simple and weighted kappa statistics and the corresponding 95% confidence intervals.
- Compare the asymptotic and exact p-values in testing whether there is an agreement between the two radiologists.

3.10.1 SAS Code

```

data radiology;
input rater1 rater2 count;
cards;
0 0 21

```

```

0 1 12
0 2 0
0 3 0
1 0 4
1 1 17
1 2 1
1 3 0
2 0 3
2 1 9
2 2 15
2 3 2
3 0 0
3 1 0
3 2 0
3 3 1
;
run;

proc freq data=radiology;
tables rater1*rater2/norow nocol nopct agree;
weight count;
test kappa;
exact kappa;
run;

```

3.10.2 Answer

Simple κ : 0.47 (0.33,0.62) Weighted κ : 0.57 (0.44,0.70)

The exact p-value is $p < 0.001$, which shows there is sufficient evidence of an agreement between the radiologists. Based on the simple and weighted kappa values, there is moderate agreement between the two radiologists.

4 Ordered Differences

4.1 Test for Ordered Differences

- What if the columns are ordinal but the rows are nominal?
 - Recall: rows could pertain to different groups.
 - Recall: Mean score tests account for location shifts in the mean response

- What if we want to know if the mean scores are monotonously increasing or decreasing across levels of the row variable?
 - Example: Showing increasing prevalence of disease in specific environments.

4.2 Test for Ordered Differences

The Jonckheere-Terpstra test is a nonparametric test that is based on sums of Mann-Whitney test statistics

- Rank-based test

This test detects whether there are differences in the trend of group effects.

- Are the scores strictly increasing or decreasing?

i Note

The null hypothesis H_0 assumes that the distribution of ordered responses is the same for all groups. For one-sided tests, this tests detect whether $d_1 \leq d_2 \leq \dots d_s$ where d_i corresponds to the i^{th} group effect.

i Note

The Jonckheere-Terpstra test can be performed by adding the **JT** option to the **tables** statement. An exact implementation of the Jonckheere-Terpstra test can also be performed by adding the **JT** option to the **exact** statement.

4.3 Example

Investigators conducted a randomized clinical trial in four hospitals, where patients were assigned to one of four surgical procedures for the treatment of severe duodenal ulcers. The treatments include:

- v + d: vagotomy and drainage
- v + a: vagotomy and antrectomy (removal of 25% of gastric tissue)
- v + h: vagotomy and hemigastrectomy (removal of 50% of gastric tissue)
- gre: gastric resection (removal of 75% of gastric tissue)

The response measured was the severity (none, slight, moderate) of symptoms, which is expected to increase directly with the proportion of gastric tissue removed. Investigators wanted to determine whether the distribution of the severity of duodenal ulcers is increasing as the amount of gastric tissue removed increased. Ignore hospital effects.

4.3.1 SAS Code

```
data operate;
input hospital trt $ severity $ wt @@;
datalines;
1 v+d none 23 1 v+d slight 7 1 v+d moderate 2
1 v+a none 23 1 v+a slight 10 1 v+a moderate 5
1 v+h none 20 1 v+h slight 13 1 v+h moderate 5
1 gre none 24 1 gre slight 10 1 gre moderate 6
2 v+d none 18 2 v+d slight 6 2 v+d moderate 1
2 v+a none 18 2 v+a slight 6 2 v+a moderate 2
2 v+h none 13 2 v+h slight 13 2 v+h moderate 2
2 gre none 9 2 gre slight 15 2 gre moderate 2
3 v+d none 8 3 v+d slight 6 3 v+d moderate 3
3 v+a none 12 3 v+a slight 4 3 v+a moderate 4
3 v+h none 11 3 v+h slight 6 3 v+h moderate 2
3 gre none 7 3 gre slight 7 3 gre moderate 4
4 v+d none 12 4 v+d slight 9 4 v+d moderate 1
4 v+a none 15 4 v+a slight 3 4 v+a moderate 2
4 v+h none 14 4 v+h slight 8 4 v+h moderate 3
4 gre none 13 4 gre slight 6 4 gre moderate 4
;
```

4.4 Example

Investigators conducted a randomized clinical trial in four hospitals, where patients were assigned to one of four surgical procedures for the treatment of severe duodenal ulcers. The treatments include:

- v + d: vagotomy and drainage
- v + a: vagotomy and antrectomy (removal of 25% of gastric tissue)
- v + h: vagotomy and hemigastrectomy (removal of 50% of gastric tissue)
- gre: gastric resection (removal of 75% of gastric tissue)

The response measured was the severity (none, slight, moderate) of symptoms, which is expected to increase directly with the proportion of gastric tissue removed. Investigators wanted to determine whether the distribution of the severity of duodenal ulcers is different across all treatments, ignoring the hospital effects.

4.4.1 SAS Code

```
data operate;
input hospital trt $ severity $ wt @@;
datalines;
1 v+d none 23 1 v+d slight 7 1 v+d moderate 2
1 v+a none 23 1 v+a slight 10 1 v+a moderate 5
1 v+h none 20 1 v+h slight 13 1 v+h moderate 5
1 gre none 24 1 gre slight 10 1 gre moderate 6
2 v+d none 18 2 v+d slight 6 2 v+d moderate 1
2 v+a none 18 2 v+a slight 6 2 v+a moderate 2
2 v+h none 13 2 v+h slight 13 2 v+h moderate 2
2 gre none 9 2 gre slight 15 2 gre moderate 2
3 v+d none 8 3 v+d slight 6 3 v+d moderate 3
3 v+a none 12 3 v+a slight 4 3 v+a moderate 4
3 v+h none 11 3 v+h slight 6 3 v+h moderate 2
3 gre none 7 3 gre slight 7 3 gre moderate 4
4 v+d none 12 4 v+d slight 9 4 v+d moderate 1
4 v+a none 15 4 v+a slight 3 4 v+a moderate 2
4 v+h none 14 4 v+h slight 8 4 v+h moderate 3
4 gre none 13 4 gre slight 6 4 gre moderate 4
;

proc freq data=operate order=data;
weight wt;
tables trt*severity / jt;
run;
```

4.4.2 Answer

The JT statistic was estimated to be 35,697, corresponding to a one-sided p-value $p = 0.005$, which is sufficient evidence to claim that the severity of the duodenal ulcer increases with the amount of gastric tissue removed.