

EAB 770 - Chapter 6

Miguel Fudolig

1 Sets of $s \times r$ tables

1.1 Outline

- General Mantel-Haenszel Method
- Mantel-Haenszel Applications
- Repeated Measures

1.2 General Mantel-Haenszel Method

- Recall: 2x2, sx2, 2xr, sxr tables all use Mantel-Haenszel Statistics
 - Able to account for ordinality unlike Pearson Chi-Square tests
 - Less stringent sample size requirements
- MH statistics in sets of contingency tables removes the confounding influence of the stratification variable.

1.3 General Mantel-Haenszel Method

Note

H_0 : The response variable is distributed at random with respect to the groups defined by the row variables

Depending on the nature of the row and column variables, the Mantel-Haenszel statistic has different forms:

- Q_{GMH} : General Mantel-Haenszel statistic
- Q_{SMH} : Mean Score Statistic
- Q_{CSMH} : Correlation Statistic

1.4 General Mantel-Haenszel Method

Table 1: Extended Mantel-Haenszel Statistics

MH Statis- tic	Alternative Hypothesis	SAS Output Label	Degrees of Freedom	Scale Requirements	Nonparametric Equivalents
Q_{GMH}	General Association	General Association	$(s - 1)(r - 1)$	None	
Q_{SMH}	Mean score location shifts	Row Means Scores Differ	$(s - 1)$	Column Variable Ordinal	Kruskal-Wallis
Q_{CSMH}	Linear Association	Nonzero correlation	1	Row and Column Variable Ordinal	Spearman Correlation

1.5 Example

Investigators conducted a randomized clinical trial in four hospitals, where patients were assigned to one of four surgical procedures for the treatment of severe duodenal ulcers. The treatments include:

- v + d: vagotomy and drainage
- v + a: vagotomy and antrectomy (removal of 25% of gastric tissue)
- v + h: vagotomy and hemigastrectomy (removal of 50% of gastric tissue)
- gre: gastric resection (removal of 75% of gastric tissue)

The response measured was the severity (none, slight, moderate) of symptoms, which is expected to increase directly with the proportion of gastric tissue removed. Investigators wanted to determine whether the distribution of the severity of duodenal ulcers is different across all treatments, **controlling** for hospital effects.

1.5.1 SAS Data set

```
data operate;
input hospital trt $ severity $ wt @@;
datalines;
1 v+d none 23 1 v+d slight 7 1 v+d moderate 2
1 v+a none 23 1 v+a slight 10 1 v+a moderate 5
1 v+h none 20 1 v+h slight 13 1 v+h moderate 5
1 gre none 24 1 gre slight 10 1 gre moderate 6
2 v+d none 18 2 v+d slight 6 2 v+d moderate 1
```

```

2 v+a none 18 2 v+a slight 6 2 v+a moderate 2
2 v+h none 13 2 v+h slight 13 2 v+h moderate 2
2 gre none 9 2 gre slight 15 2 gre moderate 2
3 v+d none 8 3 v+d slight 6 3 v+d moderate 3
3 v+a none 12 3 v+a slight 4 3 v+a moderate 4
3 v+h none 11 3 v+h slight 6 3 v+h moderate 2
3 gre none 7 3 gre slight 7 3 gre moderate 4
4 v+d none 12 4 v+d slight 9 4 v+d moderate 1
4 v+a none 15 4 v+a slight 3 4 v+a moderate 2
4 v+h none 14 4 v+h slight 8 4 v+h moderate 3
4 gre none 13 4 gre slight 6 4 gre moderate 4
;

```

1.6 Example

The response measured was the severity (none, slight, moderate) of symptoms, which is expected to increase directly with the proportion of gastric tissue removed. Investigators wanted to determine whether the distribution of the severity of duodenal ulcers is different across all treatments, **controlling** for hospital effects.

1.6.1 SAS Code

```

proc freq data=operate order=data;
weight wt;
tables hospital*trt*severity / cmh;
tables hospital*trt*severity / cmh scores=modridit;
run;

```

①

②

① Table Scores

② Modridit scores

1.6.2 Answer

Both variables are ordinal, hence we can test for correlation using Q_{CSMH} . Using modridit scores, we get $Q_{CSMH} = 6.93$, $p=0.0085$. Hence, we have sufficient evidence that the proportion of gastric tissue removed is associated with symptom severity in patients.

1.7 Exercise

The following data were collected in a study of shoulder harness usage in observations for a sample of North Carolina cars. The researchers wanted to test for an association between car size and usage after adjusting for geographical area and location.

```
data shoulder;
input area $ location $ size $ usage $ count @@;
datalines;
coast urban large no 174 coast urban large yes 69
coast urban medium no 134 coast urban medium yes 56
coast urban small no 150 coast urban small yes 54
coast rural large no 52 coast rural large yes 14
coast rural medium no 31 coast rural medium yes 14
coast rural small no 25 coast rural small yes 17
piedmont urban large no 127 piedmont urban large yes 62
piedmont urban medium no 94 piedmont urban medium yes 63
piedmont urban small no 112 piedmont urban small yes 93
piedmont rural large no 35 piedmont rural large yes 29
piedmont rural medium no 32 piedmont rural medium yes 30
piedmont rural small no 46 piedmont rural small yes 34
mountain urban large no 111 mountain urban large yes 26
mountain urban medium no 120 mountain urban medium yes 47
mountain urban small no 145 mountain urban small yes 68
mountain rural large no 62 mountain rural large yes 31
mountain rural medium no 44 mountain rural medium yes 32
mountain rural small no 85 mountain rural small yes 43
;
```

1.8 Exercise

The following data were collected in a study of shoulder harness usage in observations for a sample of North Carolina cars. The researchers wanted to test for an association between car size and usage after adjusting for geographical area and location.

1.8.1 SAS Code

```
proc freq data=shoulder order=data;
weight count;
tables area*location*size*usage / cmh scores=modridit;
```

```
;
run;
```

1.8.2 Answer

Both variables are ordinal. We can use the extended correlation statistic Q_{CSMH} . Using modridit scores, $Q_{CSMH} = 6.64$, $p = 0.01$. There is sufficient evidence to conclude that car size and shoulder harness usage are correlated.

1.9 Repeated Measures

- Commonly used in longitudinal studies
 - Cohort studies
- Dental experiments: multiple teeth from each subject
- Mantel-Haenszel statistics can be calculated to determine whether there is an association between a repeated measurement factor (usually time) and a response variable adjusting for an effect of subject.
 - Controlling for subject adjusts for the variability between subjects, which focuses the analysis to within-subject.

1.10 Repeated Measures

- Consider an experiment with n subjects.
 - The treatment is applied to each subject.
 - The response is measured at t time points.
 - Cohort studies/ longitudinal RCT
- In general, the repeated measures can come from multiple subjects. Responses can come from:
 - Same location
 - Same family/litter
 - Matched pairs

1.11 Repeated Measures

Consider a repeated measures design with t time points and a response variable with C levels.

Table 6.5 One-Sample Repeated Measures Data

Subject	Time Point				
	1	...	j	...	t
1	y_{11}	...	y_{1j}	...	y_{1t}
$\vdots$	$\vdots$	$\ddots$	$\vdots$	$\ddots$	$\vdots$
i	y_{i1}	...	y_{ij}	...	y_{it}
$\vdots$	$\vdots$	$\ddots$	$\vdots$	$\ddots$	$\vdots$
n	y_{n1}	...	y_{nj}	...	y_{nt}

i Note

Each time point will have a measured response value. For a sample size n , we will end up with n contingency tables with shape $t \times C$

Table 6.6 Contingency Table Layout for Subject i

Time Point	Response Category			Total
	1	...	c	
1	n_{i11}	...	n_{i1c}	n_{i1+}
$\vdots$	$\vdots$	$\ddots$	$\vdots$	$\vdots$
t	n_{it1}	...	n_{itc}	n_{it+}
Total	n_{i+1}	...	n_{i+c}	n_i

i Note

$n_{ijk} = 1$ if response k is measured at time j . Otherwise, $n_{ijk} = 0$

1.12 Repeated Measures: Hypotheses

- Mantel-Haenszel statistics can be used to test the hypothesis of no association between time and response.
 - Assumption: marginal totals per subject in response and time is fixed.
 - H_0 : The response variable is distributed at random with respect to the time points
 - The null hypothesis is referred to as the hypothesis of interchangeability or marginal homogeneity.
- As in previous chapters, Q_{CSMH} , Q_{SMH} , and Q_{GMH} pertain to different alternatives to no association.

1.13 Repeated Measures: Caution

Warning

Odds ratio estimates and CI limits assumes independence between contingency table entries, which is broken in repeated measures data. Breslow-Day assumptions are also not satisfied.

The Breslow-Day statistic as well as the MH and logit estimates of the odds ratio should be taken with a grain of salt.

1.14 Example

A running shoe company produces a new model of running shoe that includes a harder material for the insert that corrects for overpronation. However, the company is concerned that the material will induce heel tenderness as a result of some loss of cushioning on the strike of each step. It conducted a study on 87 runners who used the new shoe for a month. Researchers asked the participants whether they experienced occasional heel tenderness before and after they used the new shoe. Is there a difference in proportion of people reporting more heel tenderness with the new running shoes compared with the older running shoes?

1.14.1 SAS Data

```
data pump;
input subject time $ response $ @@;
datalines;
1 before no 1 after no 2 before no 2 after no
3 before no 3 after no 4 before no 4 after no
```

5 before no 5 after no 6 before no 6 after no
7 before no 7 after no 8 before no 8 after no
9 before no 9 after no 10 before no 10 after no
11 before no 11 after no 12 before no 12 after no
13 before no 13 after no 14 before no 14 after no
15 before no 15 after no 16 before no 16 after no
17 before no 17 after no 18 before no 18 after no
19 before no 19 after no 20 before no 20 after no
21 before no 21 after no 22 before no 22 after no
23 before no 23 after no 24 before no 24 after no
25 before no 25 after no 26 before no 26 after no
27 before no 27 after no 28 before no 28 after no
29 before no 29 after no 30 before no 30 after no
31 before no 31 after no 32 before no 32 after no
33 before no 33 after no 34 before no 34 after no
35 before no 35 after no 36 before no 36 after no
37 before no 37 after no 38 before no 38 after no
39 before no 39 after no 40 before no 40 after no
41 before no 41 after no 42 before no 42 after no
43 before no 43 after no 44 before no 44 after no
45 before no 45 after no 46 before no 46 after no
47 before no 47 after no 48 before no 48 after no
49 before no 49 after yes 50 before no 50 after yes
51 before no 51 after yes 52 before no 52 after yes
53 before no 53 after yes 54 before no 54 after yes
55 before no 55 after yes 56 before no 56 after yes
57 before no 57 after yes 58 before no 58 after yes
59 before no 59 after yes 60 before no 60 after yes
61 before no 61 after yes 62 before no 62 after yes
63 before no 63 after yes 64 before yes 64 after no
65 before yes 65 after no 66 before yes 66 after no
67 before yes 67 after no 68 before yes 68 after no
69 before yes 69 after yes 70 before yes 70 after yes
71 before yes 71 after yes 72 before yes 72 after yes
73 before yes 73 after yes 74 before yes 74 after yes
75 before yes 75 after yes 76 before yes 76 after yes
77 before yes 77 after yes 78 before yes 78 after yes
79 before yes 79 after yes 80 before yes 80 after yes
81 before yes 81 after yes 82 before yes 82 after yes
83 before yes 83 after yes 84 before yes 84 after yes
85 before yes 85 after yes 86 before yes 86 after yes
87 before yes 87 after yes

;

1.15 Example

Is there a difference in proportion of people reporting more heel tenderness with the new running shoes compared with the older running shoes?

1.15.1 SAS Code

```
data pump;
input subject time $ response $ @@;
datalines;
1 before no 1 after no 2 before no 2 after no
3 before no 3 after no 4 before no 4 after no
5 before no 5 after no 6 before no 6 after no
7 before no 7 after no 8 before no 8 after no
9 before no 9 after no 10 before no 10 after no
11 before no 11 after no 12 before no 12 after no
13 before no 13 after no 14 before no 14 after no
15 before no 15 after no 16 before no 16 after no
17 before no 17 after no 18 before no 18 after no
19 before no 19 after no 20 before no 20 after no
21 before no 21 after no 22 before no 22 after no
23 before no 23 after no 24 before no 24 after no
25 before no 25 after no 26 before no 26 after no
27 before no 27 after no 28 before no 28 after no
29 before no 29 after no 30 before no 30 after no
31 before no 31 after no 32 before no 32 after no
33 before no 33 after no 34 before no 34 after no
35 before no 35 after no 36 before no 36 after no
37 before no 37 after no 38 before no 38 after no
39 before no 39 after no 40 before no 40 after no
41 before no 41 after no 42 before no 42 after no
43 before no 43 after no 44 before no 44 after no
45 before no 45 after no 46 before no 46 after no
47 before no 47 after no 48 before no 48 after no
49 before no 49 after yes 50 before no 50 after yes
51 before no 51 after yes 52 before no 52 after yes
53 before no 53 after yes 54 before no 54 after yes
```

```

55 before no 55 after yes 56 before no 56 after yes
57 before no 57 after yes 58 before no 58 after yes
59 before no 59 after yes 60 before no 60 after yes
61 before no 61 after yes 62 before no 62 after yes
63 before no 63 after yes 64 before yes 64 after no
65 before yes 65 after no 66 before yes 66 after no
67 before yes 67 after no 68 before yes 68 after no
69 before yes 69 after yes 70 before yes 70 after yes
71 before yes 71 after yes 72 before yes 72 after yes
73 before yes 73 after yes 74 before yes 74 after yes
75 before yes 75 after yes 76 before yes 76 after yes
77 before yes 77 after yes 78 before yes 78 after yes
79 before yes 79 after yes 80 before yes 80 after yes
81 before yes 81 after yes 82 before yes 82 after yes
83 before yes 83 after yes 84 before yes 84 after yes
85 before yes 85 after yes 86 before yes 86 after yes
87 before yes 87 after yes
;

```

```
proc freq data=pump order=data;
```

```
tables subject*time*response/ noprint cmh; *noprint does not print the 87 contingency tables
run;
```

①

① There's no weight statement because each line is an observational unit.

1.15.2 Answer

- We can construct 2x2 tables for each runner and control for each subject. The resulting CMH tables will have the same results because of the shape of the contingency tables.
- $Q_{CSMH}=5$, $p=0.03$, which means there is strong evidence that there is a difference in heel tenderness before and after using the new shoe.
- All extended Mantel-Haenszel statistic will yield the same result because the resulting contingency table for each subject is a 2x2 table.

1.15.3 McNemar's test

```

data shoes;
input before $ after $ count;
datalines;
yes yes 19
yes no 5

```

```

no yes 15
no no 48
;

proc freq data=shoes order=data;
weight count;
tables before*after/agree;
run;

```

The McNemar's test yields the same result as Q_{CSMH} .

1.16 Exercise

The DRUG2 dataset shows the data from a study in which 46 patients were each treated with three drugs (A, B, and C). The response to each drug was recorded as favorable (F) or unfavorable (U). Use the appropriate test to determine whether the three drugs have equal probabilities for favorable response.

- Which Mantel-Haenszel statistic is the most appropriate for this example?
- At a significance level of 0.01, is there sufficient evidence against the hypothesis that the three drugs have equal probabilities?

1.16.1 SAS Data

```

1 data drug;
2 input druga $ drugb $ drugc $ count;
3 datalines;
4 F F F 6
5 F F U 16
6 F U F 2
7 F U U 4
8 U F F 2
9 U F U 4
10 U U F 6
11 U U U 6
12 ;
13 run;
14 data drug2; set drug;
15 keep patient drug response;
16 retain patient 0;

```

```

17 do i=1 to count;
18 patient=patient+1;
19 drug='A'; response=druga; output;
20 drug='B'; response=drugb; output;
21 drug='C'; response=drugc; output;
22 end;
23 run;

```

1.17 Exercise

- Which Mantel-Haenszel statistic is the most appropriate for this example?
- At a significance level of 0.01, is there sufficient evidence against the hypothesis that the three drugs have equal probabilities?

1.17.1 SAS Code

```

proc freq data=drug2 order=data;
tables patient*drug*response/cmh noprint;
run;

```

1.17.2 Answer

The most appropriate MH statistic to examine is “Row Means Differ”.

The statistic $Q_{SMH} = 8.47$ yields a p-value $p = 0.0145$. At a significance level of 0.01, there is marginally sufficient evidence that at least one drug has a different probability of a favorable response.

1.18 Ordinal Response with Missing Data

Experiments using a repeated measures design tend to include plenty of missing data.

- Missing data decreases the power of the study.
 - Including missing data in the analysis still gives a higher power than excluding incomplete cases in the analysis.
- Techniques to deal with missing data include:
 - Multiple imputation
 - Omission
 - Adjusting ANOVA scores