

EAB 770 - Chapter 9

Miguel Fudolig

1 Logistic Regression - Polytomous Response

1.1 Outline

- Ordinal Response
 - Proportional Odds Model
 - Multiple Qualitative Explanatory Variables
 - Partial Proportional Odds Model
- Nominal Response
 - Generalized Logits Model
 - Exact Methods: Generalized Logits Model

2 Ordinal Response Variable

2.1 Polytomous Variables?

Recall Chapter 8: dichotomous response variables can be analyzed using logistic regression.

Note

What if the response variable has more than two levels?

2.2 Polytomous variables

- Polytomous variables are usually described by the [multinomial distribution](#)

💡 Tip

Unlike the binomial distribution where we only need one probability of success, more than one probability of success is needed to be defined in multinomial distributions.

2.3 Cumulative probabilities

Consider π_{hij} as the probability of success of the h^{th} strata and i^{th} explanatory variables and the j^{th} level of the response variable. We can define θ_{hij} such that:

$$\theta_{hiJ} = \sum_{j=1}^J \pi_{hij}$$

i Note

If $J = 1$, then $\theta_{hi1} = \pi_{hi1}$. If $J = 2$, then $\theta_{hi2} = \pi_{hi1} + \pi_{hi2}$. θ_{hiJ} is the *cumulative probability* of the response levels from $j = 1$ to $j = J$.

i Note

If there are $J + 1$ levels, then $\pi_{J+1} = 1 - \sum_{j=1}^J \pi_{hij}$.

2.4 Cumulative logits

Cumulative logits can be defined based on the cumulative probabilities θ_{hij} . For a specific level k out of K levels,

$$\text{logit}(\theta_{hik}) = \log\left(\frac{\theta_{hik}}{1 - \theta_{hik}}\right) = \log\left(\frac{\sum_{j=1}^k \pi_{hij}}{1 - \sum_{j=1}^k \pi_{hij}}\right) = \log\left(\frac{\sum_{j=1}^k \pi_{hij}}{\sum_{j=k+1}^K \pi_{hij}}\right)$$

2.5 Cumulative logits

Suppose we define a linear predictor η_{hij} such that

$$\eta_{hij} = \text{logit}(\theta_{hij}) = \alpha_j + \mathbf{x}_{hi}' \boldsymbol{\beta}_j$$

where $x'_{hi}\beta_k$ represents the matrix product of the explanatory/strata variables x_{hi} and the coefficients for each response level β_k .

2.6 Proportional Odds Assumption

i Proportional Odds

The proportional odds assumption applies the following condition for a response variable with K total variable levels:

$$\mathbf{1} = \mathbf{2} = \mathbf{3} = \dots = \mathbf{K} =$$

Key assumption: the effect of the explanatory variables should not be disproportionate across all levels.

2.7 Proportional Odds

A consequence of the proportional odds assumption is that the difference between the levels of the explanatory variables are the same across different response levels.

i Example

Consider comparing the m^{th} and n^{th} level of the explanatory variable in the h^{th} strata for the first and second level of the response variable.

$$\eta_{hm1} - \eta_{hn1} = (\mathbf{x}'_{hm} - \mathbf{x}'_{hn}) \mathbf{1} = (\mathbf{x}'_{hm} - \mathbf{x}'_{hn})$$

$$\eta_{hm2} - \eta_{hn2} = (\mathbf{x}'_{hm} - \mathbf{x}'_{hn}) \mathbf{2} = (\mathbf{x}'_{hm} - \mathbf{x}'_{hn})$$

These are the same across all levels of the response variable. The difference between η correspond to the log odds, hence the odds ratios between explanatory variables are the same across all response levels.

2.8 Proportional Odds

Another consequence of the proportional odds assumption is that the difference between the response levels does not depend on the strata or explanatory variables.

i Example

Consider comparing the first and second level of the response variable.

$$\eta_{hm1} - \eta_{hm2} = \alpha_1 - \alpha_2 + (\mathbf{x}'_{hm} - \mathbf{x}'_{hm}\beta) = \alpha_1 - \alpha_2$$

2.9 Solving for Individual Probabilities

Individual probabilities can be calculated by taking the differences between cumulative logits.

$$\pi_1 = \theta_1; \pi_2 = \theta_2 - \theta_1; \pi_k = \theta_k - \theta_{k-1}$$

2.10 Proportional Odds Model in SAS

- We can use the LOGISTIC, GENMOD, and GLIMMIX procedures to fit the proportional odds model.
- Dealing with ordinal variables means we need to specify order of the data.
- We can use the ESTIMATE and CONTRAST statements to determine and test the odds ratios between response levels, explanatory variable levels, and strata.
 - ESTIMATE: EXP option to get the estimated odds ratio, ILINK option to get the estimated cumulative probability θ .
 - CL option is used to get the confidence limits for odds ratio or estimated probabilities.

2.11 Example

Male and female subjects received an active or placebo treatment for their arthritis pain, and the subsequent extent of improvement was recorded as marked, some, or none.

2.11.1 Questions

- Write the correct form of the linear predictor.
- Use the proportional odds model to determine whether there is an effect of sex and treatment.
- List the odds ratio estimates, 95% confidence intervals, and corresponding interpretations.
- Estimate the probability of marked improvement for females in the active group.

2.11.2 Data

Table 9.1 Arthritis

Sex	Treatment	Impr
		Marked
Female	Active	16
Female	Placebo	6
Male	Active	5
Male	Placebo	1

2.12 Example

Male and female subjects received an active or placebo treatment for their arthritis pain, and the subsequent extent of improvement was recorded as marked, some, or none.

2.12.1 Data Set

```
data arthritis;  
input sex $ treatment $ improve $ count @@;  
datalines;
```

```
female active marked 16 female active some 5 female active none 6
female placebo marked 6 female placebo some 7 female placebo none 19
male active marked 5 male active some 2 male active none 7
male placebo marked 1 male placebo some 0 male placebo none 10
;
```

2.12.2 SAS Code

```
proc logistic data=arthritis order=data plots=effect(polybar x=treatment*sex); ①
freq count;
class treatment sex / param=ref;
model improve = sex treatment sex*treatment / selection=forward include=2 scale=none aggregate=none;
estimate "odds between marked vs. some+no improvement for female in placebo group" intercept 1 sex 1;
estimate "odds between marked+some vs. no improvement for female in placebo group" intercept 1 sex 1;
estimate "probability of marked improvement for female in the placebo group" intercept 1 sex 1;
estimate "probability of marked improvement for female in the active trt group" intercept 1 sex 1;
run;
```

- ① The polybar plot creates stacked bar plots showing probability and cumulative probability plots for each level dictated by x.
- ② `selection=forward include=2` tests whether the interaction is needed to be included in the model.
- ③ The `ilink` option calculates the cumulative probability, while `exp` calculates odds ratios. The reason it returns the probability for marked is that the marked level was inputted as the lowest level.

2.12.3 Answer

The linear predictor can be written as

$$\eta_k = \alpha_k + \beta_f x_f + \beta_a x_a$$

where $x_f = 1$ when the participant identifies as female, $x_a = 1$ when the participant is in the active treatment group. Using the residual chi-square test, the interaction term is not significant ($\chi^2 = 0.28, p = 0.60$).

Using the proportional odds model, we determine that there is a significant effect of sex ($\chi^2 = 6.2, p = 0.01$) and treatment ($\chi^2 = 14.4, p = 0.001$).

Females have 3.74 (95% CI: 1.33, 10.55) times higher odds of marked improvement compared to males.

The odds of improvement after undergoing active treatment is 6.03 (95% CI: 2.39, 15.2) times higher compared to the placebo group.

2.13 Exercise

Researchers collected information on subjects' exposure to general air pollution, exposure to pollution in their jobs, and whether they smoked. The response measured was chronic respiratory disease status. Subjects were assigned to one of four possible categories.

2.13.1 Questions

- Use the proportional odds main effects model to test the effect of air pollution, job exposure, and smoking status in the response.
- Estimate the odds ratio between Response I+II+III vs. IV for current vs. ex-smokers.
- Estimate the probability of developing a response level III for an ex-smoker in a high air pollution environment with job exposure.

2.13.2 Data Set

```
data respire;
input air $ exposure $ smoking $ level $ count @@;
datalines;
low no non 1 158 low no non 2 9
low no ex 1 167 low no ex 2 19
low no cur 1 307 low no cur 2 102
low yes non 1 26 low yes non 2 5
low yes ex 1 38 low yes ex 2 12
low yes cur 1 94 low yes cur 2 48
high no non 1 94 high no non 2 7
high no ex 1 67 high no ex 2 8
high no cur 1 184 high no cur 2 65
high yes non 1 32 high yes non 2 3
high yes ex 1 39 high yes ex 2 11
high yes cur 1 77 high yes cur 2 48
low no non 3 5 low no non 4 0
low no ex 3 5 low no ex 4 3
low no cur 3 83 low no cur 4 68
low yes non 3 5 low yes non 4 1
low yes ex 3 4 low yes ex 4 4
low yes cur 3 46 low yes cur 4 60
high no non 3 5 high no non 4 1
```

```

high no ex 3 4 high no ex 4 3
high no cur 3 33 high no cur 4 36
high yes non 3 6 high yes non 4 1
high yes ex 3 4 high yes ex 4 2
high yes cur 3 39 high yes cur 4 51
;

```

2.14 Exercise

Researchers collected information on subjects' exposure to general air pollution, exposure to pollution in their jobs, and whether they smoked. The response measured was chronic respiratory disease status. Subjects were assigned to one of four possible categories.

2.14.1 SAS Code

```

proc logistic data=respire order=data plots=effect(polybar x=air*exposure*smoking);;
freq count;
class air exposure smoking / param=reference;
model level = air exposure smoking air*exposure exposure*smoking air*smoking air*exposure*smoking;
estimate "odds between I+II+III vs. IV for current vs. ex smokers" smoking 1 -1 /exp cl category;
estimate "probability of at most III level for ex smoker, high air pollution, job exposure" air 1 1 1;
estimate "probability of at most II level for ex smoker, high air pollution, job exposure" exposure 1 1 1;
run;

```

2.14.2 Answer

There is no evidence of an air pollution effect ($\chi^2 = 0.18, p = 0.68$). Exposure ($\chi^2 = 82.1, p < 0.001$) and smoking history ($\chi^2 = 209.9, p < 0.001$) are significant.

The odds of being assigned to Response Level III and lower for current smokers is 0.23 times the odds for ex-smokers.

The probability of receiving a response level III and below for an ex smoker, high air pollution, and with job exposure is 0.9351. The probability of receiving a response level II and below for an ex smoker, high air pollution, and with job exposure is 0.8512. Hence, the estimated probability of receiving a response level III is 0.0839 .

2.15 Partial Proportional Odds Model

Sometimes, the $\beta_k = \beta$ assumption does not hold for all explanatory variables, especially if some levels disproportionately affect the odds of success.

Tip

This assumption can be tested using the score test for the proportional odds assumption provided by the LOGISTIC procedure.

To address this, we specify unequal slopes for the explanatory variables that violate the proportional odds assumption.

$$\text{logit}(\theta_{hik}) = \alpha_k + x'_{hi}\beta_k + x'_{h'i'}\gamma$$

where x'_{hi} includes the explanatory variables with unequal slopes β_k and $x'_{h'i'}$ includes the explanatory variables with equal slopes γ . This model is called the **partial proportional odds model**.

2.16 Partial Proportional Odds Model in SAS

- If the score test for the proportional odds assumption is significant at the 0.10 level, we might need to use a partial proportional odds model.
- We can specify the model to have unequal slopes using the UNEQUALSLOPES option in the MODEL statement. UNEQUALSLOPES=X specifies that only the explanatory variable X has unequal slopes
- Testing for unequal slopes can be done using the TEST statement
 - LABEL: TEST statements effectively labels the test.
 - If the test yields a significant result, there is sufficient evidence for unequal slopes for that explanatory variable.
 - If the test is not significant, we can assume the proportional odds model holds.

2.17 Example

The table contains data from a study by a bicycling clothing manufacturer who wanted to assess their test glove. It was designed to combat the hand problems experienced by cyclists dealing with carpal tunnel syndrome. Cyclists wore either the company's standard gel glove or the new test glove for a week's worth of their standard rides. Then they reported whether they experienced major, moderate, or no relief from their usual symptoms (numbness and nerve pain) on the bike.

2.17.1 Questions

- Assume an unequal slopes, main effect model and test for the equality of slopes for glove and gender.
- Perform logistic regression and interpret the results.

2.17.2 Data

Table 9.6 Cycling Glo

Glove Type	Gender	Relief From	
		Major	Mo
Test	Female	12	
Test	Male	8	
Gel	Female	5	
Gel	Male	8	

2.18 Example

The table contains data from a study by a bicycling clothing manufacturer who wanted to assess their test glove. It was designed to combat the hand problems experienced by cyclists dealing with carpal tunnel syndrome. Cyclists wore either the company's standard gel glove or the new test glove for a week's worth of their standard rides. Then they reported whether they experienced major, moderate, or no relief from their usual symptoms (numbness and nerve pain) on the bike.

2.18.1 SAS Code

```
proc logistic order=data data=wrists;  
freq count;  
class glove gender / param=reference order=data;  
model relief= glove gender / scale=none aggregate unequalslopes; ①  
pogender: test genderfemale_major=genderfemale_moderate; ②  
poglove: test glovetest_major=glovetest_moderate; ③  
run;  
  
proc logistic order=data data=wrists;  
freq count;  
class glove gender / param=reference order=data;  
model relief= glove gender / scale=none aggregate unequalslopes=glove; ④  
run;
```

- ① `unequal slopes` tells SAS to provide unequal slopes for each explanatory level at all response levels.
- ② `pogender` is the label for testing whether the slopes should be the same for different response levels for the gender explanatory variable
- ③ `poglove` is the label for testing whether the slopes should be the same for different response levels for the glove explanatory variable
- ④ Only assumes unequal slopes for glove, but proportional odds assumption holds for gender.

2.18.2 Answer

Using a partial proportional odds model, there is a significant effect of the type of glove and gender. the odds of experiencing major relief for females is 2.27 (95% CI: 1.10, 4.66) times that of males. The odds of major relief when using the test glove is 1.4 (0.60, 3.21) times that of gel. Additionally, the odds of moderate relief when using the test glove is 2.91 (1.33, 6.36) times that of gel.

2.19 Exercise

Consider the arthritis example discussed earlier. Test the hypothesis of equal slopes for both sex and treatment.

2.20 Exercise

Consider the arthritis example discussed earlier. Test the hypothesis of equal slopes for both sex and treatment.

2.20.1 SAS Code

```
proc logistic data=arthritis order=data plots=effect(polybar x=treatment*sex);;
freq count;
class treatment sex / param=reference;
model improve = sex treatment / unequalslopes scale=none aggregate;
equal_sex: test sexfemale_marked=sexfemale_some;
equal_treatment: test treatmentactive_marked=treatmentactive_some;
run;
```

2.20.2 Answer

We suppose the slopes are unequal, then test whether there is evidence against the null hypothesis of equal slopes using **test**.

There is insufficient evidence to claim that the β for sex ($\chi^2 = 1.35, p = 0.25$) and treatment ($\chi^2 = 0.03, p = 0.86$).

3 Nominal Response Variables

3.1 Generalized Logits Model

When the response variables are nominal, the concept of cumulative probabilities is less applicable. The **generalized logits model** is used for nominal polytomous response variables.

Generalized Logit Model

The logit function is the logarithm of the ratio of the probability of the level of interest, π_j and the *reference* category π_r .

$$\text{logit}\theta_j = \log\left(\frac{\pi_j}{\pi_r}\right) = \alpha_j + \sum_g \beta_{jg}x_g$$

where g comprises of ALL explanatory variables

Reference level

For the reference level, $\alpha_r = 0, \beta_{rg} = 0$.

3.2 Generalized Logits Model

The resulting expression for π_j can be written as:

$$\pi_j = \frac{\exp[\alpha_j + \sum_g \beta_{jg} x_g]}{\sum_{k=1}^r \exp[\alpha_k + \sum_g \beta_{kg} x_g]}$$

with the constraint $\alpha_r = 0, \beta_{rg} = 0$.

3.3 Generalized Logits Model: SAS

Implementing generalized logit models in SAS

The LOGISTIC procedure will model the logit models for all levels simultaneously. If there are r levels of the response variable, there will be $(r - 1)$ degrees of freedom for a single logit model.

Link function

SAS does not know whether your response variable is ordinal or nominal. The generalized logits model can be specified using `link=glogit` in the `model` statement.

Syntax

- Unless specified, the last variable level based on the specified order in the `class` statement is set as the reference level.
- Syntax is the same as the PO model, but the assumptions are different.

3.4 Example

Schoolchildren in experimental learning settings were surveyed to determine which learning style they preferred. Investigators were interested in whether response was associated with school and their program, which was either a standard school day or also included after-hours care.

3.4.1 Question

Use logistic regression to test for program and school effects and calculate the relevant odds ratios.

3.4.2 SAS Data Set

```
data school;
input school program $ style $ count @@;
datalines;
1 regular self 10 1 regular team 17 1 regular class 26
1 after self 5 1 after team 12 1 after class 50
2 regular self 21 2 regular team 17 2 regular class 26
2 after self 16 2 after team 12 2 after class 36
3 regular self 15 3 regular team 15 3 regular class 16
3 after self 12 3 after team 12 3 after class 20
;

proc freq data=school order=data;
weight count;
tables school*program*style/ nocol norow nopct;
run;
```

3.5 Example

Schoolchildren in experimental learning settings were surveyed to determine which learning style they preferred. Investigators were interested in whether response was associated with school and their program, which was either a standard school day or also included after-hours care. Use logistic regression to test for program and school effects and calculate the relevant odds ratios.

3.5.1 SAS Code

```
proc logistic data=school order=data;
freq count;
class school (ref=first) program (ref=last) / param=ref;
model style= program school school*program / link=gllogit selection=forward include=2 scale=none;
run;
```

3.5.2 Answer

There is sufficient evidence of a program ($\chi^2 = 10.9, p = 0.004$) and school ($\chi^2 = 14.8, p = 0.005$).

The odds of preferring self-directed learning for those assigned to after-hours care is 0.47 (0.27-0.82) times that of those who assigned to regular class hours. The odds of students from school 2 preferring self-directed learning is 2.95 (1.48-5.91) times that of students from school 1. Similarly, the odds of students from school 3 preferring self-directed learning is 3.72 (1.76-7.90) times that of students from school 1.

The odds of preferring team-based learning for those assigned to after-hours care is 0.48 (0.27-0.82) times that of those who assigned to regular class hours. The odds of students from school 2 preferring self-directed learning is 1.20 (0.64-2.23) times that of students from school 1. Similarly, the odds of students from school 3 preferring self-directed learning is 1.93 (0.99-3.75) times that of students from school 1.

3.6 Exercise

The data set “firecracker” shows data of injuries due to firecrackers from three community health centers in rural Philippines. The age groups were coded as A: <25 years of age, B: 25 years of age. Use a generalized logit main effect model to determine whether there is a significant effect of age and clinic in the occurrence of face and finger injury compared to other parts of the body.

3.6.1 Question

Use the estimate statement to test and estimate the odds ratio between clinics 1 and 2 for fingers vs. other injury and face vs. other injury.

3.6.2 Data

```
data firecracker;  
input clinic $ agegroup $ injury $ count;  
datalines;  
1 A Fingers 25  
1 A Face 18  
1 A Other 5  
1 B Fingers 13  
1 B Face 5  
1 B Other 3
```

```

2 A Fingers 42
2 A Face 7
2 A Other 12
2 B Fingers 23
2 B Face 5
2 B Other 5
3 A Fingers 10
3 A Face 4
3 A Other 4
3 B Fingers 9
3 B Face 7
3 B Other 5
;
run;

```

3.7 Exercise

The data set “firecracker” shows data of injuries due to firecrackers from three community health centers in rural Philippines. The age groups were coded as A: <25 years of age, B: 25 years of age. Use a generalized logit main effect model to determine whether there is a significant effect of age and clinic in the occurrence of face and finger injury compared to other parts of the body.

3.7.1 SAS Code

```

proc logistic data=firecracker order=data;
freq count;
class clinic agegroup;
model injury= clinic agegroup clinic*agegroup/link=glogit selection=forward include=2 scale=1;
estimate "clinic 1 vs clinic 2 - face" clinic 1 -1/ exp cl category="Face";
estimate "clinic 1 vs clinic 2 - fingers" clinic 1 -1/ exp cl category="Fingers";
run;

```

3.7.2 Answer

There is a significant effect of clinic ($\chi^2 = 11.9, p = 0.02$), but no evidence of an age effect ($\chi^2 = 0.02, p = 0.99$).

The odds of receiving patients with face injuries in clinic 1 is 4.08 (1.37-12.2) times that of clinic 2. Meanwhile, the odds of receiving patients with finger injuries in clinic 1 is 1.25 (0.49-3.16) times that of clinic 2.

3.8 Exact Generalized Logits Test

- Exact methods are more appropriate for tables with small or zero counts.

Exact Method Warning

Generalized logits tests have more parameters and exact methods are typically more computationally expensive. When using exact methods, try to keep the model small for relatively small data sets.

Exact Method in SAS

The **exact** statement can perform joint tests as well as calculate exact confidence intervals for odds ratios. Use the “Exact” p-value column in the “Probability” row in the exact conditional tests.

3.9 Example

The following data come from a study in which a physical therapy practice experimented with different treatment modalities in a strengthening program. The therapists rotated patients through three programs for maintenance after basic strength had been attained. The strengthening programs included free weights, resistance bands, and weight machines. The therapists are interested in whether gender and age affect choice of strengthening programs.

3.9.1 Data Set

```
data pt;
format therapy $11.;
input gender $ age $ therapy $ count @@;
datalines;
males adult machines 2 males senior machines 10
males adult freeweights 13 males senior freeweights 9
males adult bands 3 males senior bands 3
females adult machines 3 females senior machines 8
females adult freeweights 9 females senior freeweights 0
```

```
females adult bands 1 females senior bands 1
;
```

3.10 Example

The following data come from a study in which a physical therapy practice experimented with different treatment modalities in a strengthening program. The therapists rotated patients through three programs for maintenance after basic strength had been attained. The strengthening programs included free weights, resistance bands, and weight machines. The therapists are interested in whether gender and age affect choice of strengthening programs.

Warning

Small cell counts means we need to use exact methods.

3.10.1 SAS Code

```
proc logistic data=pt;
title "asymptotic";
freq count;
class gender(ref='females') age(ref='senior') / param=ref;
model therapy(ref='machines') = gender age / link=glogit scale=none aggregate;
run;

proc logistic data=pt;
title "exact";
freq count;
class gender(ref='females') age(ref='senior') / param=ref;
model therapy(ref='machines') = gender age / link=glogit scale=none aggregate;
exact gender age/joint estimate=both;
run;
```

3.10.2 Answer

There is a marginal effect of gender ($p=0.099$), *but there is a significant effect of age* ($\chi^2 = 12.2$, $p=0.003$).

The odds of males preferring bands is 4.1 (0.49-59.55) times than that of females. Additionally, the odds of males preferring freeweights is 4.49 (0.93-30.2) times than that of females.

The odds of an adult preferring bands is 5.08 (0.60-53.2) times that of seniors. However, the odds of an adult preferring freeweights is 12.47 (2.7-84.0) times that of seniors.