EAB 770 - Chapter 8

Miguel Fudolig

1 Chapter 8: Logistic Regression: Dichotomous Response

1.1 Outline

- Dichotomous Explanatory Variables
  - The LOGISTIC procedure
  - CLASS statement
- Qualitative Explanatory Variables
- Continuous and Ordinal Explanatory Variables
- Diagnostics
- Likelihood methods
  - The GENMOD procedure

1.2 Review: Mantel-Haenszel Tests

- Mantel-Haenszel statistics investigate association between two variables with adjustment for a stratification variable.
  - General association
  - Location shifts for means
  - Linear/increasing trends -Mantel-Haenszel does not involve statistical models.
  - Can't be used for estimation

1.3 Logistic Regression

- Statistical modeling method that enables us to:
  - make inferences
  - estimate/predict outcome variables

- Describes the relationship between a categorical response variable and multiple explanatory variables
  - Continuous explanatory variables can be included in the analysis
  - Responses can be dichotomous or polytomous

1.4 Logistic Regression

- Possible applications include epidemiology, medical research, agronomy, and social science research.

1.5 Logistic Regression: Dichotomous Variables

- When both response and explanatory variables are dichotomous, it corresponds to a 2×2 table.
  - Rows pertain to groups/populations/subpopulations
- Logistic regression enables us to write a model for the probability of success for each group.

1.6 Logistic Regression: Model

Let θ_{mi} be the probability of success fo the i^{th} group in the m^{th} strata.The logistic model can be used to describe the variation among $\{\theta_{mi}\}$

$$\theta_{mi} = \frac{1}{1 + \exp[-\eta_{mi}]}$$

1.7 Logistic Regression: Model

Where

$$\eta_{mi} = \alpha + \sum_{k=1}^{t} \beta_k x_{mik}$$

and t is the number of explanatory variables.

> **Note**
>
> η_{mi} is referred to as the linear predictor of the model.

1.8 Logistic Regression: Model

Using algebra, we can invert the equation to the following form:

$$\eta_{mi} = \alpha + \sum_{k=1}^{t} \beta_k x_{mik} = \log\left[\frac{\theta_{mi}}{1-\theta_{mi}}\right]$$

Note

$\log\left[\frac{\theta_{mi}}{1-\theta_{mi}}\right]$ is referred to as the **logit** transformation of θ_{mi}, which is the logarithm of the odds.

1.9 Odds and Odds Ratios

The odds of success for an individual in the m^{th} strata and i^{th} group can be calculated as

$$\theta_{mi}/(1-\theta_{mi}) = e^{\eta_{mi}}$$

The odds ratio between the i^{th} and j^{th} group in the m^{th} strata can easily be calculated.

$$OR = \frac{odds_{mj}}{odds_{mi}} = e^{\eta_{mj}}/e^{\eta_{mi}} = e^{\eta_{mj}-\eta_{mi}}$$

1.10 Advantages

- The logit transformation enables us to monitor the change in log odds.
- Easy to calculate ods ratio.
- The linear predictor can take on values from $(-\infty, \infty)$, but the values of θ_{mi} will be constrained in [0,1].

Tip

If you used linear regression to model a proportion, it assumes that the response variable can be negative or above 1.

1.11 Fit Statistics

- The model performance can be measured using the **deviance** and **goodness-of-fit** statistics.

> **Note**
>
> Both use chi-square statistics to assess model fit.

1.12 Chi-square statistics

> **Note**
>
> Likelihood ratio chi-square measures the difference between two models by using the ratio of the two likelihoods as a chi-square statistic.

> **Note**
>
> Pearson chi-square goodness of fit uses observed and expected values to test for a goodness of fit.

1.13 Chi-square statistics

- If the chi-square statistic divided by the degrees of freedom is higher than one, then the data has some variation that is unaccounted for in the model.
  - This is called overdispersion
- The high ratio between the chi-square statistic and degrees of freedom can also mean that the data does not follow a binomial distribution well.
  - Heterogeneity
- Both cases are accounted for using generalized linear mixed models/generalized estimating equations.

1.14 Chi-square statistics: sample size

Sample size guidelines for the chi-square assumption for these fit statistics include:

- Each group should have at least ten subjects.
- 80% of the predicted counts should be at least 5.

- No 0 counts, remaining 20% of the counts should be above 2.
- Failure to satisfy these assumptions means exact tests should be used.

1.15 Chi-square statistics: hypothesis test

The null hypothesis for both deviance and goodness-of-fit tests is that the departures of the observed data from the predicted values based on the model are random.

- Failure to reject the null hypothesis means we have no evidence against good agreement between observed and predicted values based on the model.

1.16 Fit Statistics

- The Akaike Infomation Criterion (AIC) and the Bayesian Information Criterion (BIC) can also be used to assess model fit.

> **Note**
>
> The Bayesian Information Criterion is also known as the Schwarz Criterion (SC).

> **Note**
>
> AIC predicts estimation error based on the provided model. BIC also predicts estimation error but uses a different penalization scheme for errors.

- The AIC (or BIC) for the full model (intercept + covariates) should be lower than the AIC (or BIC) of the intercept-only model.

1.17 The Main Effect Model

Recall that the linear predictor can be written as:

$$\eta_i = \alpha + \beta_1 x_{i1} + \beta_2 x_{i2} + \ldots$$

> **Note**
>
> α is the intercept of the linear model

> **Note**
>
> β_k are the coefficients of the explanatory variables. These coefficients will tell us the effect of predictor x_{ik}.

1.18 The Main Effect Model

- The current form of the linear predictor assumes the explanatory variables are in quantitative form.
- When the explanatory variables are categorical, we need to create dichotomous dummy variables to denote each level of the variables.
  - The case $x_{ik} = 0$ is called the reference level of the variables. When all $x_{ik} = 0$ the linear predictor collapses to the intercept α.

> **Note**
>
> For dummy dichotomous variables, odds ratios can be calculated as an effect size of association.

1.19 The Main Effect Model

> **Note**
>
> The main effect model follows a regression framework, which means we can use ANOVA to test for effect significance.

1.20 The LOGISTIC Procedure

The LOGISTIC procedure can perform logistic regression in SAS.

- Can take both categorical and continuous explanatory variable
- Syntax is slightly different from the FREQ procedure.

1.21 The LOGISTIC Procedure

The LOGISTIC procedure provides estimates for β as in regression analysis.

1.21.1 Sample Syntax

```
proc logistic data=data;
freq count;                                                    ①
model y(event='Yes') = x1 x2 /scale=none aggregate;            ②
run;
```

① `freq` functions similar to `weight` in the `FREQ` procedure. When the data consists of an observation for each row, this statement is not needed.

② `model` specifies the model for the response `y` using the explanatory variables `x1` and `x2`. `event` tells SAS what the "success" response level is. The `scale` option produces goodness-of-fit statistics. The `aggregate` option requests that SAS treat each unique combination of the explanatory variable values as a distinct group in computing the goodness-of-fit statistics.

1.22 Example

The data shows people who visited a clinic and required a catheterization. The response is the presence of coronary artery disease. The explanatory variables are ECG segment depression and sex. Use logistic regression methods to obtain odds ratio estimates and create a logistic regression model to estimate the probability of developing CA disease based on sex and ECG segment depression.

Table 3.7 Coronary Artery Disease Data

Sex	ECG	No Disease	Disease	Total
Female	$<$ 0.1 ST segment depression	11	4	15
Female	$\geq$ 0.1 ST segment depression	10	8	18
Male	$<$ 0.1 ST segment depression	9	9	18
Male	$\geq$ 0.1 ST segment depression	6	21	27

1.22.1 Data Set

```
data coronary;
input sex ecg ca count @@;
datalines;
0 0 0 11 0 0 1 4
```

```
0 1 0 10 0 1 1 8
1 0 0 9 1 0 1 9
1 1 0 6 1 1 1 21
;
```

1.23 Example

The data shows people who visited a clinic and required a catheterization. The response is the presence of coronary artery disease. The explanatory variables are ECG segment depression and sex. Use logistic regression methods to obtain odds ratio estimates and create a logistic regression model to estimate the probability of developing CA disease based on sex and ECG segment depression.

1.23.1 SAS Code

```
proc logistic data=coronary ;
freq count;
model ca(event='1')=sex ecg / scale=none aggregate;
run;
```

1.23.2 Answer

Both chi-square tests had p-values above 0.05, which meant there was no evidence that the errors were not random.

The AIC values for the full model are lower than the intercept-only model, which meant model performance increased after including our covariates.

Assuming the probability of developing CA is given by $\theta_C A$, the logistic model was estimated to be:

$$logit\ [\hat{\theta_{CA}}] = -1.17 + 1.28 * SEX + 1.05 * ECG$$

There is a significant effect of sex (β = 1.28, p=0.01) and segment depression (β = 1.05, p=0.03) on the probability of developing CA disease. In terms of odds ratios, the odds that males develop CA is 3.58 (95% CI: 1.35,9.52) times that of females. Additionally, the odds of developing CA with higher segment depression is 2.87 (95% CI: 1.08, 7.62) times that of patients with lower segment depression in their ECG.

1.24 Exercise

A simple random sample of individuals from different age groups were asked about their caffeine and exercise habits.

1.24.1 Data

Age Group	Caffeine	Yes Exercise	No Exercise
Young Adult	Yes Caffeine	121	158
Young Adult	No Caffeine	36	97
Older Adult	Yes Caffeine	44	132
Older Adult	No Caffeine	15	130

1.24.2 Questions

- Use logistic regression to estimate the logistic model for the data. Use Young Adult and No caffeine as the reference levels.
- Is there a significant age group effect on exercise habits?
- Is there a significant effect of caffeine intake on exercise habits?
- Calculate and interpret the appropriate odds ratios for age group and caffeine intake.
- Does the model fit the data well?

1.25 Exercise

A simple random sample of individuals from different age groups were asked about their caffeine and exercise habits.

1.25.1 SAS Code

```
data caf;
input agegrp  caffeine  exercise  count;
datalines;
0 1 1 121
0 1 0 158
0 0 1 36
0 0 0 97
1 1 1 44
1 1 0 132
1 0 1 15
```

```
1 0 0 130
;

proc logistic;
freq count;
model exercise(event='1')=agegrp caffeine / scale=none aggregate;
run;
```

1.25.2 Answer

Suppose the probability of exercise is given by θ,The estimated logistic regression model is: $logit(\theta) = -1.08 - 0.93 * AGEGRP + 0.84 * CAFFEINE$.

There is a significant effect of age in the likelihood of exercising. The odds of an older participant reporting regular exercise is 0.39 (95% CI: 0.28, 0.56; p<0.001) times that of younger participants.

There is a significant effect of caffeine intake on exercise. The odds of a participant reporting regular exercise when they consume caffeine is 2.3 (95% CI: 1.6,3.3; p<0.001) times that of those who do not consume caffeine.

The deviance (p=0.39) and goodness-of-fit chi-square tests (p=0.40) showed no evidence against the randomness of errors, which indicates good model fit. The AIC and SC values also decreased after including covariates, also supporting good fit.

1.26 CLASS statement

- The `class` statement can create dummy variable coding automatically through the LOGISTIC procedure.
  - The REF option allows us to set the reference level for each variable.

```
proc logistic;
class cat;
model y = x cat/scale=none aggregate;
run;
```

1.27 Model parametrizations

> **Note**
>
> **Effect parametrization** produces deviation from the *grand mean*. This is the default parametrization in the LOGISTIC procedure.
>
> - The reference level corresponds to $X = -1$.

> **Note**
>
> **Reference cell parametrization** produces measureents based on reference level. This can be specified in the class statement as `param=ref` in the LOGISTIC procedure.
>
> - The reference level corresponds to $X = 0$.

1.28 Example

The data is based on a study on prison sentencing for persons convicted of a burglary or larceny. The researchers wanted to test whether the type of crime (nonresidential or other) and prior arrests (some or none) affect whether the offender was sent to prison.

1.28.1 Data

Table 8.5 Sentencing Data

Type	Prior Arrest	Prison	No Prison	Total
Nonresidential	Some	42	109	151
Nonresidential	None	17	75	92
Other	Some	33	175	208
Other	None	53	359	412

1.28.2 Questions

- Perform a logistic regression analysis to estimate the appropriate logistic model for the data. Comment on the significance of the main effects and estimate the odds ratio and

the 95% CI.
- Using effect parametrization
- Using reference cell parametrization

1.29 Example

The data is based on a study on prison sentencing for persons convicted of a burglary or larceny. The researchers wanted to test whether the type of crime (nonresidential or other) and prior arrests (some or none) affect whether the offender was sent to prison.

- Perform a logistic regression analysis to estimate the appropriate logistic model for the data. Comment on the significance of the main effects and estimate the odds ratio and the 95% CI.
- Using effect parametrization
- Using reference cell parametrization

1.29.1 SAS Code

```
data sentence;
input type $ prior $ sentence $ count @@;
datalines;
nrb some y 42 nrb some n 109
nrb none y 17 nrb none n 75
other some y 33 other some n 175
other none y 53 other none n 359
;

proc logistic data=sentence order=data;
title "effect";
class type(ref="nrb") prior(ref="none") / ;                                    ①
freq count;
model sentence(event='y') = type prior / scale=none aggregate;
run;


proc logistic data=sentence order=data;
title "reference";
class type(ref="nrb") prior(ref="none") / param=ref;                           ②
freq count;
model sentence(event='y') = type prior / scale=none aggregate;
```

```
run;
```

① Default: effect parametrization
② Reference cell parametrization used after `param=ref` is specified.

1.29.2 Answer

Effect parametrization: $logit(\theta_{ij}) = -1.48 - 0.30 * TYPE + 0.17 * PRIOR$

Reference Cell parametrization: $logit(\theta_{ij}) = -1.36 - 0.59 * TYPE + 0.35 * PRIOR$

The odds of sending an offender to prison after committing other crimes is 0.55 times (95% CI: 0.38, 0.81) that when committing non-residential crimes. There is strong evidence against the null hypothesis of no difference.

The odds of sending an offender with some prior arrests are 1.42 times (95% CI: 0.974, 2.06) times more likely to be sent to prison than those who have no prior arrests. There is moderate evidence against the null hypothesis of no difference.

1.30 Exercise

Recall the age x caffeine x exercise habits exercise. Use the class statement to perform logistic regression using effect and reference cell parametrization. Comment on the significance of the main effects and estimate the odds ratio and the 95% CI.

1.31 Exercise

Recall the age x caffeine x exercise habits exercise. Use the class statement to perform logistic regression using effect and reference cell parametrization. Comment on the significance of the main effects and estimate the odds ratio and the 95% CI.

1.31.1 SAS Code

```
title "effect";
proc logistic data=caf;
freq count;
class agegrp (REF='0') caffeine (REF='0');
model exercise(event='1')=agegrp caffeine / scale=none aggregate;
run;
```

```
title "ref cell";
proc logistic data=caf;
freq count;
class agegrp (REF='0') caffeine (REF='0')/param=ref;
model exercise(event='1')=agegrp caffeine / scale=none aggregate;
run;
```

1.31.2 Answer

Effect parametrization: $logit(\theta_{ij}) = -1.12 - 0.47 * AGEGRP + 0.42 * CAFFEINE$

Reference Cell parametrization: $logit(\theta_{ij}) = -1.08 - 0.93 * AGEGRP + 0.84 * CAFFEINE$

There is strong evidence of an effect of age (p<0.001). The odds of older adults reporting regular exercise is 0.39 (95% CI: 0.28, 0.56) times that of younger adults.

There is strong evidence of an association with caffeine consumption (p<0.001). The odds of reporting regular exercise for participants who consumed caffeine is 2.32 times that of those who do not consume caffeine.

1.32 Saturated models

- Saturated models often include an interaction term $A * B$, which tests the difference of the effect of B in the different levels of A.

> **Simple Effects**
>
> The effect of B conditional on the level of A is referred to as *simple effects*.

1.33 Interaction Models

The interaction model in the logistic regression framework can be written as:

$$logit(\theta) = \alpha + \beta_1 x_1 + \beta_2 x_2 + \beta_{12} x_1 x_2$$

> **Tip**
>
> Interaction terms should be tested first ($x_1 x_2$). If the interaction is not significant, it can be dropped from the model, leaving you with a main effects model.

1.34 Interaction models

1.34.1 Interaction tests

- Interaction tests are Wald chi-square tests with $(a-1)(b-1)$ degrees of freedom, where a is the number of levels of A and b is the number of levels of B.
- The degrees of freedom is also equal to the number of coefficients that should be allotted to the interactions.
  - Interactions can be visually recognized using interaction plots.

1.34.2 Figure



1.35 Non-Dichotomous Explanatory Variables

- What if the explanatory variables have more than two levels?
  - Nominal
  - Ordinal
  - Continuous

1.36 Non-Dichotomous Explanatory Variables

- Recall: Regression models
  - If an explanatory variable has n levels, $(n-1)$ dichotomous dummy variables are needed to be produced to include the explanatory variable in the regression model.
  - In interaction models, this leads to more interaction terms.

For a variable with $(a-1)$ levels

$$logit(\theta) = \alpha + \beta_1 x_{a1} + \beta_2 x_{a2} + ... + \beta_{(a-1)} x_{a,a-1}$$

- The matrix form of the predictor variables is called the Design Matrix.

1.37 Non-Dichotomous Explanatory Variables

For a model with two variables/factors: one with $(a-1)$ levels and the other with $(b-1)$ levels, the interaction model can be written as:

$$logit(\theta) = \alpha + \beta_{a1} x_{a1} + \beta_{a2} x_{a2} + ... + \beta_{a,(a-1)} x_{a,a-1} + \beta_{b1} x_{b1} + \beta_{b2} x_{b2} + ... + \beta_{b(b-1)} x_{b,b-1} + Interaction$$

$$Interaction = \beta_{a1b1} x_{a1} x_{b1} + \beta_{a1b2} x_{a1} x_{b2} + \beta_{a2b1} x_{a2} x_{b1} + ... + \beta_{a,a-1;b,b-1} x_{a,a-1} x_{b,b-1}$$

1.38 Nominal Explanatory Variables

Since there is no order in the levels of explanatory variables, we might be interested in the difference between specific levels

- Example: How did the difference in placebo and test treatments compare between Hospital 1 and Hospital 2?
- Example: How did the difference in placebo and test treatments compare in Hospital 2?

1.39 Contrasts

- Contrasts are used in performing a formal test for estimable effects
- The null hypotheses for these tests are usually the hypothesis of no effect
  - "There is no main effect of Variable/ Factor A"
  - "There is no interaction"
  - "The estimated treatment mean of the first level of Factor A and first level of Factor B is zero"

1.40 Contrasts: Estimable Effects

Suppose we refer to $logit(\theta_{ij})$ as η_{ij}.

> **Estimable Effects**
>
> Estimable effects are effects that can be expressed as the sum or difference of linear predictors η_{ij}.

Example: How did the difference in placebo and test treatments compare between Hospital 1 and Hospital 2?

- The null hypothesis can be expressed as $\eta_{t1} - \eta_{p1} - (\eta_{t2} - \eta_{p2}) = 0$. This effect is estimable!
- The expression can be expressed in terms of β_i by substituting the linear predictor expressions.

1.41 Contrasts: Joint Tests

Suppose we have an explanatory variable/factor A with three levels, which requires **two** different dummy variables. The regression model can be expressed as:

$$logit(\theta) = \eta = \alpha + \beta_1 X_{a1} + \beta_2 X_{a2}$$

To test for a main effect of A, we need to test the following null hypotheses simultaneously: $\beta_1 = 0$ and $\beta_2 = 0$. These could be done using **joint tests**.

- The joint tests use Wald statistics similar to an ANOVA.

1.42 Contrasts: SAS

We use the `contrast` statement in the `LOGISTIC` procedure to test these contrasts.

> **Note**
>
> In SAS, we focus on the coefficients of each term defined in the linear predictor. Each linear predictor is broken down into the effect and interaction terms and simplified. The resulting **coefficients** would then make it to the contrast statement.

1.43 Example

Suppose we are interested in modeling the mortality rate of cardiothoracic surgery across 4 hospitals. The resulting logistic regression model would be:

$$\eta = \alpha + \beta_1 X_{h1} + \beta_1 X_{h2} + \beta_3 X_{h3}$$

Suppose we want to test for a difference in mortality between hospital 1 and hospital 4.

1.43.1 Reference Cell

The effect due to Hospital 1 corresponds to the case $(X_{h1}, X_{h2}, X_{h3} = 1, 0, 0)$ yields a regression model given by: $\alpha + \beta_1$. Hospital 4 corresponds to the case $(X_{h1}, X_{h2}, X_{h3} = 0, 0, 0)$, which yields a regression model given by: α.

Then the difference between Hospital 1 and Hospital 4 can be calculated as

$$\alpha + \beta_1 - \alpha = \beta_1$$

. T he corresponding null hypothesis of no difference can the be written as

$$1 * \beta_1 + 0 * \beta_2 + 0 * \beta_3 = 0$$

In SAS, we can code this as:

```
contrast A 1 0 0 ;
```

1.43.2 Effect

The effect due to Hospital 1 corresponds to the case $(X_{h1}, X_{h2}, X_{h3} = 1, 0, 0)$ yields a regression model given by: $\alpha + \beta_1$. Hospital 4 corresponds to the case $(X_{h1}, X_{h2}, X_{h3} = -1, -1, -1)$, which yields a regression model given by: $\alpha - \beta_1 - \beta_2 - \beta_3$.

Then the difference between Hospital 1 and Hospital 4 can be calculated as

$$\alpha + \beta_1 - [\alpha - \beta_1 - \beta_2 - \beta_3] = 2\beta_1 + \beta_2 + \beta_3$$

.

The corresponding null hypothesis of no difference can the be written as:

$$2\beta_1 + 1 * \beta_2 + 1 * \beta_3 = 0$$

In SAS, we can code this as:

```
contrast A 2 1 1 ;
```

1.44 Exercise

Consider a Factor A with three levels (A_1, A_2, A_3), with A_3 as the reference level. Provide the corresponding null hypotheses and formulate the SAS contrast statements for both reference cell and effect parametrization.

- There is no difference between the first and third levels of A.
- The mean effect of A_1 and A_2 is equal to the effect of A_3.

1.45 Exercise

Consider a Factor A with three levels (A_1, A_2, A_3), with A_3 as the reference level. Provide the corresponding null hypotheses and formulate the SAS contrast statements for both reference cell and effect parametrization.

- There is no effect difference between the first and third levels of A.
- The mean effect of A_1 and A_2 is equal to the effect of A_3.

The logit model can be expressed as:

$$\eta = \alpha + \beta_1 X_1 + \beta_2 X_2$$

1.45.1 Item 1

Reference Cell: Effect $A_1 = \alpha + \beta_1$ Effect $A_3 = \alpha$

No effect difference between the first and third levels: $A_1 - A_3 = 0 \rightarrow \beta_1 = 0$

```
contrast A 1 0;
```

Effect parametrization: Effect $A_1 = \alpha + \beta_1$ Effect $A_3 = \alpha - \beta_1 - \beta_2$

No effect difference between the first and third levels: $A_1 - A_3 = 0 \rightarrow 2\beta_1 + \beta_2 = 0$

```
contrast A 2 1;
```

1.45.2 Item 2

Reference Cell: Effect $A_1 = \alpha + \beta_1$ Effect $A_2 = \alpha + \beta_2$ Effect $A_3 = \alpha$

No effect difference between the first and third levels: $(A_1 + A_2)/2 - A_3 = 0 \rightarrow \beta_1 + \beta_2 = 0$

```
contrast A 1 1;
```

Effect parametrization: Effect $A_1 = \alpha + \beta_1$ Effect $A_2 = \alpha + \beta_2$ Effect $A_3 = \alpha - \beta_1 - \beta_2$

No effect difference between the first and third levels: $(A_1 + A_2)/2 - A_3 = 0 \rightarrow 1 * \beta_1 + 1 * \beta_2 = 0$

```
contrast A 1 1;
```

1.46 Exercise

Consider a Factor A with three levels (A_1, A_2, A_3) and Factor B with two levels (B_1, B_2), with A_3 and B_2 as the reference level. Suppose θ is a probability of observing success for a dichotomous response variable y.

- Please write the saturated/interaction logistic model for θ

Provide the corresponding null hypothesis and formulate the SAS contrast statements for both reference cell and effect parametrizations.

- "There is no interaction between A and B".

1.47 Exercise

Consider a Factor A with three levels (A_1, A_2, A_3) and Factor B with two levels (B_1, B_2), with A_3 and B_2 as the reference level. Suppose θ is a probability of observing success for a dichotomous response variable y.

- Please write the saturated/interaction logistic model for θ

$$\eta = \alpha + \beta_{A1} x_{A1} + \beta_{A2} x_{A2} + \beta_{B1} x_{B1} + \beta_{A1B1} x_{A1} x_{B1} + \beta_{A2B1} x_{A2} x_{B1}$$

Provide the corresponding null hypotheses and formulate the SAS contrast statements for both reference cell and effect parametrizations.

- "There is no interaction between A and B".

- The joint tests of the following hypotheses should be tested: $\beta_{A1B1} = 0$ and $\beta_{A2B1} = 0$.

```
contrast A*B 1 0,
         A*B 0 1;
```

1.48 Implementing Qualitative Explanatory Variables in SAS

- Use the `class` statement in `LOGISTIC` procedure to identify qualitative variables.
- Interaction between Factors A and B are included in the model as $A * B$
  - To include the saturated model of Factors A and B, you can write (response variable)=`A|B`.
  - You can add more than two factors in these statements.
  - Three-way interaction: `A*B*C`
  - Saturated model for a three-factor model: `A|B|C`
- If you wish to only include up to two-way interactions in a three-factor model: `A|B|C@2`
- You can introduce contrasts in the `contrast` statement.
- `effectplot` can be used to create interaction plots.
- `oddsratio` calculates odds ratios and graphs of confidence limits for odds ratios.

1.49 Example

The data set is from a cross-sectional prevalence study conducted to investigate textile worker complaints about respiratory symptoms experienced while working in the mills. Investigators were interested in whether occupational environment was related to prevalence of respiratory ailments associated with the disease byssinosis.

- Use logistic regression to model the probability of complaining based on workplace condition, years of employment, and smoking history. Include only up to two-way interaction effects.
- Is there evidence of any two-way interactions at a significance level of 0.10?
- Generate an interaction plot for these cases.
- Generate the main effects using contrast statements.

1.49.1 Table

Table 8.9 Byssinosis Complaints

Workplace Condition	Years of Employment	Smoking	Complaints Yes	Complaints No
Dusty	< 10	Yes	30	203
Dusty	< 10	No	7	119
Dusty	≥ 10	Yes	57	161
Dusty	≥ 10	No	11	81
Not Dusty	< 10	Yes	14	1340
Not Dusty	< 10	No	12	1004
Not Dusty	≥ 10	Yes	24	1360
Not Dusty	≥ 10	No	10	986

1.49.2 SAS Data Set

```
data byss;
input work $ years $ smoke $ status $ count @@;
datalines ;
dusty <10 yes yes 30 dusty <10 yes no 203
dusty <10 no yes 7 dusty <10 no no 119
dusty >=10 yes yes 57 dusty >=10 yes no 161
dusty >=10 no yes 11 dusty >=10 no no 81
not <10 yes yes 14 not <10 yes no 1340
not <10 no yes 12 not <10 no no 1004
not >=10 yes yes 24 not >=10 yes no 1360
not >=10 no yes 10 not >=10 no no 986
;
```

1.50 Example

The data set is from a cross-sectional prevalence study conducted to investigate textile worker complaints about respiratory symptoms experienced while working in the mills. Investigators were interested in whether occupational environment was related to prevalence of respiratory ailments associated with the disease byssinosis.

1.50.1 Questions

- Use logistic regression to model the probability of complaining based on workplace condition, years of employment, and smoking history. Include only up to two-way interaction effects.

- Is there evidence of any two-way interactions at a significance level of 0.10?
- Generate an interaction plot for these cases.
- Generate the main effects using contrast statements.

1.50.2 SAS Code

```
proc logistic data=byss;
freq count;
class work years(ref=first) smoke(ref=first) /param=ref;
model status(event=last) = work|years|smoke@2 /scale=none aggregate;
effectplot interaction (x=work) / at(smoke=all years=all) link noobs;
oddsratios work;
contrast 'work' work 1;
contrast 'years' years 1;
contrast 'smoke' smoke 1;
run;
```

1.50.3 Answer

$Work * Smoke$ (p = 0.07) is significant at the 0.10 level.

1.51 Exercise

The following data come from a study on urinary tract infections. Investigators applied three treatments to patients who had either a complicated or uncomplicated diagnosis of urinary tract infection. Since complicated cases of urinary tract infections are difficult to cure, investigators were interested in whether the pattern of treatment differences is the same across diagnoses.

- Are we interested in main effects or interaction?
- Is there a significant interaction effect?
- Calculate the odds ratio between the diagnoses of UTI for different treatments.
- Generate plots for the interaction and odds ratios.

1.51.1 Table

Table 8.7 Urinary Tract Infection Data

Diagnosis	Treatment	Cured	Not Cured	Proportion Cured
Complicated	A	78	28	0.736
Complicated	B	101	11	0.902
Complicated	C	68	46	0.596
Uncomplicated	A	40	5	0.889
Uncomplicated	B	54	5	0.915
Uncomplicated	C	34	6	0.850

1.51.2 SAS Data Set

```
data uti;
input diagnosis : $13. treatment $ response $ count @@;
datalines;
complicated A cured 78 complicated A not 28
complicated B cured 101 complicated B not 11
complicated C cured 68 complicated C not 46
uncomplicated A cured 40 uncomplicated A not 5
uncomplicated B cured 54 uncomplicated B not 5
uncomplicated C cured 34 uncomplicated C not 6
;
```

1.52 Exercise

The following data come from a study on urinary tract infections. Investigators applied three treatments to patients who had either a complicated or uncomplicated diagnosis of urinary tract infection. Since complicated cases of urinary tract infections are difficult to cure, investigators were interested in whether the pattern of treatment differences is the same across diagnoses.

- Are we interested in main effects or interaction?
- Is there a significant interaction effect?
- Calculate the odds ratio between the diagnoses of UTI for different treatments.
- Generate plots for the interaction and odds ratios.

1.52.1 SAS Code

```
data uti;
input diagnosis : $13. treatment $ response $ count @@;
datalines;
complicated A cured 78 complicated A not 28
complicated B cured 101 complicated B not 11
complicated C cured 68 complicated C not 46
uncomplicated A cured 40 uncomplicated A not 5
uncomplicated B cured 54 uncomplicated B not 5
uncomplicated C cured 34 uncomplicated C not 6
;

proc logistic data=uti order=data;
freq count;
class diagnosis treatment /param=ref;
model response(event='cured') = diagnosis|treatment /scale=none aggregate;
effectplot interaction (x=treatment) / at(treatment=all) link noobs;
oddsratios diagnosis;
run;
```

1.52.2 Answer

We are more interested in the interaction effect.

There is no significant interaction effect between diagnosis and treatment.

Odds Ratio Estimates and Wald Confidence Intervals			
Odds Ratio	Estimate	95% Confidence Limits	
diagnosis complicated vs uncomplicated at treatment=A	0.348	0.125	0.971
diagnosis complicated vs uncomplicated at treatment=B	0.850	0.281	2.574
diagnosis complicated vs uncomplicated at treatment=C	0.261	0.101	0.671

Figure 1: Odds Ratios

1.53 Continuous and Ordinal Explanatory Variables

- Sometimes, continuous explanatory variables are unique in the data set. The sample size requirements for the goodness-of-fit tests will not be satisfied.
  - Income

  - Age
  - Years in University

- There should be at least 5 observations for the rarer outcome per parameter in the full model.
  - If there are 50 observations of the rarer outcome, then the maximum number of parameters supported is 10.

> **Note**
>
> Continuous variables should **NOT** be included in the class statement.

1.54 Model Selection: Logistic Regression

- We can use the residual score statistic Q_{RS} to determine up to what extent does the residuals from the model are linearly associated with other potential explanatory variables.
  - DF: difference in number of parameters between models
- The `SELECTION=FORWARD` option should be included in the `MODEL` statement. The `LOGISTIC` procedure will then start from an intercept model and build it up to the most parsimonious model.
  - The `INCLUDE` option tells SAS how many of the variables listed should always be included in the model.

1.55 Model Selection: Logistic Regression

- SAS also provides the Hosmer and Lemeshow Goodness-of-Fit test that tests the adequacy for these data by including the `lackfit` option.
  - Null hypothesis: The model is adequate, i.e., the residuals are explained by the model.
- The UNITS option specifies at what increment of the continuous variable will SAS calculate the odds ratios.

1.56 Example

The data set CORONARY is from the same study on coronary artery disease as previously analyzed; in addition, the continuous variable AGE is an explanatory variable. The variable ECG is now treated as an ordinal variable, with values 0, 1, and 2. ECG is coded 0 if the ST segment depression is less than 0.1, 1 if it equals 0.1 or higher but less than 0.2, and 2 if the ST segment depression is greater than or equal to 0.2. The variable AGE is age in years.

- What is the resulting logistic model?
- Calculate the odds ratio for patients that differ in age by 10 years.

1.56.1 SAS Data Set

```
data coronary;
input sex ecg age ca @@ ;
datalines;
0 0 28 0 1 0 42 1 0 1 46 0 1 1 45 0
0 0 34 0 1 0 44 1 0 1 48 1 1 1 45 1
0 0 38 0 1 0 45 0 0 1 49 0 1 1 45 1
0 0 41 1 1 0 46 0 0 1 49 0 1 1 46 1
0 0 44 0 1 0 48 0 0 1 52 0 1 1 48 1
0 0 45 1 1 0 50 0 0 1 53 1 1 1 57 1
0 0 46 0 1 0 52 1 0 1 54 1 1 1 57 1
0 0 47 0 1 0 52 1 0 1 55 0 1 1 59 1
0 0 50 0 1 0 54 0 0 1 57 1 1 1 60 1
0 0 51 0 1 0 55 0 0 2 46 1 1 1 63 1
0 0 51 0 1 0 59 1 0 2 48 0 1 2 35 0
0 0 53 0 1 0 59 1 0 2 57 1 1 2 37 1
0 0 55 1 1 1 32 0 0 2 60 1 1 2 43 1
0 0 59 0 1 1 37 0 1 0 30 0 1 2 47 1
0 0 60 1 1 1 38 1 1 0 34 0 1 2 48 1
0 1 32 1 1 1 38 1 1 0 36 1 1 2 49 0
0 1 33 0 1 1 42 1 1 0 38 1 1 2 58 1
0 1 35 0 1 1 43 0 1 0 39 0 1 2 59 1
0 1 39 0 1 1 43 1 1 0 42 0 1 2 60 1
0 1 40 0 1 1 44 1
;
```

1.57 Example

The data set CORONARY is from the same study on coronary artery disease as previously analyzed; in addition, the continuous variable AGE is an explanatory variable. The variable ECG is now treated as an ordinal variable, with values 0, 1, and 2. ECG is coded 0 if the ST segment depression is less than 0.1, 1 if it equals 0.1 or higher but less than 0.2, and 2 if the ST segment depression is greater than or equal to 0.2. The variable AGE is age in years.

1.57.1 SAS Code

```
proc logistic descending;
model ca=sex ecg age ecg*ecg age*age
sex*ecg sex*age ecg*age /selection=forward include=3 details lackfit;   ①
units age=10;                                                           ②
run;
```

① `include=3` indicates we are including the first three variables in the model. In this case, it's sex, ecg, and age.

② Looks at the odds ratio due to a 10-point difference in age.

1.57.2 Answer

The logistic model is as follows:

$$logit(\theta) = -5.64 + 1.35 * Sex + 0.873 * ECG + 0.09 * Age$$

The odds of developing a coronary artery disease in patients older than 10 years is 2.5 times higher.

1.58 Exercise

Chapter8-Corn.csv includes the data of 82 plants from a randomized controlled trial that tests the effect of a new formulation of pesticide as well as the effect of using cover crops in planting corn. The farmers are also interested if the yield (measured in kg) is associated to the plant discoloration.

- Use the forward selection approach to determine if an interaction term between cover crop and pesticide needs to be included in the model.
- What is the resulting logistic model?
- Calculate the odds ratio for corn plants that have a yield difference of 2 kg.

1.59 Exercise

Chapter8-Corn.csv includes the data of 82 plants from a randomized controlled trial that tests the effect of a new formulation of pesticide as well as the effect of using cover crops in planting corn. The farmers are also interested if the yield (measured in kg) is associated to the plant discoloration.

1.59.1 SAS Code

```
proc logistic data=corn;
class pesticide (ref="No") covercrop (ref="No");
model Discolored(event="Yes") = pesticide covercrop yield pesticide*covercrop/selection=forwa
run;
```

1.59.2 Answer

There was no evidence to indicate that a pesticide*covercrop interaction needed to be included ($Q_{RS} = 0.08, p = 0.78$). The Hosmer-Lemeshow goodness-of-fit test also provides no evidence against model adequacy ($\chi^2 = 6.9, p = 0.55$).

The resulting logistic model is:

$$logit(\theta) = 0.38 - 0.68 * Pesticide - 1.10 * CoverCrop - 0.27 * Yield$$

The odds of discoloration decreases by 0.42 (1-0.58) times when the yield increases by 2 kg.

2 Model Diagnostics

2.1 Diagnostics

- We are interested in how a model fits our data and the lack of fit.
  - Lack of fit: Should we still consider other variables to include in the model?
- Model fit: Pearson chi-square and deviance chi-square tests
- Lack of fit: Pearson *residuals* and deviance *residuals*

Note

- Pearson residuals look at the difference between the observed and expected occurrences in a group. A value of 2 is usually indicative of a lack of fit.
- Deviance residuals are the components that make up the deviance statistic
- Likelihood residuals: weighted average of standardized Pearson and Deviance residuals

2.2 Diagnostics

- These residuals can be viewed in graphical form.

> **Tip**
>
> In SAS, we can view the residual plots by specifying `PLOTS=INFLUENCE` in the `PROC LOGISTIC` statement. `stdres` specifies plots for standardized residuals
>
> ```
> proc logistic plots(label)=influence(unpack stdres);
> ```

2.3 Example

Consider a modified version of the UTI data set. Generate the standardized residual plot, likelihood residual plot, and the Pearson chi-square residual plot.

```
data uti2;
input diagnosis : $13. treatment $ response trials;
datalines;
complicated A 78 106
complicated B 101 112
complicated C 68 114
uncomplicated A 40 45
uncomplicated B 54 59
uncomplicated C 34 40
;
```

2.4 Example

Consider a modified version of the UTI data set. Generate the standardized residual plot, likelihood residual plot, and the Pearson chi-square residual plot.

2.4.1 SAS Code

```
proc logistic data=uti2
plots(label)=influence(unpack stdres);          (1)
class diagnosis treatment / param=ref;
model response/trials = diagnosis treatment;    (2)
run;
```

① `plots=influence` yield all residual plots. `stdres` generates the standardized residual plot.
② The model can also be expressed in terms of # of success/sample size;

2.4.2 Answer

Residuals did not exceed 2, which indicate good fit.

2.5 Alternative Estimation Methods

- We have been using maximum likelihood (ML) estimation to estimate values of β, which leads to odds ratio estimates.
  - What if there are no successes or failures in your data? (Complete separation)
  - What if there is a small number of successes in the data?
  - What if the sample size is small?
- These conditions might yield non-converging solutions
  - Infinite odds ratios
  - Nonsensical odds ratio estimates

2.6 Complete Separation

Table 8.11 Infinite Odds Ratio Example

Factor	Response=Yes	Response=No
1	15	0
2	0	34

Figure 2: Complete Separation: One level includes all successes or all failures.

Note

If nearly all the observations have a probability of 1 of being allocated to the correct response group, then the data configuration may be one of **quasi-complete separation**.

2.7 Dealing with non-convergence

- Exact tests
  - Non-convergence due to small cell counts
  - Quasi-complete separation

Tip

The `EXACT` statement calculates the exact p-values for each variable/factor. Adding a `joint` option performs joint tests for the variables included. We use the results of the joint *probability* tests as its results are analogous to the Fisher's exact test.

Warning

The output of the exact statement typically yields a one-sided p-value for median unbiased estimates and one-sided confidence intervals. To get a two-sided 95% confidence interval, we can use a one-sided 97.5% confidence interval by specifying `alpha = 0.025`.

- Firth's penalized likelihood
  - Complete separation/quasi-complete separation when there are continuous explanatory variables
  - Applies a penalty in the likelihood function.

Tip

Adding the `FIRTH` option to the model statement tells SAS to use the Firth penalized likelihood. `CLPARM=PL` uses the likelihood profile consistent with the Firth penalized likelihood method.

2.8 Example

Consider the following data set:

- Does this display complete separation, quasi-complete separation, or neither?
- Calculate exact p-values for the effect of treatments A and B as well as the joint tests.
- Calculate exact estimates for the odds ratio.

```
data preclinical;
input treatA $ treatB $ response $ count @@;
datalines;
no no yes 0 no no no 0
yes no yes 8 yes no no 0
no yes yes 0 no yes no 2
yes yes yes 21 yes yes no 6
;
```

2.9 Example

Consider the following data set:

- Does this display complete separation, quasi-complete separation, or neither?
- Calculate exact p-values for the effect of treatments A and B as well as the joint tests.
- Calculate exact estimates for the odds ratio.

2.9.1 SAS Code

```
proc logistic data=preclinical;
freq count;
class treatA (ref="no") treatB (ref="no") /param=ref ;
model response (event="yes") = treatA treatB ;
exact treatA treatB /joint estimate=both alpha=0.025;
run;
```

2.9.2 Answer

The data shows quasi-complete separation.

The odds of a successful response is for those who received Treatment A is 7.06 (CI: 0.52, ∞) times that of those who did not receive Treatment A. The exact p-value is 0.07.

The odds of a successful response is for those who received Treatment B is 0.35 (CI: 0, 2.85) times that of those who did not receive Treatment B.The exact p-value is 0.18.

2.10 Example: Complete Separation

Consider the following data set:

- Does this display complete separation, quasi-complete separation, or neither?
- Calculate exact p-values for the effect of treatments A and B as well as the joint tests.
- Calculate estimates for the odds ratio from the Firth penalized likelihood approach.

```
data complete;
input gender region count response @@;
datalines;
0 0 0 1 0 0 5 0
0 1 1 1 0 1 0 0
1 0 0 1 1 0 175 0
1 1 53 1 1 1 0 0
;
```

2.11 Example: Complete Separation

Consider the following data set:

- Does this display complete separation, quasi-complete separation, or neither?
- Calculate exact p-values for the effect of treatments A and B as well as the joint tests.
- Calculate estimates for the odds ratio from the Firth penalized likelihood approach.

2.11.1 SAS Code

```
proc logistic data=complete;
freq count;
model response = gender region / firth clparm=pl;
exact gender region/estimate=both;
run;
```

2.11.2 Answer

Based on the exact tests, there appears to be no effect of gender (p=0.5). However, there is an effect of region.

There is complete separation in region. We can use the penalized likelihood approach to calculate odds ratio estimates. The estimated odds ratio between region=1 and region = 0 is 10.5 (95% CI: 7.55, 16.26).

3 The GENMOD procedure

3.1 The GENMOD procedure

- The GENMOD procedure uses generalized estimating equations in performing logistic regression when the responses are correlated.
  - Similar syntax as the LOGISTIC procedure.
- Reference levels are included in the estimates, but are automatically zeroed out by the GENMOD procedure.
  - You can set the reference levels by applying the `ref` option to each variable in the class statement.

3.1.1 Sample Code

```
proc genmod;
freq count;
class A B;
model response = A B A*B/link=logit dist=binomial type3 aggregate;   ①
estimate '[NAME OF CONTRAST]' A 1 -1/exp cl;                          ②
run;
```

① We need to specify the link function and the distribution as GENMOD can be used to model other distributions like Poisson. `type3` invokes the Type III tests of fixed effects, which looks at partial testing of each effect.

② GENMOD does not output odds ratio by default. Hence, we use the `estimate` statement. The `estimate` statement follows the `contrast` statement, but since it calculates effect sizes, fractions cannot be eliminated by multiplying a factor unlike contrast statements.

3.2 Example

Consider the UTI example. Perform a logistic regression analysis using PROC GENMOD.

3.3 Example

Consider the UTI example. Perform a logistic regression analysis using PROC GENMOD.

3.3.1 SAS Code

```
proc genmod data=uti;
freq count;
class diagnosis treatment;
model response = diagnosis treatment /link=logit dist=binomial type3 aggregate;
contrast 'treatment' treatment 1 0 -1 ,
                     treatment 0 1 -1;                                          (1)
contrast 'A-B' treatment 1 -1 0;
contrast 'A-C' treatment 1 0 -1;
estimate 'odds ratio of diagnosis' diagnosis 1 -1/exp;
run;
```

(1) This contrast should yield the same values as that of the Type 3 test for treatment.

3.4 Exercise

Consider the Byssinosis example. Perform a logistic regression analysis by applying an interaction model that includes up to two-way interactions in PROC GENMOD.

- Test for the main effect of workplace condition using contrasts. Compare the chi-square and p-values obtained through the Type 3 tests.
- Test for the interaction of years of employment and smoking for dusty workplace conditions.
- Calculate the odds ratio of developing byssinosis between the two work conditions.

3.4.1 SAS Data Set

```
data byss;
input work $ years $ smoke $ status $ count @@;
datalines ;
dusty <10 yes yes 30 dusty <10 yes no 203
dusty <10 no yes 7 dusty <10 no no 119
dusty >=10 yes yes 57 dusty >=10 yes no 161
dusty >=10 no yes 11 dusty >=10 no no 81
not <10 yes yes 14 not <10 yes no 1340
not <10 no yes 12 not <10 no no 1004
not >=10 yes yes 24 not >=10 yes no 1360
not >=10 no yes 10 not >=10 no no 986
;
```

3.5 Exercise

Consider the Byssinosis example. Perform a logistic regression analysis by applying an interaction model that includes up to two-way interactions in PROC GENMOD.

3.5.1 SAS Code

```
proc genmod data=byss;
freq count;
class work years smoke;
model status(event='yes') = work years smoke work*years work*smoke years*smoke /link=logit di
contrast 'work' work 1 -1; * same as the Type 3 Tests;
contrast 'interaction' work*years 1 -1 -1 1;
estimate 'OR for work' work 1 -1/e exp;
run;
```

3.5.2 Answer

The workplace effect has a $\chi^2 = 114.10$, p<.0001. There is sufficient evidence of a workplace main effect.

The interaction term between years of work and smoking yielded a $\chi^2 = 0.91$, p=0.34. There is insufficient evidence of a workplace main effect.

The estimated odds of developing byssinosis for dusty work places is 11.6 (95% CI: 7.96, 17.03) times that of non-dusty places