EAB 770 - Chapter 15

Miguel Fudolig

1 Generalized Estimating Equations

1.1 Outline

- Generalized Estimating Equations Methodology
  - Vs. Generalized Linear model and Weighted Least Squares
- Marginal model
- Wald Statistic
- Count Data
- Proportional Odds Data
- Missing Values
- Overdispersion

1.2 Generalized Estimating Equations (GEE)

- Recall: Weighted Least Squares method
  - Can handle categorical outcome variables but cannot handle continuous explanatory variables.
  - No missing values.
  - Also needs a high sample size.
- Generalized Estimating equations can also do what the WLS method can.

> **Note**
>
> GEEs can handle missing observations, continuous explanatory variables, and time-changing explanatory variables.

Applications

Examples of GEE applications include crossover studies, longitudinal data analysis, observational studies, and non-Gaussian responses.

1.3 GEE: Motivation

- We need to account for correlations when the data is correlated
  - Longitudinal study: Within-subject corelations
  - Cross-sectional study with clusters: Between-subject correlations.
- Correlations could lead to an underestimation of standard errors of between-subject effects and the overestimation of the standard errors of within-subject effects.

Note

What sets GEE apart from other methods is that we can account for the covariance between outcomes by specifying a structure for the covariance matrix.

- Can account for discrete and continuous response data.

1.4 GEE: Framework

Recall: GLM model such that

$$\eta(\mu) = X\beta = \alpha + \beta_1 X_1 + \beta_2 X_2 ...$$

To estimate the coefficients , we need to assume a form for the working correlation matrix, which we denote as R.

Note

R may depend on a vector of unknown parameters α. All components of R should be estimated as part of the analysis.

1.5 Working Correlation Matrix Forms

1.5.1 Independence

Independence: $\mathbf{R} = \boldsymbol{R}_0 = \boldsymbol{I}$.

$$\mathbf{R} = \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

1.5.2 Exchangeable/Compound Symmetry

Exchangeable:

$$\text{Corr}(y_{ij}, y_{i,j'}) = \left\{ \begin{matrix} 1 & j = j' \\ \alpha & j \neq j' \end{matrix} \right\}$$

$$\mathbf{R} = \begin{bmatrix} 1 & \alpha & \alpha & \alpha \\ \alpha & 1 & \alpha & \alpha \\ \alpha & \alpha & 1 & \alpha \\ \alpha & \alpha & \alpha & 1 \end{bmatrix}$$

1.5.3 M-dependent

$$\text{Corr}(y_{ij}, y_{i,j+s}) = \left\{ \begin{matrix} 1 & s = 0 \\ \alpha_s & s = 1, 2, \ldots, m \\ 0 & s > m \end{matrix} \right]$$

$$\mathbf{R} = \begin{bmatrix} 1 & \alpha_1 & \alpha_2 & 0 \\ \alpha_1 & 1 & \alpha_1 & \alpha_2 \\ \alpha_2 & \alpha_1 & 1 & \alpha_1 \\ 0 & \alpha_2 & \alpha_1 & 1 \end{bmatrix}$$

1.6 Working Correlation Matrix Forms

1.6.1 AR(1) - Auto-Regressive

Auto-regressive (AR-1):

$$\text{Corr}(y_{ij}, y_{i,j+s}) = \alpha^s \quad s = 0, 1, 2, \ldots, t_i - j$$

$$\mathbf{R} = \begin{bmatrix} 1 & \alpha & \alpha^2 & \alpha^3 \\ \alpha & 1 & \alpha & \alpha^2 \\ \alpha^2 & \alpha & 1 & \alpha \\ \alpha^3 & \alpha^2 & \alpha & 1 \end{bmatrix}$$

1.6.2 ANTE(1) - antedependence model

$$\sum_{ik} = \begin{bmatrix} \sigma_1^2 & \sigma_1\sigma_2\rho_1 & \sigma_1\sigma_3\rho_1\rho_2 & \sigma_1\sigma_4\rho_1\rho_2\rho_3 & \sigma_1\sigma_5\rho_1\rho_2\rho_3\rho_4 & \sigma_1\sigma_6\rho_1\rho_2\rho_3\rho_4\rho_5 \\ & \sigma_2^2 & \sigma_2\sigma_3\rho_2 & \sigma_2\sigma_4\rho_2\rho_3 & \sigma_2\sigma_5\rho_2\rho_3\rho_4 & \sigma_2\sigma_6\rho_2\rho_3\rho_4\rho_5 \\ & & \sigma_3^2 & \sigma_3\sigma_4\rho_3 & \sigma_3\sigma_5\rho_3\rho_4 & \sigma_3\sigma_6\rho_3\rho_4\rho_5 \\ & & & \sigma_4^2 & \sigma_4\sigma_5\rho_4 & \sigma_4\sigma_6\rho_4\rho_5 \\ & & & & \sigma_5^2 & \sigma_5\sigma_6\rho_5 \\ & & & & & \sigma_6^2 \end{bmatrix}$$

1.6.3 Unstructured (Fixed)

$$\text{Corr}(y_{ij}, y_{i,j'}) = \begin{Bmatrix} 1 & j = j' \\ \alpha_{jj'} & j \neq j' \end{Bmatrix}$$

$$\mathbf{R} = \begin{bmatrix} 1 & \alpha_{21} & \alpha_{31} & \alpha_{41} \\ \alpha_{21} & 1 & \alpha_{32} & \alpha_{42} \\ \alpha_{31} & \alpha_{32} & 1 & \alpha_{43} \\ \alpha_{41} & \alpha_{42} & \alpha_{43} & 1 \end{bmatrix}$$

1.7 Estimating the Covariance Matrix

- For GEE models, we use the empirical sandwich estimator which is based on the Fisher information matrix.

Fisher Information Matrix

The Fisher information matrix is the variance of the score function.
The score function measures the steepness of the logarithm of the likelihood function.

Quasi-likelihood Information Criteria (QIC)

You can use the QIC to compare the fit of different working correlation structures on the data set.
-Lower QIC, better fit.

1.8 Contrasts

You can still test any hypothesis with the form

$$H_0 : C\beta = 0$$

as a contrast with an approximate chi-square distribution under H_0. The degrees of freedom is equal to the number of rows of the contrast matrix.

1.9 GEE: Notes

Note

The GEEs reduce to GLM estimating equations for cross-sectional data.

Missing data

While the GEE method handles missing values, it requires missing data to be completely at random (MCAR). MCAR implies that missing values are unrelated to the data. More information about missing data can be found here

Note

GEE methods are robust to an assigned correlation structure!

1.10 GEE: Marginal Model

GEEs produce marginal model, which yields estimates that are **population averaged**

- Conditional models, such as the conditional logistic regression and generalized linear mixed models (`GLIMMIX`), focus on individual-based predictions.

1.11 GEE in SAS

- The `GENMOD` procedure can perform GEE using the `REPEATED` statement.
- The `REPEATED` statement includes a `SUBJECT` option that sets the clustering variable.
  - The `TYPE` option requests the working correlation structure.
  - The `COVB` option requests the parameter estimate covariance matrix
  - The `CORRW` option specifies the final working correlation.
- GEE can be used for both binary and count data. When unspecified, it uses a Gaussian distribution.
  - `DIST` option can be used to specify distributions.
  - `DIST=BIN` uses binary
  - `DIST=POISSON` uses Poisson distribution to model count data

Tip

The `GLIMMIX` procedure can also perform a GEE-type analysis by adding `_residual_` to all random effects.

1.12 Example

The data shown is from a hypothetical study of the effects of air pollution children. Researchers followed 25 children and recorded whether they were exhibiting wheezing symptoms during the periods of evaluation at ages 8, 9, 10, and 11. The response is recorded as 1 for symptoms and 0 for no symptoms. Explanatory variables included age, city, and a passive smoking index with values 0, 1, and 2 that reflected the degree of smoking in the home.

1.12.1 SAS Data Set

```
data children;
   input id city$ @@;
   do i=1 to 4;
      input age smoke symptom @@;
      output;
   end;
   datalines;
 1 steelcity  8 0 1  9 0 1  10 0 1  11 0 0
 2 steelcity  8 2 1  9 2 1  10 2 1  11 1 0
 3 steelcity  8 2 1  9 2 0  10 1 0  11 0 0
 4 greenhills 8 0 0  9 1 1  10 1 1  11 0 0
 5 steelcity  8 0 0  9 1 0  10 1 0  11 1 0
 6 greenhills 8 0 1  9 0 0  10 0 0  11 0 1
 7 steelcity  8 1 1  9 1 1  10 0 1  11 0 0
 8 greenhills 8 1 0  9 1 0  10 1 0  11 2 0
 9 greenhills 8 2 1  9 2 0  10 1 1  11 1 0
10 steelcity  8 0 0  9 0 0  10 0 0  11 1 0
11 steelcity  8 1 1  9 0 0  10 0 0  11 0 1
12 greenhills 8 0 0  9 0 0  10 0 0  11 0 0
13 steelcity  8 2 1  9 2 1  10 1 0  11 0 1
14 greenhills 8 0 1  9 0 1  10 0 0  11 0 0
15 steelcity  8 2 0  9 0 0  10 0 0  11 2 1
16 greenhills 8 1 0  9 1 0  10 0 0  11 1 0
17 greenhills 8 0 0  9 0 1  10 0 1  11 1 1
18 steelcity  8 1 1  9 2 1  10 0 0  11 1 0
19 steelcity  8 2 1  9 1 0  10 0 1  11 0 0
20 greenhills 8 0 0  9 0 1  10 0 1  11 0 0
21 steelcity  8 1 0  9 1 0  10 1 0  11 2 1
22 greenhills 8 0 1  9 0 1  10 0 0  11 0 0
23 steelcity  8 1 1  9 1 0  10 0 1  11 0 0
24 greenhills 8 1 0  9 1 1  10 1 1  11 2 1
25 greenhills 8 0 1  9 0 0  10 0 0  11 0 0
;
```

1.12.2 SAS Code

```
proc genmod data=children descending;
   class id city;
   model symptom = city age smoke  /
```

```
                    link=logit dist=bin type3;
   repeated subject=id / type=exch covb corrw ;
   estimate 'age' age 1/exp;
run;
```

1.12.3 Answer

Age is the "time" factor. There is no effect of city ($\chi^2 = 0.01$. p=0.94) and there is moderate evidence of a smoking index effect ($\chi^2 = 3.59$. p=0.06). There is moderate evidence of an age effect ($\chi^2 = 2.74$. p=0.10), which means there is moderate evidence of changes in incidence of wheezing as age changes. The odds of wheezing is estimated to decrease by 27% when the child turns a year older.

1.13 Exercise

The DIAGNOS2 dataset includes the data from a one population observational study involving 793 subjects. For each subject, two diagnostic procedures (standard and test) were carried out at each of two times. The results of the four evaluations were classified as positive or negative. Use an interaction model and GEE techniques to determine whether there is a significant between-group (procedure) effect and within-group (time) effect.

-Hint: The response variable is binary. - Use an exchangeable (compound symmetry) working covariance matrix

1.13.1 SAS Data Set

```
data diagnos;
   input std1 $ test1 $ std2 $ test2 $ count;
   do i=1 to count;
    output;
   end;
   datalines;
Neg Neg Neg Neg 509
Neg Neg Neg Pos   4
Neg Neg Pos Neg  17
Neg Neg Pos Pos   3
Neg Pos Neg Neg  13
Neg Pos Neg Pos   8
Neg Pos Pos Neg   0
Neg Pos Pos Pos   8
```

```
Pos Neg Neg Neg  14
Pos Neg Neg Pos   1
Pos Neg Pos Neg  17
Pos Neg Pos Pos   9
Pos Pos Neg Neg   7
Pos Pos Neg Pos   4
Pos Pos Pos Neg   9
Pos Pos Pos Pos 170
;
data diagnos2;
   set diagnos;
    drop std1 test1 std2 test2;
   subject=_n_;
   time=1; procedure='standard';
   response=std1; output;
    time=1; procedure='test';
   response=test1; output;
   time=2; procedure='standard';
    response=std2; output;
    time=2; procedure='test';
   response=test2; output;
run;
```

1.14 Exercise

The DIAGNOS2 dataset includes the data from a one population observational study involving 793 subjects. For each subject, two diagnostic procedures (standard and test) were carried out at each of two times. The results of the four evaluations were classified as positive or negative. Use an interaction model and GEE techniques to determine whether there is a significant between-group (procedure) effect and within-group (time) effect.

1.14.1 SAS Code

```
proc genmod data=diagnos2 descending;
   class procedure time subject;
   model response = procedure time procedure*time  /
                   link=logit dist=bin type3;
   repeated subject=subject / type=exch covb corrw ;
   lsmeans procedure/diff exp cl ilink;
run;
```

1.14.2 Answer

The correlation between time points is approximately 0.8.

There is no significant interaction effect between procedure and time (χ^2=2.49, p=0.11). There is strong evidence of a procedure effect ($ ^2 = 8.17, p=0.0043). The odds of a positive response is 13% (OR = 1.1265, 95% CI: 1.04, 1.22) higher in the standard procedure compared to the test procedure.

1.15 GEE for Count Data

We can use the Poisson and negative binomial distributions to model count data.

> **Note**
>
> We can call the Poisson distribution using `DIST=POISSON`. For negative binomial, we can use `DIST = NB`

1.16 Example

The FRACTURE2 data set comes from a study where researchers evaluated a new drug to treat osteoporosis in women past menopause. In a double-blind study, a group of women were assigned the treatment and a group of women were assigned the placebo. Both groups of women were provided with calcium supplements, given nutritional counseling, and encouraged to be physically active through the exercise programs made available to them. The study ran for three years, and the number of fractures occurring in each of those years was recorded. The length of each of the years, the corresponding risk periods, is 12 months. However, there were a few drop-outs in the third year, and those risk periods were set at 6 months. The offset variable is log of months at risk, as contained in the variable `LMONTHS`.

1.16.1 SAS Data Set

```
data fracture;
    input ID age center $ treatment $ year1 year2 year3 @@;
    total=year1+year2+year3;
    lmonths=log(12);
    datalines;
 1  56 A p 0 0 0   2 71 A p 1 0 0   3 60 A p 0 0 1   4 71 A p 0 1 0
 5  78 A p 0 0 0   6 67 A p 0 0 0   7 49 A p 0 0 0
 9  75 A p 1 0 0   8 68 A p 0 0 0  11 82 A p 0 0 0
```

```
13   56 A p 0 0 0  12 71 A p 0 0 0  15 66 A p 1 0 0
17   78 A p 0 0 0  16 63 A p 0 2 0  19 61 A p 0 0 0
21   75 A p 1 0 0  20 68 A p 0 0 0  23 63 A p 1 1 1
25   54 A p 0 0 0  24 65 A p 0 0 0  27 71 A p 0 0 0
29   56 A p 0 0 0  28 64 A p 0 0 0  31 78 A p 0 0 2
33   76 A p 0 0 0  32 61 A p 0 0 0  35 76 A p 0 0 0
37   74 A p 0 0 0  36 56 A p 0 0 0  39 62 A p 0 0 0
41   56 A p 0 0 0  40 72 A p 0 0 1  43 76 A p 0 0 0
45   75 A p 0 0 0  44 77 A p 2 2 0  47 78 A p 0 0 0
49   71 A p 0 0 0  48 68 A p 0 0 0  51 74 A p 0 0 0
53   69 A p 0 0 0  52 78 A p 1 0 0  55 81 A p 2 0 1
57   68 A p 0 0 0  56 77 A p 0 0 0  59 77 A p 0 0 0
61   75 A p 0 0 0  60 83 A p 0 0 0  63 72 A p 0 0 0  64 88 A p 0 0 0
65   69 A p 0 0 0  66 55 A p 0 0 0  67 76 A p 0 0 0  68 55 A p 0 0 0
69   63 A t 0 0 2  70 52 A t 0 0 0  71 56 A t 0 0 0  72 52 A t 0 0 0
73   74 A t 0 0 0  74 61 A t 0 0 0  75 69 A t 0 0 0  76 61 A t 0 0 0
77   84 A t 0 0 0  78 76 A t 0 1 0  79 59 A t 0 0 1  80 76 A t 0 0 0
81   66 A t 0 0 1  82 78 A t 0 0 1  83 77 A t 0 0 0  84 75 A t 1 0 0
85   75 A t 0 0 0  86 62 A t 0 0 0  87 67 A t 0 0 0  88 62 A t 0 0 0
89   71 A t 0 0 0  90 63 A t 0 0 0                   92 68 A t 0 0 0
93   69 A t 0 0 0  94 61 A t 0 0 0                   96 61 A t 0 0 0
97   67 A t 0 0 0  98 77 A t 0 0 0 91   70 A t 0 0 1 102 81 A t 0 0 0
95   49 A t 0 0 0 106 55 A t 0 0 0
99   63 A t 2 1 0 100 52 A t 0 0 0 101 48 A t 0 0 0
103 71 A t 0 0 0 104 61 A t 0 0 0 105 74 A t 0 0 0
107 67 A t 0 0 0 108 56 A t 0 0 0 109 54 A t 0 0 0
111 56 A t 0 0 0 112 77 A t 1 0 0 113 65 A t 0 0 0
115 66 A t 0 0 0 116 71 A t 0 0 0 117 71 A t 0 0 0 128 71 A t 0 0 0
119 86 A t 1 0 0 120 81 A t 0 0 0 121 64 A t 0 0 0 132 76 A t 0 0 0
123 71 A t 0 0 0 124 76 A t 0 0 0 125 66 A t 0 0 0 136 76 A t 0 0 0
1    68 B p 0 0 0   2 63 B p 0 0 0  3   66 B p 0 0 0   4 63 B p 0 0 0
5    70 B p 0 1 0   6 62 B p 0 0 0  7   54 B p 1 0 0   8 66 B p 0 0 0
9    71 B p 0 0 0  10 76 B p 0 0 0 11   72 B p 0 0 1  12 65 B p 0 1 0
13   55 B p 0 1 0  14 59 B p 0 0 2 15   61 B p 1 0 0  16 56 B p 0 1 0
17   54 B p 0 0 0  18 68 B p 0 0 0 19   68 B p 0 0 0  20 81 B p 0 0 0
21   81 B p 1 0 0  22 61 B p 2 0 1 23   72 B p 1 0 0  24 67 B p 0 0 0
25   56 B p 0 0 0  26 66 B p 0 0 0 27   71 B p 0 1 0  28 75 B p 0 1 0
29   76 B p 0 0 0  30 73 B p 2 0 0 31   56 B p 0 0 0  32 89 B p 0 0 0
33   56 B p 0 0 0  34 78 B p 0 0 0 35   55 B p 0 0 0  36 73 B p 0 0 1
37   71 B p 0 0 0  38 56 B p 0 0 0 39   69 B p 0 0 0  40 77 B p 0 0 0
41   89 B p 0 0 0  42 63 B p 0 0 0 43   67 B p 0 0 0  44 73 B p 0 0 0
45   60 B p 0 0 0  46 67 B p 0 0 0 47   56 B p 0 0 0  48 78 B p 0 0 0
```

```
49  73 B t 1 0 0  50 76 B t 0 0 0 51  61 B t 0 0 0  52 81 B t 0 0 0
53  55 B t 0 0 0  54 82 B t 0 0 0 55  78 B t 0 0 0  56 60 B t 0 0 0
57  56 B t 0 0 0  58 83 B t 0 0 0 59  55 B t 0 0 0  60 60 B t 0 0 0
61  80 B t 0 0 0  62 78 B t 0 0 0 63  67 B t 0 0 0  64 67 B t 0 0 0
65  56 B t 0 0 0  66 72 B t 0 0 0 67  71 B t 0 0 0  68 83 B t 0 0 0
69  66 B t 0 0 0  70 71 B t 0 0 1 71  78 B t 1 0 2  72 61 B t 0 0 0
73  56 B t 0 0 0  74 61 B t 0 0 0 75  55 B t 0 0 0  76 69 B t 1 1 0
77  71 B t 0 0 0  78 76 B t 0 0 0 79  56 B t 0 0 0  80 75 B t 0 0 0
81  89 B t 0 0 0  82 77 B t 0 0 0 83  77 B t 1 0 0  84 73 B t 0 0 0
85  60 B t 0 0 0  86 61 B t 0 0 0 87  79 B t 0 0 0  88 71 B t 0 0 0
89  61 B t 0 0 0  90 79 B t 0 0 0 91  87 B t 1 0 0  92 55 B t 0 0 0
93  55 B t 0 0 0  94 79 B t 0 0 0 95  66 B t 0 0 0  96 49 B t 0 0 0
97  56 B t 0 0 0  98 64 B t 0 0 0 99  88 B t 0 0 0 100 62 B t 1 0 0
101 80 B t 0 0 1 102 65 B t 0 0 0 103 57 B t 0 0 1 104 85 B t 0 0 0
;
data fracture2;
   set fracture;
   drop year1-year3;
   year=1; fractures=year1; output;
   year=2; fractures=year2; output;
   do; if center = A then do;
     if (ID=85 or ID=66 or ID=124 or ID=51) then lmonths=log(6); end;
     if center = B then do;
     if (ID=29 or ID=45 or ID=55) then lmonths=log(6); end;
   end;
   year=3; fractures=year3; output;
run;
```

1.16.2 SAS Code

```
proc genmod data=fracture2;
   class id treatment center year;
   model fractures = center treatment age year  /
                     dist=poisson type3 offset=lmonths;
  repeated subject=id(center) / type=exch corrw;
  estimate "fracture rate increase" treatment 1 -1/exp;
run;
```

1.16.3 Answer

The correlation between time points is estimated to be 0.13.

There is no significant interaction between treatment and time ($\chi^2 = 3.15$, p=0.21), which means the effect of the treatment does not change in time. There is a significant main effect of the treatment ($\chi^2 = 4.59$, p=0.03) and year ($\chi^2 = 7.64$, p=0.02), which implies significant between- and within-group differences. The fracture rate in the placebo group is expected to be 105% higher than that of the treatment group.

1.17 Exercise

Implement a GEE analysis for the FRACTURE2 data set using an unstructured working correlation matrix. Comment on the results of the Type 3 tests. Use the QIC to determine whether the unstructured working correlation matrix is better than the compound symmetry correlation matrix in describing the correlated nature of the data points.

1.18 Exercise

Implement a GEE analysis for the FRACTURE2 data set using an unstructured working correlation matrix (type = UNSTR). Comment on the results of the Type 3 tests. Use the QIC to determine whether the unstructured working correlation matrix is better than the compound symmetry correlation matrix in describing the correlated nature of the data points.

1.18.1 SAS Code

```
proc genmod data=fracture2;
   class id treatment center year;
   model fractures = center treatment age year treatment*year  /
                     dist=poisson type3 offset=lmonths;
  repeated subject=id(center) / type=unstr corrw;
  lsmeans treatment/diff exp cl ilink;
  estimate "fracture rate increase" treatment 1 -1/exp;
run;
```

1.18.2 Answer

Using the unstructured form, there is no significant interaction between treatment and time ($\chi^2 = 3.13$, p=0.21), which means the effect of the treatment does not change in time. There is a significant main effect of the treatment ($\chi^2 = 4.60$, p=0.03) and year ($\chi^2 = 7.68$, p=0.02), which implies significant between- and within-group differences. The fracture rate in the placebo group is expected to be 105% higher than that of the treatment group.

The QIC for the unstructured structure was 368.2, while the QIC for the exchangeable/compound symmetry was 367.2. Thus, the exchangeable structure fits the data better.

1.19 Overdispersion in GEE

> **Note**
>
> GEE can account for overdispersion by adding a measure of variability to adjust for overdispersion, which will reflect in the standard errors. Typically, this is applied to the sampling unit level. This is slightly different from the GLM approach, which will only change a single scale factor.
> A negative binomial GEE can also be implemented to account for overdispersion.

1.20 Example

Researchers studying the incidence of lower respiratory illness in infants took repeated observations of infants over one year. They studied 284 children and examined them every two weeks. Explanatory variables evaluated included passive smoking (one or more smokers in the household), socioeconomic status, and crowding. One outcome of interest was the total number of times, or counts, of lower respiratory infection recorded for the year. The data set LRI includes all the data from the study.

- Compare the standard errors from a naive Poisson GLM, quasi-Poisson GLM (scale=Pearson), and GEE.

1.20.1 SAS Data Set

```
data lri;
   input id count risk passive crowding ses agegroup race @@;
   logrisk =log(risk/52);
   datalines;
```

```
 1 0 42 1 0 2 2 0 96 1 41 1 0 1 2 0   191 0 44 1 0 0 2 0
 2 0 43 1 0 0 2 0 97 1 26 1 1 2 2 0   192 0 45 0 0 0 2 1
 3 0 41 1 0 1 2 0 98 0 36 0 0 0 2 0   193 0 42 0 0 0 2 0
 4 1 36 0 1 0 2 0 99 0 34 0 0 0 2 0   194 1 31 0 0 0 2 1
 5 1 31 0 0 0 2 0 100 1  3 1 1 2 3 1  195 0 35 0 0 0 2 0
 6 0 43 1 0 0 2 0 101 0 45 1 0 0 2 0  196 1 35 1 0 0 2 0
 7 0 45 0 0 0 2 0 102 0 38 0 0 1 2 0  197 1 27 1 0 1 2 0
 8 0 42 0 0 0 2 1 103 0 41 1 1 1 2 1  198 1 33 0 0 0 2 0
 9 0 45 0 0 0 2 1 104 1 37 0 1 0 2 0  199 0 39 1 0 1 2 0
10 0 35 1 1 0 2 0 105 0 40 0 0 0 2 0  200 3 40 0 1 2 2 0
11 0 43 0 0 0 2 0 106 1 35 1 0 0 2 0  201 4 26 1 0 1 2 0
12 2 38 0 0 0 2 0 107 0 28 0 1 2 2 0  202 0 14 1 1 1 1 1
13 0 41 0 0 0 2 0 108 3 33 0 1 2 2 0  203 0 39 0 1 1 2 0
14 0 12 1 1 0 1 0 109 0 38 0 0 0 2 0  204 0  4 1 1 1 3 0
15 0  6 0 0 0 3 0 110 0 42 1 1 2 2 1  205 1 27 1 1 1 2 1
16 0 43 0 0 0 2 0 111 0 40 1 1 2 2 0  206 0 36 1 0 0 2 1
17 2 39 1 0 1 2 0 112 0 38 0 0 0 2 0  207 0 30 1 0 2 2 1
18 0 43 0 1 0 2 0 113 2 37 0 1 1 2 0  208 0 34 0 1 0 2 0
19 2 37 0 0 0 2 1 114 1 42 0 1 0 2 0  209 1 40 1 1 1 2 0
20 0 31 1 1 1 2 0 115 5 37 1 1 1 2 1  210 0  6 1 0 1 1 1
21 0 45 0 1 0 2 0 116 0 38 0 0 0 2 0  211 1 40 1 1 1 2 0
22 1 29 1 1 1 2 1 117 0  4 0 0 0 3 0  212 2 43 0 1 0 2 0
23 1 35 1 1 1 2 0 118 2 37 1 1 1 2 0  213 0 36 1 1 1 2 0
24 3 20 1 1 2 2 0 119 0 39 1 0 1 2 0  214 0 35 1 1 1 2 1
25 1 23 1 1 1 2 0 120 0 42 1 1 0 2 0  215 1 35 1 1 2 2 0
26 1 37 1 0 0 2 0 121 0 40 1 0 0 2 0  216 0 43 1 0 1 2 0
27 0 49 0 0 0 2 0 122 0 36 1 0 0 2 0  217 0 33 1 1 2 2 0
28 0 35 0 0 0 2 0 123 1 42 0 1 1 2 0  218 0 36 0 1 1 2 1
29 3 44 1 1 1 2 0 124 1 39 0 0 0 2 0  219 1 41 0 0 0 2 0
30 0 37 1 0 0 2 0 125 2 29 0 0 0 2 0  220 0 41 1 1 0 2 1
31 2 39 0 1 1 2 0 126 3 37 1 1 2 2 1  221 1 42 0 0 0 2 1
32 0 41 0 0 0 2 0 127 0 40 1 0 0 2 0  222 0 33 0 1 2 2 1
33 1 46 1 1 2 2 0 128 0 40 0 0 0 2 0  223 0 40 1 1 2 2 0
34 0  5 1 1 2 3 1 129 0 39 0 0 0 2 0  224 0 40 1 1 1 2 1
35 1 29 0 0 0 2 0 130 0 40 1 0 1 2 0  225 0 40 0 0 2 2 0
36 0 31 0 1 0 2 0 131 1 32 0 0 0 2 0  226 0 28 1 0 1 2 0
37 0 22 1 1 2 2 0 132 0 46 1 0 1 2 0  227 0 47 0 0 0 2 1
38 1 22 1 1 2 2 1 133 4 39 1 1 0 2 0  228 0 18 1 1 2 2 1
39 0 47 0 0 0 2 0 134 0 37 0 0 0 2 0  229 0 45 1 0 0 2 0
40 1 46 1 1 1 2 1 135 0 51 0 0 1 2 0  230 0 35 0 0 0 2 0
41 0 37 0 0 0 2 0 136 1 39 1 1 0 2 0  231 1 17 1 0 1 1 1
42 1 39 0 0 0 2 0 137 1 34 1 1 0 2 0  232 0 40 0 0 0 2 0
```

```
43 0 33 0 1 1 2 1 138 1 14 0 1 0 1 0  233 0 29 1 1 2 2 0
44 0 34 1 0 1 2 0 139 2 15 1 0 0 2 0  234 1 35 1 1 1 2 0
45 3 32 1 1 1 2 0 140 1 34 1 1 0 2 1  235 0 40 0 0 2 2 0
46 3 22 0 0 0 2 0 141 0 43 0 1 0 2 0  236 1 22 1 1 1 2 0
47 1  6 1 0 2 3 0 142 1 33 0 0 0 2 0  237 0 42 0 0 0 2 0
48 0 38 0 0 0 2 0 143 3 34 1 0 0 2 1  238 0 34 1 1 1 2 1
49 1 43 0 1 0 2 0 144 0 48 0 0 0 2 0  239 6 38 1 0 1 2 0
50 2 36 0 1 0 2 0 145 4 26 1 1 0 2 0  240 0 25 0 0 1 2 1
51 0 43 0 0 0 2 0 146 0 30 0 1 2 2 1  241 0 39 0 1 0 2 0
52 0 24 1 0 0 2 0 147 0 41 1 1 1 2 0  242 1 35 0 1 2 2 1
53 0 25 1 0 1 2 1 148 0 34 0 1 1 2 0  243 1 36 1 1 1 2 1
54 0 41 0 0 0 2 0 149 0 43 0 1 0 2 0  244 0 23 1 0 0 2 0
55 0 43 0 0 0 2 0 150 1 31 1 0 1 2 0  245 4 30 1 1 1 2 0
56 2 31 0 1 1 2 0 151 0 26 1 0 1 2 0  246 1 41 1 1 1 2 1
57 3 28 1 1 1 2 0 152 0 37 0 0 0 2 0  247 0 37 0 1 1 2 0
58 1 22 0 0 1 2 1 153 0 44 0 0 0 2 0  248 0 46 1 1 0 2 0
59 1 11 1 1 1 1 0 154 0 40 1 0 0 2 0  249 0 45 1 1 0 2 1
60 3 41 0 1 1 2 0 155 0  8 1 1 1 3 1  250 1 38 1 1 1 2 0
61 0 31 0 0 1 2 0 156 0 40 1 1 1 2 1  251 0 10 1 1 1 1 0
62 0 11 0 0 1 1 1 157 1 45 0 0 0 2 0  252 0 30 1 1 2 2 0
63 0 44 0 1 0 2 0 158 0  4 0 0 2 3 0  253 0 32 0 1 2 2 0
64 0  9 1 0 0 3 1 159 1 36 0 1 0 2 0  254 0 46 1 0 0 2 0
65 0 36 1 1 1 2 0 160 3 37 1 1 1 2 0  255 5 35 1 1 2 2 1
66 0 29 1 0 0 2 0 161 0 15 1 0 0 1 0  256 0 44 0 0 0 2 0
67 0 27 0 1 0 2 1 162 1 27 1 0 1 2 1  257 0 41 0 1 1 2 0
68 0 36 0 1 0 2 0 163 2 31 0 1 0 2 0  258 2 36 1 0 1 2 0
69 1 33 1 0 0 2 0 164 0 42 0 0 0 2 0  259 0 34 1 1 1 2 1
70 2 13 1 1 2 1 1 165 0 42 1 0 0 2 0  260 1 30 0 1 0 2 1
71 0 38 0 0 0 2 0 166 1 38 0 0 0 2 0  261 1 27 1 0 0 2 0
72 0 41 0 0 0 2 1 167 0 44 1 0 0 2 0  262 0 48 1 0 0 2 0
73 0 41 1 0 2 2 0 168 0 45 0 0 0 2 0  263 1  6 0 1 2 3 1
74 0 35 0 0 1 2 0 169 0 34 0 1 0 2 0  264 0 38 1 1 0 2 1
75 0 45 0 0 0 2 0 170 2 41 0 0 0 2 0  265 0 29 1 1 1 2 1
76 4 38 1 0 2 2 1 171 2 30 1 1 1 2 0  266 1 43 0 1 2 2 1
77 1 42 1 0 0 2 1 172 0 44 0 0 0 2 0  267 0 43 0 1 0 2 0
78 1 42 1 1 2 2 1 173 0 40 1 0 0 2 0  268 0 37 1 0 2 2 0
79 6 36 1 1 0 2 0 174 2 31 0 0 0 2 0  269 1 23 1 1 0 2 1
80 2 23 1 1 1 2 1 175 0 41 1 0 0 2 0  270 0 44 0 0 1 2 0
81 1 32 0 0 1 2 0 176 0 41 0 0 0 2 0  271 0  5 0 1 1 3 1
82 0 41 0 1 0 2 0 177 0 39 1 0 0 2 0  272 0 25 1 0 2 2 0
83 0 50 0 0 0 2 0 178 0 40 1 0 0 2 0  273 0 25 1 0 1 2 0
84 0 42 1 1 1 2 1 179 2 35 1 0 2 2 0  274 1 28 1 1 1 2 1
```

```
85 1 30 0 0 0 2 0 180 1 43 1 0 0 2 0  275 0  7 0 1 0 3 1
86 2 47 0 1 0 2 0 181 2 39 0 0 0 2 0  276 0 32 0 0 0 2 0
87 1 35 1 1 2 2 0 182 0 35 1 1 0 2 0  277 0 41 0 0 0 2 0
88 1 38 1 0 1 2 1 183 0 37 0 0 0 2 0  278 1 33 1 1 2 2 1
89 1 38 1 1 1 2 1 184 3 37 0 0 0 2 0  279 2 36 1 1 2 2 0
90 1 38 1 1 1 2 1 185 0 43 0 0 0 2 0  280 0 31 0 0 0 2 0
91 0 32 1 1 1 2 0 186 0 42 0 0 0 2 0  281 0 18 0 0 0 2 0
92 1  3 1 0 1 3 1 187 0 42 0 0 0 2 0  282 1 32 1 0 2 2 0
93 0 26 1 0 0 2 1 188 0 38 0 0 0 2 0  283 0 22 1 1 2 2 1
94 0 35 1 0 0 2 0 189 0 36 1 0 0 2 0  284 0 35 0 0 0 2 1
95 3 37 1 0 0 2 0 190 0 39 0 1 0 2 0
;
```

1.20.2 SAS Code

```
proc genmod data=lri;                                                    ①
   class ses id race agegroup;
   model count = passive crowding ses race agegroup/
                 dist=poisson link=log offset=logrisk type3;
run;

proc genmod data=lri;                                                    ②
   class ses id race agegroup;
   model count = passive crowding ses race agegroup/
                 dist=poisson link=log offset=logrisk type3 scale=PEARSON;
run;

proc genmod data=lri;                                                    ③
   class id ses race agegroup;
   model count = passive crowding ses race agegroup /
                 dist=poisson link=log offset=logrisk type3;
   repeated subject=id / type=ind;
run;
```

① Naive Poisson approach
② Quasi-Poisson approach
③ GEE approach

1.20.3 Answer

Both the quasi-Poisson and GEE approach addressed the overdispersion issue, but the GEE approach yielded lower standard errors compared to the quasi-Poisson approach.

1.21 GEE: Other distributions

The GEE can also analyze multinomial distributions for ordinal polytomous response variables using the proportional odds model.

1.22 Example: Proportional Odds Model

A clinical trial compared two treatments for a respiratory illness (Koch et al. 1990). In each of two centers, eligible patients were randomly assigned to active treatment or placebo. During treatment, respiratory status was determined at baseline and four visits and recorded on a five-point scale of 0 for terrible to 4 for excellent. Potential explanatory variables were center, sex, and baseline respiratory status, as well as age (in years) at the time of study entry. There were 111 patients (54 active, 57 placebo), with no missing data for responses or covariates. The data is given in the RESP data set.

1.22.1 SAS Data Set

```
data resp;
   input center id treatment $ sex $ age baseline
   visit1-visit4 @@;
   visit=1;  outcome=visit1;  output;
   visit=2;  outcome=visit2;  output;
   visit=3;  outcome=visit3;  output;
   visit=4;  outcome=visit4;  output;
   datalines;
1  53 A F  32 1  2 2 4 2  2  30 A F  37 1  3 4 4 4
1  18 A F  47 2  2 3 4 4  2  52 A F  39 2  3 4 4 4
1  54 A M  11 4  4 4 4 2  2  23 A F  60 4  4 3 3 4
1  12 A M  14 2  3 3 3 2  2  54 A F  63 4  4 4 4 4
1  51 A M  15 0  2 3 3 3  2  12 A M  13 4  4 4 4 4
1  20 A M  20 3  3 2 3 1  2  10 A M  14 1  4 4 4 4
1  16 A M  22 1  2 2 2 3  2  27 A M  19 3  3 2 3 3
1  50 A M  22 2  1 3 4 4  2  16 A M  20 2  4 4 4 3
1   3 A M  23 3  3 4 4 3  2  47 A M  20 2  1 1 0 0
1  32 A M  23 2  3 4 4 4  2  29 A M  21 3  3 4 4 4
```

```
1  56 A M  25 2  3 3 2 3  2  20 A M  24 4  4 4 4 4
1  35 A M  26 1  2 2 3 2  2   2 A M  25 3  4 3 3 1
1  26 A M  26 2  2 2 2 2  2  15 A M  25 3  4 4 3 3
1  21 A M  26 2  4 1 4 2  2  25 A M  25 2  2 4 4 4
1   8 A M  28 1  2 2 1 2  2   9 A M  26 2  3 4 4 4
1  30 A M  28 0  0 1 2 1  2  49 A M  28 2  3 2 2 1
1  33 A M  30 3  3 4 4 2  2  55 A M  31 4  4 4 4 4
1  11 A M  30 3  4 4 4 3  2  43 A M  34 2  4 4 2 4
1  42 A M  31 1  2 3 1 1  2  26 A M  35 4  4 4 4 4
1   9 A M  31 3  3 4 4 4  2  14 A M  37 4  3 2 2 4
1  37 A M  31 0  2 3 2 1  2  36 A M  41 3  4 4 3 4
1  23 A M  32 3  4 4 3 3  2  51 A M  43 3  3 4 4 2
1   6 A M  34 1  1 2 1 1  2  37 A M  52 1  2 1 2 2
1  22 A M  46 4  3 4 3 4  2  19 A M  55 4  4 4 4 4
1  24 A M  48 2  3 2 0 2  2  32 A M  55 2  2 3 3 1
1  38 A M  50 2  2 2 2 2  2   3 A M  58 4  4 4 4 4
1  48 A M  57 3  3 4 3 4  2  53 A M  68 2  3 3 3 4
1   5 P F  13 4  4 4 4 4  2  28 P F  31 3  4 4 4 4
1  19 P F  31 2  1 0 2 2  2   5 P F  32 3  2 2 3 4
1  25 P F  35 1  0 0 0 0  2  21 P F  36 3  3 2 1 3
1  28 P F  36 2  3 3 2 2  2  50 P F  38 1  2 0 0 0
1  36 P F  45 2  2 2 2 1  2   1 P F  39 1  2 1 1 2
1  43 P M  13 3  4 4 4 4  2  48 P F  39 3  2 3 0 0
1  41 P M  14 2  2 1 2 3  2   7 P F  44 3  4 4 4 4
1  34 P M  15 2  2 3 3 2  2  38 P F  47 2  3 3 2 3
1  29 P M  19 2  3 3 0 0  2   8 P F  48 2  2 1 0 0
1  15 P M  20 4  4 4 4 4  2  11 P F  48 2  2 2 2 2
1  13 P M  23 3  3 1 1 1  2   4 P F  51 3  4 2 4 4
1  27 P M  23 4  4 2 4 4  2  17 P F  58 1  4 2 2 0
1  55 P M  24 3  4 4 4 3  2  39 P M  11 3  4 4 4 4
1  17 P M  25 1  1 2 2 2  2  40 P M  14 2  1 2 3 2
1  45 P M  26 2  4 2 4 3  2  24 P M  15 3  2 2 3 3
1  40 P M  26 1  2 1 2 2  2  41 P M  15 4  3 3 3 4
1  44 P M  27 1  2 2 1 2  2  33 P M  19 4  2 2 3 3
1  49 P M  27 3  3 4 3 3  2  13 P M  20 1  4 4 4 4
1  39 P M  28 2  1 1 1 1  2  34 P M  20 3  2 4 4 4
1   2 P M  28 2  0 0 0 0  2  45 P M  33 3  3 3 2 3
1  14 P M  30 1  0 0 0 0  2  22 P M  36 2  4 3 3 4
1  10 P M  37 3  2 3 3 2  2  18 P M  38 4  3 0 0 0
1  31 P M  37 1  0 0 0 0  2  35 P M  42 3  2 2 2 2
1   7 P M  43 2  3 2 4 4  2  44 P M  43 2  1 0 0 0
1  52 P M  43 1  1 1 3 2  2   6 P M  45 3  4 2 1 2
```

```
1   4 P M  44 3  4 3 4 2  2  46 P M  48 4  4 0 0 0
1   1 P M  46 2  2 2 2 2  2  31 P M  52 2  3 4 3 4
1  46 P M  49 2  2 2 2 2  2  42 P M  66 3  3 3 4 4
1  47 P M  63 2  2 2 2 2
;
```

1.22.2 SAS Code

```
proc genmod data=resp descending;
class id center sex treatment visit;
model outcome = treatment sex center age baseline
visit visit*treatment /
link=clogit dist=mult type3;
repeated subject=id*center / type=ind;
lsmeans visit*treatment/ilink cl;
slice visit*treatment/diff=all cl exp adjust=tukey;
run;
```

1.22.3 Answer

There is a significant interaction between treatment and visit (χ^2 =10.47, p=0.02). There seems to be no difference between visits for the active treatment (χ^2 = 5.4, p=0.15) and placebo treatment (χ^2 = 5.8, p=0.12). However, there is a difference between active and placebo treatment at all visits. At visit 1, the odds of a higher score for active treatment is 108% higher than the placebo treatment (OR = 2.08, 95% CI: 1.11, 3.90). At visit 2, the odds ratio increases even more to 5.86 (2.81, 12.27). At visit 3, the odds ratio decreases slightly to 3.95 (1.84, 8.47).

1.23 Missing Data

Recall: GEE can handle missing data that is missing completely at random (MCAR). The GEE can still calculate usable statistics and p-values for statistical inference.

1.24 Example: Missing Data

Consider a two-period crossover study on treatments for a skin disorder where patients were given sequences of the standard drug A, a new drug B, and a placebo. Investigators introduced a skin irritant and then applied topical treatments. Subjects were stratified by gender. The first session included 300 patients, but 50 patients failed to attend the second session one week later. Investigators determined that none of the losses to follow-up were actually due to the failure of the treatments, but they were due to the usual attrition plus a breakdown in the communication to emphasize the importance of the return visit. For this reason, although much more missing data occurred than was expected, the analysis proceeded. The SKINCROSS2 dataset includes the data from the study.

1.24.1 SAS Data Set

```
data skincross;
    input subject gender $ sequence $ Time1 $ Time2 $ @@;
    datalines;
  1  m AB    Y    Y  101 m   PA   Y    Y     201 f  AP    Y    Y
  2  m AB    Y    .  102 m   PA   Y    Y     202 f  AP    Y    Y
  3  m AB    Y    Y  103 m   PA   Y    Y     203 f  AP    Y    Y
  4  m AB    Y    .  104 m   PA   Y    Y     204 f  AP    Y    Y
  5  m AB    Y    Y  105 m   PA   Y    Y     205 f  AP    Y    Y
  6  m AB    Y    .  106 m   PA   Y    N     206 f  AP    Y    Y
  7  m AB    Y    .  107 m   PA   Y    .     207 f  AP    Y    Y
  8  m AB    Y    Y  108 m   PA   Y    N     208 f  AP    Y    Y
  9  m AB    Y    Y  109 m   PA   N    .     209 f  AP    Y    Y
 10  m AB    Y    Y  110 m   PA   N    Y     210 f  AP    Y    Y
 11  m AB    Y    .  111 m   PA   N    Y     211 f  AP    Y    Y
 12  m AB    Y    Y  112 m   PA   N    Y     212 f  AP    Y    Y
 13  m AB    Y    N  113 m   PA   N    .     213 f  AP    Y    Y
 14  m AB    Y    N  114 m   PA   N    .     214 f  AP    Y    .
 15  m AB    Y    N  115 m   PA   N    Y     215 f  AP    Y    .
 16  m AB    Y    N  116 m   PA   N    Y     216 f  AP    Y    .
 17  m AB    Y    N  117 m   PA   N    Y     217 f  AP    Y    Y
 18  m AB    Y    N  118 m   PA   N    Y     218 f  AP    Y    Y
 19  m AB    Y    .  119 m   PA   N    Y     219 f  AP    Y    Y
 20  m AB    Y    N  120 m   PA   N    Y     220 f  AP    Y    Y
 21  m AB    Y    N  121 m   PA   N    Y     221 f  AP    Y    .
 22  m AB    Y    N  122 m   PA   N    Y     222 f  AP    Y    Y
 23  m AB    Y    .  123 m   PA   N    Y     223 f  AP    Y    Y
 24  m AB    Y    N  124 m   PA   N    Y     224 f  AP    Y    Y
 25  m AB    N    Y  125 m   PA   N    Y     225 f  AP    Y    Y
 26  m AB    N    .  126 m   PA   N    Y     226 f  AP    Y    N
 27  m AB    N    .  127 m   PA   N    Y     227 f  AP    Y    N
 28  m AB    N    .  128 m   PA   N    Y     228 f  AP    Y    N
 29  m AB    N    Y  129 m   PA   N    Y     229 f  AP    Y    .
 30  m AB    N    Y  130 m   PA   N    Y     230 f  AP    Y    N
```

31	m	AB	N	.	131	m	PA	N	.	231	f	AP	Y	N
32	m	AB	N	N	132	m	PA	N	N	232	f	AP	N	Y
33	m	AB	N	N	133	m	PA	N	N	233	f	AP	N	Y
34	m	AB	N	N	134	m	PA	N	N	234	f	AP	N	Y
35	m	AB	N	N	135	m	PA	N	N	235	f	AP	N	Y
36	m	AB	N	N	136	m	PA	N	.	236	f	AP	N	Y
37	m	AB	N	N	137	m	PA	N	N	237	f	AP	N	Y
38	m	AB	N	N	138	m	PA	N	N	238	f	AP	N	N
39	m	AB	N	N	139	m	PA	N	.	239	f	AP	N	N
40	m	AB	N	N	140	m	PA	N	N	240	f	AP	N	N
41	m	AB	N	.	141	m	PA	N	N	241	f	AP	N	N
42	m	AB	N	N	142	m	PA	N	N	242	f	AP	N	N
43	m	AB	N	N	143	m	PA	N	N	243	f	AP	N	N
44	m	AB	N	N	144	m	PA	N	N	244	f	AP	N	N
45	m	AB	N	.	145	m	PA	N	N	245	f	AP	N	N
46	m	AB	N	N	146	m	PA	N	N	246	f	AP	N	N
47	m	AB	N	N	147	m	PA	N	N	247	f	AP	N	N
48	m	AB	N	N	148	m	PA	N	N	248	f	AP	N	N
49	m	AB	N	N	149	m	PA	N	N	249	f	AP	N	N
50	m	AB	N	N	150	m	PA	N	.	250	f	AP	N	N
51	m	BP	Y	Y	151	f	BA	Y	Y	251	f	PB	Y	.
52	m	BP	Y	Y	152	f	BA	Y	Y	252	f	PB	Y	Y
53	m	BP	Y	Y	153	f	BA	Y	Y	253	f	PB	Y	Y
54	m	BP	Y	Y	154	f	BA	Y	.	254	f	PB	Y	Y
55	m	BP	Y	Y	155	f	BA	Y	Y	255	f	PB	Y	Y
56	m	BP	Y	Y	156	f	BA	Y	Y	256	f	PB	Y	.
57	m	BP	Y	Y	157	f	BA	Y	Y	257	f	PB	Y	Y
58	m	BP	Y	Y	158	f	BA	Y	Y	258	f	PB	Y	.
59	m	BP	Y	N	159	f	BA	Y	Y	259	f	PB	Y	Y
60	m	BP	Y	.	160	f	BA	Y	Y	260	f	PB	Y	Y
61	m	BP	Y	N	161	f	BA	Y	.	261	f	PB	Y	Y
62	m	BP	Y	.	162	f	BA	Y	.	262	f	PB	Y	.
63	m	BP	Y	N	163	f	BA	Y	Y	263	f	PB	Y	.
64	m	BP	N	Y	164	f	BA	Y	Y	264	f	PB	Y	N
65	m	BP	N	Y	165	f	BA	Y	Y	265	f	PB	Y	N
66	m	BP	N	Y	166	f	BA	Y	Y	266	f	PB	Y	N
67	m	BP	N	Y	167	f	BA	Y	Y	267	f	PB	Y	N
68	m	BP	N	Y	168	f	BA	Y	Y	268	f	PB	Y	N
69	m	BP	N	Y	169	f	BA	Y	Y	269	f	PB	N	Y
70	m	BP	N	.	170	f	BA	Y	.	270	f	PB	N	Y
71	m	BP	N	N	171	f	BA	Y	N	271	f	PB	N	Y
72	m	BP	N	N	172	f	BA	Y	N	272	f	PB	N	.

```
 73   m BP   N   N  173 f   BA  N   Y    273 f   PB    N   .
 74   m BP   N   N  174 f   BA  N   Y    274 f   PB    N   Y
 75   m BP   N   N  175 f   BA  N   .    275 f   PB    N   Y
 76   m BP   N   N  176 f   BA  N   Y    276 f   PB    N   Y
 77   m BP   N   N  177 f   BA  N   Y    277 f   PB    N   .
 78   m BP   N   N  178 f   BA  N   Y    278 f   PB    N   Y
 79   m BP   N   N  179 f   BA  N   .    279 f   PB    N   Y
 80   m BP   N   N  180 f   BA  N   .    280 f   PB    N   Y
 81   m BP   N   .  181 f   BA  N   Y    281 f   PB    N   Y
 82   m BP   N   N  182 f   BA  N   Y    282 f   PB    N   Y
 83   m BP   N   .  183 f   BA  N   Y    283 f   PB    N   Y
 84   m BP   N   N  184 f   BA  N   Y    284 f   PB    N   Y
 85   m BP   N   .  185 f   BA  N   Y    285 f   PB    N   Y
 86   m BP   N   N  186 f   BA  N   Y    286 f   PB    N   Y
 87   m BP   N   N  187 f   BA  N   Y    287 f   PB    N   Y
 88   m BP   N   N  188 f   BA  N   Y    288 f   PB    N   Y
 89   m BP   N   N  189 f   BA  N   Y    289 f   PB    N   Y
 90   m BP   N   N  190 f   BA  N   Y    290 f   PB    N   N
 91   m BP   N   N  191 f   BA  N   Y    291 f   PB    N   N
 92   m BP   N   .  192 f   BA  N   Y    292 f   PB    N   .
 93   m BP   N   N  193 f   BA  N   Y    293 f   PB    N   N
 94   m BP   N   N  194 f   BA  N   Y    294 f   PB    N   N
 95   m BP   N   N  195 f   BA  N   Y    295 f   PB    N   N
 96   m BP   N   N  196 f   BA  N   Y    296 f   PB    N   N
 97   m BP   N   N  197 f   BA  N   Y    297 f   PB    N   N
 98   m BP   N   N  198 f   BA  N   N    298 f   PB    N   N
 99   m BP   N   N  199 f   BA  N   N    299 f   PB    N   N
100   m BP   N   .  200 f   BA  N   N    300 f   PB    N   N
;
data skincross2;
   set skincross;
   period=1;
   treatment=substr(sequence, 1, 1);
   carry='N';
   response=Time1;
   output;
   period=2;
   Treatment=substr(sequence, 2, 1);
   carry = substr(sequence, 1, 1);
   if carry='P' then carry='N';
   response=Time2;
   output;
```

```
run;
```

1.24.2 SAS Code

```
proc genmod data=skincross2 descending;
   class subject treatment period gender carry;
   model response = treatment period gender carry
                 gender*period /type3
                 dist=bin link=logit;
   repeated subject=subject / type=exch;
run;

proc genmod data=skincross2 descending;
   class subject treatment period gender;
   model response = treatment period gender gender*period
                    /type3
                    dist=bin link=logit;
   repeated subject=subject / type=exch;
   estimate 'OR:A-B' treatment 1 -1 0 /exp;
   estimate 'OR:A-P' treatment 1 0 -1 / exp;
   estimate 'OR:B-P' treatment 0 1 -1 / exp;
   lsmeans treatment/cl diff adjust=tukey ilink exp;
run;
```

1.24.3 Answer

The carryover effect is not significant (χ^2 = 1.08, p=0.58). Running the analysis without the carryover effect shows that there is strong evidence of an interaction between period and gender (χ^2 = 3.93,p=0.047). There is also a significant effect of the treatment. (χ^2 = 40.02, p<0.0001). Upon further inspection, we determine that treatment A and P are significantly different (OR: 2.97, 95% CI:1.78-4.97), A and B are significantly different (OR: 3.53, 95% CI: 2.17, 5.75), but B and P are not significantly different (OR: 1.19, 95% CI: 0.72, 1.96).