EAB 770 - Chapter 14

Miguel Fudolig

1 Weighted Least Squares

1.1 Outline

- Weighted Least Squares
  - Methodology and Framework
- Modeling of Means
- Survey Sample Data
- Repeated Measures

1.2 Weighted Least Squares

- Commonly used in modeling mean scores, proportions, and functions.
  - General idea: Model the distribution of the response variable across levels of the explanatory variables.
- In a contingency table:
  - Categorical response: columns
  - Explanatory variables: rows

Table 14.2 General Contingency Table

Group	Response				Total
	1	2	$\cdots$	r	
1	n_{11}	n_{12}	$\cdots$	n_{1r}	n_{1+}
2	n_{21}	n_{22}	$\cdots$	n_{2r}	n_{2+}
$\cdots$	$\cdots$	$\cdots$	$\cdots$	$\cdots$	$\cdots$
s	n_{s1}	n_{s2}	$\cdots$	n_{sr}	n_{s+}

> **Note**
>
> Each row is also referred to as subpopulations.

1.3 Weighted Least Squares

1.4 Weighted Least Squares Framework

The proportion of subjects in each group who have each response is written as

$$p_{ij} = n_{ij}/n_{i+}$$

We consider the rows of the contingency table to be simple random samples from the multinomial distribution.

> **Note**
>
> Groups are assumed to be independent, so the contingency table is distributed as product multinomial. This assumption enables us to create an estimated covariance matrix for proportions in each row.

1.5 Response Functions

Response functions are similar to link functions in previous chapters. These response functions could be `means`, `logits`, `cumulative logits` or `marginal` probabilities.

> **Note**
>
> The weighted least squares can model the following:
>
> - Estimated proportions p_{ij}
> - Mean scores, which are simple linear functions of the proportions (weighted means)
> - Transformations such as logits, cumulative logits, or marginal probabilities, kappa statistic, etc.

1.6 Saturated Model

We are interested in modeling the saturated model where there are as many parameters as there are response functions.

> **Note**
>
> If groups have sufficient sample size, then the variation among the response functions can be investigated by fitting linear regression models with weighted least squares.

1.7 Weighted Least Squares in SAS

- The `CATMOD` procedure is used to model categorical data using WLS.
  - `WEIGHT` statement is similar to PROC FREQ.
  - `RESPONSE` statement tells the `CATMOD` procedure what the response functions are.
    - `MEANS`
    - `LOGITS`
    - `CLOGITS`: cumulative logit
    - `MARGINAL`: marginal probabilities
  - `MODEL` statement is similar to the `LOGISTIC` and `GENMOD` syntax.
    - `FREQ` option shows the frequencies of the underlying contingency table.
    - `PROB` option shows the frequencies of the underlying contingency table.
    - `DESIGN` option prints the response functions and model matrix.
  - `POPULATION` statement defines the variables that define the population groups.

1.8 Example

The data set COLDS shows the number of periods with colds in children. We want to test whether there is an effect of sex and residence on the mean number of periods with a cold.

1.8.1 SAS Data Set

```
data colds;
   input sex $ residence $ periods count @@;
   datalines;
female rural 0 45 female rural 1 64  female rural 2 71
female urban 0 80 female urban 1 104 female urban 2 116
male rural   0 84 male   rural 1 124 male   rural 2 82
male urban   0 106 male  urban 1 117 male   urban 2 87
run;
```

1.8.2 SAS Code

```
proc catmod data=colds;
   population sex residence;
   weight count;
   response means;
   model periods = sex residence sex*residence/freq prob design;
   contrast "male vs. female in urban" sex 2 sex*residence -2;         (1)
   contrast "male vs. female in urban" all_parms 0 2 0 -2;
   contrast "female: rural vs urban" residence 2 sex*residence 2;
run;
```

(1) The numbers are based on the design matrix outputted by `catmod`.

1.8.3 Answer

There is no significant interaction between sex and residence (p=0.76). There is a significant effect of sex ($\chi^2 = 11.57$, p=0.007).

1.9 Example

The data SELECTION_CONFIDENCE shows the self-reported confidence of working professionals in interpreting a bar chart provided to respondents. Use an interaction model to test whether educational attainment (graduate or non-graduate level) or STEM association (STEM degree or non-STEM) has an effect in the perception of confidence in assessing data visualization.

- -1: Not confident, 0: neutral, 1: confident

1.9.1 SAS Data Set

```
data selection_confidence;
input stem $ graduate $ confidence $ count @@;
datalines;
yes yes NC 28 yes yes N 30 yes yes C 43
yes no NC 17 yes no N 25 yes no C 15
no yes NC 30 no yes N 10 no yes C 17
no no NC 8 no no N 10 no no C 38
;
```

1.9.2 SAS Code

```
proc catmod data=selection_confidence order=data;
population stem graduate;
weight count;
response -1 0 1;                                                    (1)
model confidence= stem graduate stem*graduate/freq prob design;
contrast "graduate level: stem vs non-stem" all_parms 0 2 0 2;
contrast "non-graduate: stem vs non-stem" all_parms 0 2 0 -2;
contrast "STEM: graduate degree vs. non-grad" all_parms 0 0 2 2;
contrast "non-STEM: graduate degree vs. non-grad" all_parms 0 0 2 -2;
run;
```

(1) Alternatively,you can use `response means;`

1.9.3 Answer

There is a significant interaction between STEM association and educational attainment ($\chi^2 =$ 22.63,p<0.0001). There is no significant difference between STEM graduate and non-graduate degree holders (χ^2= 2.03, p =0.16), but there is a significant difference between non-STEM graduate and non-graduate degree holders ($\chi^2 = 25.26$, p<0.0001).

1.10 Weighted Least Squares: Analyzing Survey Data

- The `CATMOD` procedure can also be used to analyze complex surveys.
  - If the data given are the actual response functions, we need to identify the various groups in PROC CATMOD to express cross-classification relationships.

- The FACTOR statement can specify the internal variables that define the grouping.
- The *RESPONSE* option specifies the desired model effects
- PROFILE lists the values of the internal variables.

1.11 Example

The dataset WBEING shows the ordered categorical estimates, covariance matrix, and standard errors of the well-being index for a cross-classification based on sex and age. Use the WLS method to estimate a main effect model and test whether there is an effect of sex and age group on the well-being index scores.

1.11.1 SAS Data Set

```
data wbeing(type=est);
   input   b1-b5   _type_ $  _name_ $  b6-b10 #2;
   datalines;
 7.93726   7.92509   7.82815   7.73696   8.16791  parms    .
 7.24978   7.18991   7.35960   7.31937   7.55184
 0.00739   0.00019   0.00146  -0.00082   0.00076  cov      b1
 0.00189   0.00118   0.00140  -0.00140   0.00039
 0.00019   0.01172   0.00183   0.00029   0.00083  cov      b2
-0.00123  -0.00629  -0.00088  -0.00232   0.00034
 0.00146   0.00183   0.01050  -0.00173   0.00011  cov      b3
 0.00434  -0.00059  -0.00055   0.00023  -0.00013
-0.00082   0.00029  -0.00173   0.01335   0.00140  cov      b4
 0.00158   0.00212   0.00211   0.00066   0.00240
 0.00076   0.00083   0.00011   0.00140   0.01430  cov      b5
-0.00050  -0.00098   0.00239  -0.00010   0.00213
 0.00189  -0.00123   0.00434   0.00158  -0.00050  cov      b6
 0.01110   0.00101   0.00177  -0.00018  -0.00082
 0.00118  -0.00629  -0.00059   0.00212  -0.00098  cov      b7
 0.00101   0.02342   0.00144   0.00369   0.00253
 0.00140  -0.00088  -0.00055   0.00211   0.00239  cov      b8
 0.00177   0.00144   0.01060   0.00157   0.00226
-0.00140  -0.00232   0.00023   0.00066  -0.00010  cov      b9
-0.00018   0.00369   0.00157   0.02298   0.00918
 0.00039   0.00034  -0.00013   0.00240   0.00213  cov     b10
-0.00082   0.00253   0.00226   0.00918   0.01921
;
```

1.11.2 SAS Code

```
proc catmod data=wbeing;
response read b1-b10;
factors sex $ 2, age $ 5 /
_response_ = sex age
profile = (male '25-34' ,
male '35-44',
male '45-54' ,
male '55-64',
male '65-74' ,
female '25-34',
female '35-44',
female '45-54',
female '55-64' ,
female '65-74');
model _f_ = _response_/design;
contrast '25-34 vs. 35-44' all_parms 0 0 1 -1 0 0;
contrast '25-34 vs. 45-54' all_parms 0 0 1 0 -1 0;
contrast '25-34 vs. 55-64' all_parms 0 0 1 0 0 -1;
contrast '25-34 vs. 65-74' all_parms 0 0 2 1 1 1;
run;
```

1.11.3 Answer

There is a significant effect of sex ($\chi^2 = 47.1$, p<0.001) and moderate evidence of an effect of age ($\chi^2 = 9.21$, p=0.06).

1.12 Weighted Least Squares: Repeated Measures

The Weighted Least Squares can also be used when there are multiple response functions per group.

> **Note**
>
> If there is a category response variable with c levels measured at t time points, the cross-classification of possible outcomes results in $r = c^t$ response profiles.
> For marginal proportions, generalized logits, or cumulative logits, we have $c(t-1)$ profiles.
> For correlated mean scores for ordinal responses, we have t correlated mean scores.

SAS Implementation

The REPEATED statement indicates that we are modeling repeated measures data.

- Can specify a specific model matrix
- Can specify a name for each repeated measurement factor
- Can specify type, number of levels, and labels for each level.
- The factors can be named anything as long as it is identical to another variable name in the model.

1.13 Weighted Least Squares: Repeated Measures

Let's consider different cases:

- One Population, Dichotomous Response
- Two Populations, Polytomous Response
- One Population Regression Analysis of Logits

1.14 Example

The data set below includes data from 46 subjects who were treated with three Drugs (A, B, and C). The response to each drug was recorded as favorable or unfavorable. We want to test the hypothesis that the probability of a favorable response is the same for all three drugs.

1.14.1 SAS Data Set

```
data drug;
input druga $ drugb $ drugc $ count;
datalines;
F F F 6
F F U 16
F U F 2
F U U 4
U F F 2
U F U 4
U U F 6
U U U 6
;
```

1.14.2 SAS Code

```
proc catmod data=drug;
weight count;
response marginals;
model druga*drugb*drugc=_response_ / design oneway cov; *oneway displays the one-way frequenc
repeated drug 3 / _response_=drug;
contrast 'A versus B' all_parms 0 1 -1;
contrast 'A versus C' all_parms 0 2 1;
run;
```

1.14.3 Answer

There is strong evidence of a drug effect (χ^2 = 6.58, p=0.04). Upon further investigation, there is no significant difference between A and B, but there is strong evidence of a difference between the rate of favorable response in drugs A and C (χ^2= 5.79, p=0.02).

1.15 Example

The data includes the unaided distance vision data from 30-39 year-old employees of United Kingdom Royal Ordnance factories during the years 1943-1946. We are interested in determining whether the marginal vision grade distributions are the same in the right eye as in the left eye, whether the marginal probabilities for each score differ between females and males, and whether differences between right eye and left eye vision are the same for females and males.

1.15.1 SAS Data Set

```
data vision;
   input gender $ right left count;
   datalines;
F 1 1 1520
F 1 2  266
F 1 3  124
F 1 4   66
F 2 1  234
F 2 2 1512
F 2 3  432
F 2 4   78
F 3 1  117
```

```
   F 3 2  362
   F 3 3 1772
   F 3 4  205
   F 4 1   36
   F 4 2   82
   F 4 3  179
   F 4 4  492
   M 1 1  821
   M 1 2  112
   M 1 3   85
   M 1 4   35
   M 2 1  116
   M 2 2  494
   M 2 3  145
   M 2 4   27
   M 3 1   72
   M 3 2  151
   M 3 3  583
   M 3 4   87
   M 4 1   43
   M 4 2   34
   M 4 3  106
   M 4 4  331
   ;
```

1.15.2 SAS Code

```
proc catmod data=vision;
   weight count;
   population gender;
   response marginals;
   model right*left=gender _response_(gender='F')
                           _response_(gender='M') / design; *_response_(gender='F') means the
   repeated eye 2;
contrast 'Interaction' all_parms 0 0 0 0 0 0 1 0 0 -1  0  0,
                       all_parms 0 0 0 0 0 0 0 1 0  0 -1  0,
                       all_parms 0 0 0 0 0 0 0 0 1  0  0 -1;
run;
```

1.15.3 Answer

There is a significant nested effect of the eye location (right/left) in the vision of the female group ($\chi^2 = 11.98$, p=0.008), but not for males ($\chi^2 = 3.68$, p=0.30). The vision levels also differ between genders ($\chi^2 = 142$, p<0.001).